


The logo for K Power, featuring a large, stylized 'K' in red and orange, with the word 'POWER' in a similar color scheme below it.

COMPUTER CRAZINESS

PAUL SOMERSON AND STEPHEN MANES

For the Commodore 64, VIC 20; Apple II, II Plus, IIe; IBM PC, PCjr; Atari 400, 800, XL Series; Radio Shack Color Computer; and TI 99/4A.





COMPUTER CRAZINESS

PAUL SOMERSON AND STEPHEN MANES

A H A R D / S O F T P R E S S B O O K



SCHOLASTIC INC.
New York Toronto London Auckland Sydney Tokyo

To Jean, paragon of patience

No part of this publication may be reproduced in whole or in part, or stored in a retrieval system, or transmitted in any form or by any means, electronic, mechanical, photocopying, recording, or otherwise, without written permission, of the publisher. For information regarding permission write to Scholastic Inc., 730 Broadway, New York, NY 10003.

ISBN 0-590-33175-2

Copyright © 1984 Hard/Soft Inc. All rights reserved.
Published by Scholastic Inc.

Program adaptations by McMullen & McMullen, Inc.

Designed by Gene Siegel

12 11 10 9 8 7 6 5 4 3 2 1

6

4 5 6 7 8/8

Junior High School Library
Beaver Dam, WI 53005

Printed in U.S.A.

14

Hi There, Bozo!

Did you know your computer can do amazing magic tricks?

Can you keep your car from crashing on the world's fastest Grand Prix course?

Have you ever given your parents a report card?

Would you like your computer to grant your wishes?

With this book and your home computer you'll be able to do all these things — and a whole lot more!

Just be sure to read our NOT-SO-CRAZY TIPS AND TRICKS before you begin!

NOT-SO-CRAZY TIPS AND TRICKS:

The mad programmers answer your questions

You may be tempted to skip this section and go right on to the programs. You know what we say to that? **DON'T!!!!**

Running BASIC programs is fun. So is typing them in. But there are times when programming can drive *you* crazy. In this section, we'll give you lots of tips to help you keep your sanity!

Will these programs run on my computer?

If you have an IBM Personal Computer or IBM PCjr, the answer is YES!
All you have to do is type in the Program Listing. You shouldn't need to make any changes at all.

If you have any of the following computers, the answer is still YES!

- Apple II
- Apple II Plus
- Apple IIe
- Atari 400, 800 or XL Series
- Coleco ADAM
- Commodore 64 or VIC-20
- Radio Shack TRS-80 Color Computer
- Texas Instruments 99/4A

With these computers, you'll usually have to change the Program Listing slightly. Most of the time only a line or two will be different.

Just be sure to read the "If You Have . . ." section that appears after every Program Listing. It always tells you *exactly* what changes you'll need to make for your computer. (The ADAM needs special changes. Keep reading to learn how to make them.)

It's a good idea to read the "If You Have . . ." information *before* you start typing in your program. If this is your own copy of *Computer Crazy*, you may want to use a pen or pencil to mark the Program Listing with the changes you'll need.

There's more information about each of these computers at the end of this section. Be sure to read it before starting out!

If you have any other computer, the answer is still YES! But you may have to experiment until you find the exact changes your machine needs to run each program. That should be simple. We've tried hard to use only the parts of BASIC that are available on most machines.

Will my program run as soon as I type it in?

If a program runs as soon as you type it in, you're an absolutely amazing typist — and a very lucky one! Everybody makes little typing mistakes — even us experts! But in a computer program, little mistakes can cause BIG trouble.

If a program doesn't seem to run right, or if you get an error message — something like "Syntax Error" or "SN?" or "You Dummy!" — you'll have to go hunting for your mistakes. Programmers call them bugs. Remember, you put them there, and your program won't run right until you squish out every last one! The official word for that is "debugging."

How do I get the bugs out of my program?

When you type in a program, you must type each line exactly as it appears in the listing. If you miss even one space or one comma, your program probably won't run the way it should.

First, check your program line by line against the listing in the book. It's a good idea to LIST only a few lines at a time, using commands like LIST 100-150. That way, the whole LIST won't scroll up your screen before you get a chance to read it.

You'll probably slap your head when you discover the idiotic goofs you made. With some computers, you'll have to type the whole line in again to make a correction. With others, you can just type in the changes. Learn which way your computer works: It'll save you lots of typing.

Second, be sure you've made the changes your particular machine needs. *Always* look at the "If You Have . . ." section for your particular computer.

If your program still doesn't run, don't give up! There are plenty of things you may have done wrong. How do we know? Because we do them all the time ourselves! Here are some things to watch out for:

Rotten Leftovers

When you start typing in a program, be sure to type NEW and hit the ENTER (or RETURN) key. If you don't, you may find leftover lines from an old program popping up in your new one. It can be a real mess to get them out!

The Curse of the Missing Line Number

It's easy to skip a line by accident when you're typing a long program. Remember: Every program in this book includes a REM statement on line 10 with the name of the program. The next line in every program is line 100, and the lines count up by tens from there, like this:

```
10 REM USELESS PROGRAM
100 INPUT X
110 INPUT Y
120 PRINT Z
```

The only line numbers that won't end in 0 will be lines you add from the "If You Have . . ." section. And the only time the lines will skip — say from 240 to 260 — is when the "If You Have . . ." section tells you to remove a line for your machine.

One Out of Two

Some things have to come in pairs, or your computer will get terribly upset. What kinds of things? Mainly quotation marks (") and left and right parentheses (()). Leaving one out is just about guaranteed to get you some sort of error message. That's why it's important to check a statement like

```
300 PRINT CHR$(INT(RND(1)*(3+2)/(4+3)))
```

very carefully. If you're not positive, count the parentheses and quotation marks on your *other* personal computer — your fingers!

A Letter's Not a Number

Your computer is very, very fussy about letters and numbers. Remember, a one (1) is not a small L (l) or a capital i (I), even though they may look

almost the same. And as far as the computer is concerned, a capital letter O (O) is not the same as the number zero (0).

In our Program Listings, you'll always find a slash through the number zero (0) to keep you from getting confused. But your fingers may do something your brain never had in mind!

Here are some program lines that will cause your computer to scratch its head and give up. We stole them from our own wastebaskets. Each line contains just one error. Can you spot the goofs?

```
100 GOT0 240
200 IF I=2 THEN 500
300 N=J
400 R=22I
500 P=4I6
```

A Space Isn't Nothing!

Especially in graphics programs — programs that draw pictures — spaces are terribly important. If the Program Listing shows a space, be sure you don't leave it out. For example, in

```
100 F$="MURBLEHEAD"
110 PRINT F$;" IS A JERK!"
```

if you leave out the space between the first quotation mark and the word IS in line 110, you'll get this:

```
MURBLEHEADIS A JERK!
```

You can probably guess what prints out if you leave the space in.

But don't add spaces where there are none in the listing. That may cause problems that are even worse!

Semicolons and Commas

Semicolons and commas have special jobs in BASIC programs. Semicolons let you PRINT more than one thing on each line. In PRINT statements, commas do the same thing a little differently, but the programs in this book hardly ever use them that way.

In DATA statements, commas separate each item of data. If you leave one out or put an extra one in, it will confuse your computer. If you get a message like "Out of Data" when everything in your program looks right, chances are you left out a comma — or perhaps a whole DATA statement.

Mysterious Letters

Sometimes the letters and names used for variables in a program can get confusing. It's easy to type an E when you meant to type a D. They're right next to each other on the keyboard. So watch out! One mistake like that can keep an almost perfect program from running at all.

A Real Live Bug?

If you've done everything you can think of, and you're positive we've missed a bug in our own program, please write and describe the problem. We'll try to come up with a solution!

Why did you include the Sample Runs?

The Sample Runs are here for two reasons. First, they give you an idea of what the programs will do. That way, you can decide whether or not you want to type in a particular program.

The Sample Runs are also here to help you make sure your program is running the way it should. But remember: There are some things you can do on a computer that you can't do on a printed page. Many of these programs create pictures that move across your screen. There's no way we can show that here.

And some of the programs do different things every time you run them. So when you run your program, it may not do *exactly* what the Sample Run does — but it should come close.

How come the program listings are in ALL CAPITALS?

If you've already peeked at the Program Listings, you'll notice they're entirely in capital letters. That's because some computers only have capitals. If your machine is one of them, you won't have any problems with any of these programs.

But if you've got a computer that can give you both capitals *and* lower-case letters, you may run into some problems when you RUN these programs. Why? Let's say we ask a question that has a yes or no answer. You type in "yes" — all lowercase. But the program will be looking for "YES" — all capitals.

The solution? It's easy. Just press the key called CAPS LOCK or ALPHA LOCK or just plain LOCK on your computer. Do it when you type in your programs and when you run them. That way, all the letters will appear in CAPITALS!

What's an ENTER key? My computer doesn't have one!

One of the strangest and most important keys on the keyboard is the one you use to INPUT data. The strange part is that computer companies

can't agree on what to call it. Some call it the ENTER key. Some call it the RETURN key. And some put a weird symbol on it and no lettering at all!

It's always in roughly the same place — to the right of the letters on the keyboard. Our Program Listings always call it the ENTER key. If you've got a RETURN key, you'll need to type that word into your PRINT statements wherever we've used the word ENTER.

That brings up another rule we've used in this book. Whenever a program asks you to answer a question or make a choice, you have to hit the ENTER (or RETURN) key before anything will happen. That means you can use the BACKSPACE key to correct any mistakes before you ENTER your answer into the computer.

If you're a good programmer, you'll also notice that we've tried hard to keep you from entering an answer the computer isn't expecting. If we ask for a number between 1 and 4, you won't be able to enter 5 — or 0!

My machine doesn't have a BREAK key. Or does it?

The BREAK key is something else that's different on every machine. Sometimes it's just a key marked BREAK. Sometimes it's really two keys — CTRL and BREAK or some other combination that you have to hold down at the same time. We'll tell you exactly which combination works for your computer in a second. Be sure you know how to use it!

Why? Well, some of our programs will run forever unless you BREAK out of them — or turn off your machine! Sometimes a typing error will make the same thing happen when it's not supposed to. And when you're debugging a program, you may only want to run a small part of it and then quit. It's the same with computers as it is with cars — knowing how to get your machine to stop when you want it to is a really lucky BREAK.

How can I keep my programs from disappearing when I turn off the computer?

Many of the programs in this book will take you only a few minutes to type in. The long ones may take a few hours. And once you turn your computer off, all your hard work will disappear.

But if you have a cassette recorder or disk drive, you can SAVE your programs and RUN them again and again. SAVES are different on almost every machine. The instruction manual that came with your computer or disk drive should tell you how to SAVE your work — and LOAD it back into your computer again.

You don't have to wait until a program is working to SAVE it. Especially with longer programs, you should SAVE now and then as you type the program in. That way, you'll be able to LOAD everything you've done right back into your machine if your dog or cat or little sister accidentally pulls the plug — or if you've had enough of programming for one day and decide to go to the movies.

If you've got a friend with the same kind of computer as yours, you can each type in different programs and SAVE them on disk or cassette. Then you can swap your disks or cassettes. It's another great way to save yourself some typing!

I've got a printer. What about me?

You're really in luck! With a little thought, you'll be able to print out your very own personalized program runs. Do computer-makers agree on the way to get information to your printer? Not any more than they agree on the name of the ENTER (RETURN?) key!

With some machines, it's easy to print out your program runs. All you do is type something like PR#1, and everything you see on the screen will magically turn up on your printer, too!

With other machines it's a little trickier. You may have to change PRINT statements to LPRINT to get your printer to notice them. The best advice we can give is to check the manuals for your computer.

But remember this: If you try to run a program and absolutely *nothing* happens, it's a good bet that you tried to send something to your printer — and your printer wasn't turned on! On most machines, that's an excellent way to stop a program cold!

I can think of a better way to do some of the things you did. Why did you do them your way?

There are usually lots of different ways to get your computer to do the same thing. If you run this program:

```
10 PRINT "JESKLER LOVES PRUNES"  
20 PRINT "JESKLER LOVES PRUNES"
```

you'll get exactly the same result as:

```
10 FOR A=1 TO 2  
20 PRINT "JESKLER LOVES PRUNES"  
30 NEXT A
```

We can think of at least ten other ways to get the same two lines printed out on the screen.

Which way is best? It all depends. If we've done something that looks silly to you, it may be because we've tried to get it to work with machines that can't do some things yours can.

We've tried to write our programs so they'll run on as many machines as possible. If we've left out your favorite BASIC command or done something in

a roundabout way, we apologize. But if we've done something *really* stupid, let us know about it!

Will I mess up my computer if I make changes in your programs?

No! In fact, we *hope* you'll try to improve on these programs — and use them as a starting point for your own. But it's probably a good idea to get them running in their original versions first.

How do I run a program once I've typed it in?

Just type RUN and hit the ENTER or RETURN key!

What else do I have to read in this chapter?

Just the section on your own computer. After that — happy programming!

What if I have an IBM Personal Computer or PCjr?

You're in luck! These programs will run on your machine without any changes at all!

You should be sure to start off in BASIC by typing the command KEY OFF (and hitting the ENTER key). If you have a color monitor or TV set, you should next type WIDTH 40 (and the ENTER key again). You can choose the colors that you see on the screen by giving COLOR commands before you begin — or by writing them into the beginning of the program. The manual that comes with your computer will give you the details.

Remember to set the CAPS LOCK key so that you only get CAPITAL letters. To BREAK out of a program on a PC, hold down the CTRL key and press the BREAK key (it's also called SCROLL LOCK). To BREAK out on a PCjr, press the Fn key and then the letter B (for break). The PC's ENTER key has no lettering on it — just a funny arrow with a bent tail.

What if I have an Apple II Computer?

Many of these programs will run on your machine without any changes at all. In some programs, you'll have to remove one RANDOMIZE statement and change the way the program clears the screen. The "If You Have . . ." section after each Program Listing will tell you exactly what to do.

Be sure you run these programs in Applesoft BASIC — Apple's Integer BASIC just won't work. When you need to BREAK out of a program, hold down the CONTROL key and press the letter C. Your machine has a RETURN key instead of an ENTER key. If you have an Apple IIe, you should set the CAPS LOCK key so that you only get CAPITAL letters.

What if I have an Atari Computer?

Atari BASIC is very unusual. It doesn't allow string arrays like `N$(3)`. It makes you `DIMension` all your string variables before you can use them. It requires a special way of entering string commands. It lacks features such as the `TAB` function.

Most of these programs will need simple changes and additions. The "If You Have . . ." section after each Program Listing will tell you exactly what to do.

Your machine has a `BREAK` key to let you break out of a program. It has a `RETURN` key instead of an `ENTER` key. And you can change the colors that appear on your screen—check your manual for the details.

What if I have a Coleco ADAM?

Many of these programs will run on your machine without any changes at all. In some programs, you'll have to remove one `RANDOMIZE` statement and change the way the program clears the screen. Because the version of BASIC in ADAM is almost exactly like the one for the Apple, you'll usually be able to follow the directions for Apple computers in the "If You Have . . ." section after each Program Listing.

But since the ADAM does not display as many characters across the screen as Apples do, you will have to make changes to some programs—especially the ones that make pictures. The following steps will help you get the programs running on your ADAM:

1. Enter the program, making the changes needed for the Apple Computer.
2. Now look at the changes for the TRS-80 Color Computer. *Don't* change any of the lines you already changed for the Apple. *Don't* change any lines that include the words `CLS`, `LINE INPUT`, `RND` or `SOUND`. Make all other changes needed for the Color Computer.

When you need to `BREAK` out of a program, hold down the `CONTROL` key and press the letter `C`. Your machine has a `RETURN` key instead of an `ENTER` key. And you should set the `LOCK` key so that you only get CAPITAL letters.

What if I have a Commodore 64 or VIC-20?

Many of these programs will run on your machine without any changes at all. In some programs, you'll have to remove a `RANDOMIZE` statement. In some, you'll also need to change one number in `RND` statements and change the way the program clears the screen. The "If You Have . . ." section after each Program Listing will tell you exactly what to do.

Your machine has a `RETURN` key instead of an `ENTER` key. To `BREAK` out of a program, press the `RUN/STOP` key. You may want to change the col-

ors that appear on your screen. Your computer's manual will tell you exactly how to do it.

Since the VIC prints fewer characters (letters and numbers) on the screen than most other home computers, you may want to adjust the PRINT statements so that words don't "wrap" around the ends of lines. The easiest way is to divide one PRINT statement into two. If you do this, be sure the first PRINT statement keeps its original line number.

What if I have a Texas Instruments 99/4A Computer?

TI BASIC is unusual in many ways. It uses special characters instead of AND and OR. It uses the term SEG\$ instead of the more common MID\$. It's very fussy about what you can put after the word THEN.

Most of these programs will need simple changes and additions. The "If You Have . . ." section after each Program Listing will tell you exactly what to do.

Your machine has an ENTER key. To BREAK out of a program, hold down the FCTN key and press the number 4 key. You should set the ALPHA LOCK key so that you only get CAPITAL letters. You may want to change the colors that appear on your screen. Your computer's manual will tell you how to do it.

What if I have a TRS-80 Color Computer?

Many of these programs will run on your machine without any changes at all. In some, you'll have to take out one RANDOMIZE statement and change the way RND works. The "If You Have . . ." section after each Program Listing will tell you exactly what to do.

Your machine has an ENTER key and a BREAK key. You may want to change the colors that appear on your screen. Your computer's manual will tell you how to do it.

NOW, ON TO THE FUN!

Making History

Silicon is one of the most plentiful elements on earth. You find it in sand — and in computer chips. You'll find it in this program, too!
The sands of time won't wait until you type it in.

□ Sample Run

AS YOU PROBABLY KNOW, THERE IS A
VERY SOPHISTICATED TIMING CIRCUIT
ON A SILICON CHIP INSIDE YOUR
COMPUTER.

TO SEE THE VERY FIRST SILICON
TIMING CIRCUIT EVER MADE, HIT THE
ENTER KEY?

[illegible]

(continued)

[illegible]

[There's more — when you get to the end of the program, you'll see something like this:]

```

XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX
  X                                     X
    X                                 X
      X                             X
        X                         X
          X                     X
            X                 X
              X             X
                X         X
                  X     X
                    X X
                      X X
                        X:::X
                          X:::::X
                            X:::::X
                              X:::::X
                                X:::::X
                                  X:::::X
                                    X:::::X
                                      X:::::X
                                        X:::::X
                                          X:::::X
                                            X:::::X
                                              X:::::X
                                                X:::::X
                                                  X:::::X
                                                    X:::::X
                                                    X:::::X
XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX

```

□ Program Listing

```

10 REM MAKING HISTORY
100 PRINT
110 PRINT "AS YOU PROBABLY KNOW, THERE IS A"
120 PRINT "VERY SOPHISTICATED TIMING CIRCUIT"
130 PRINT "ON A SILICON CHIP INSIDE YOUR"
140 PRINT "COMPUTER."
150 PRINT
160 PRINT "TO SEE THE VERY FIRST SILICON"
170 PRINT "TIMING CIRCUIT EVER MADE, HIT THE"
180 PRINT "ENTER KEY";
190 INPUT D$
200 PRINT CHR$(12)
210 GOSUB 440
220 FOR A=1 TO 10
230 PRINT TAB(A+3);"X";
240 IF Z<=A-1 THEN GOSUB 490
250 PRINT TAB(25-A);"X"
260 NEXT A

```

(continued)

```

270 K=1
280 FOR B=10 TO 1 STEP -1
290 PRINT TAB(B+3);"X";
300 IF Z=>B-1 THEN 330
310 PRINT TAB(14);":";
320 GOTO 340
330 GOSUB 560
340 PRINT TAB(25-B);"X"
350 NEXT B
360 GOSUB 440
370 K=0
380 Z=Z+1
390 IF Z=10 THEN END
400 PRINT CHR$(12)
410 GOTO 210
420 END
430 REM --- FLATS SUBROUTINE ---
440 FOR C=1 TO 27
450 PRINT "H";
460 NEXT C
470 RETURN
480 REM --- TOP BEADS SUBROUTINE ---
490 IF K=0 THEN 530
500 FOR D=(19-K) TO 1 STEP -1
510 PRINT ":";
520 NEXT D
530 K=A*2
540 RETURN
550 REM --- BOTTOM BEADS SUBROUTINE ---
560 FOR E=2 TO K
570 PRINT ":";
580 NEXT E
590 K=(12-B)*2
600 RETURN

```

☐ If You Have . . .

APPLE II

Change: 180 PRINT "RETURN KEY";
 200 HOME
 400 HOME

ATARI

Add: 95 DIM D\$(1)
 242 IF Z>A-2 THEN 253
 245 IF A>1 THEN PRINT "X"
 246 IF A>1 THEN 260

```

252 GOTO 260
253 AR=21-(A*2)
254 IDX=AR:GOSUB 615:PRINT "X"
306 IF B=10 THEN PRINT ":X"
307 IF B=10 THEN 350
331 IF Z=B-1 THEN IDX=(2*(10-B))+1
332 IF Z=B-1 THEN 345
333 PRINT "X"
334 GOTO 350
345 GOSUB 615:PRINT "X"
465 PRINT
565 FOR E=2 TO K
615 FOR AA=1 TO IDX
625 PRINT " ";:NEXT AA:RETURN

```

Change:

```

180 PRINT "RETURN KEY";
200 PRINT CHR$(125)
230 IDX=(A+2):GOSUB 615:PRINT "X";
250 IDX=(20-A):GOSUB 615:PRINT "X"
290 IDX=(B+2):GOSUB 615:PRINT "X";
300 IF Z>=B-1 THEN 330
310 IDX=(10-B):GOSUB 615:PRINT ":";
340 IDX=(10-B)
400 PRINT CHR$(125)
560 IF K<=3 THEN 590

```

COMMODORE 64

Add:

```

215 PRINT
565 IF K=1 THEN 580

```

Change:

```

180 PRINT "RETURN KEY";
200 PRINT CHR$(147)
230 PRINT TAB(A+2);"X";
250 PRINT TAB(24-A);"X"
290 PRINT TAB(B+2);"X";
310 PRINT TAB(13);":";
340 PRINT TAB(24-B);"X"
400 PRINT CHR$(147)

```

COMMODORE VIC-20

Add:

```

215 PRINT
565 IF K=1 THEN 580

```

Change:

```

180 PRINT "RETURN KEY"
200 PRINT CHR$(147)
220 FOR A=1 TO 8
230 PRINT TAB(A+2);"X";
250 PRINT TAB(20-A);"X"
280 FOR A=8 TO 1 STEP -1
290 PRINT TAB(B+2);"X";
310 PRINT TAB(11);":";

```

(continued)

```

340 PRINT TAB(20-B);"X"
390 IF Z=8 THEN END
400 PRINT CHR$(147)
440 FOR C=1 TO 22
500 FOR D=(15-K) TO 1 STEP -1
590 K=(10-B)*2

```

TEXAS INSTRUMENTS 99/4A

Delete: 140,180

Add: 242 GOSUB 490
615 END

Change: 110 PRINT "AS YOU PROBABLY KNOW, THERE IS A VERY
SOPHISTICATED"
120 PRINT "TIMING CIRCUIT ON A"
130 PRINT "CHIP INSIDE YOUR COMPUTER"
160 PRINT "TO SEE THE VERY FIRST"
170 PRINT "SILICON TIMING CIRCUIT EVER MADE, HIT
THE ENTER KEY";
200 CALL CLEAR
240 IF Z>A-1 THEN 250
300 IF Z>=B-1 THEN 330
390 IF Z=10 THEN 615
400 CALL CLEAR

TRS-80 COLOR COMPUTER

Add: 445 FOR C=1 TO 18
565 FOR E=2 TO K

Change: 200 CLS
220 FOR A=4 TO 10
280 FOR B=9 TO 4 STEP -1
400 CLS
440 PRINT TAB(4);"H";
560 K=(11-B)*2

Sticky Computer

Once in a while it happens: Your crazy computer makes things sticky for you.
Try this program to see if you can get the bubble gum out!

☐ Sample Run

ENTER YOUR WHOLE NAME? CHARLIE FOXTROT

THIS COMPUTER IS ACTING UP AGAIN. I
THINK IT HAS SOME STICKY KEYS. HIT
THE ENTER KEY TO TEST IT?

CC
HHHHHHHHHHHHHHHHHHHHHHHHHHHHHHHHHHHHHH
AA
RRRRRRRRRRRRRRRRRRRRRRRRRRRRRRRRRRRRRR
LL
II
EE

FF
OOOOOOOOOOOOOOOOOOOOOOOOOOOOOOOOOOOOOO
XX
TT
RRRRRRRRRRRRRRRRRRRRRRRRRRRRRRRRRRRRRR
OOOOOOOOOOOOOOOOOOOOOOOOOOOOOOOOOOOOOO
TT

CC
HHHHHHHHHHHHHHHHHHHHHHHHHHHHHHHHHHHHHH
AA
RRRRRRRRRRRRRRRRRRRRRRRRRRRRRRRRRRRRRR
LL
II
EE

FF
OOOOOOOOOOOOOOOOOOOOOOOOOOOOOOOOOOOOOO
XX
TT
RRRRRRRRRRRRRRRRRRRRRRRRRRRRRRRRRRRRRR
OOOOOOOOOOOOOOOOOOOOOOOOOOOOOOOOOOOOOO
TT

Break

☐ Program Listing

```
10 REM STICKY COMPUTER
100 PRINT "ENTER YOUR WHOLE NAME";
110 INPUT WNAME$
120 LNAME=LEN(WNAME$)
130 IF LNAME=0 THEN 100
140 PRINT CHR$(12)
150 PRINT "THIS COMPUTER IS ACTING UP AGAIN. I"
160 PRINT "THINK IT HAS SOME STICKY KEYS. HIT"
170 PRINT "THE ENTER KEY TO TEST IT";
180 INPUT D$
190 FOR C=1 TO 25
200 PRINT
210 NEXT C
220 PRINT CHR$(12)
230 FOR A=1 TO LNAME
240 FOR B=1 TO 39
250 PRINT MID$(WNAME$,A,1);
260 NEXT B
270 PRINT
280 NEXT A
290 PRINT
300 GOTO 230
```

☐ If You Have . . .

APPLE II

Change: 140 HOME
170 PRINT "THE RETURN KEY TO TEST IT";
220 HOME

ATARI

Add: 95 DIM D\$(1),WNAME\$(40)

Change: 140 PRINT CHR\$(125)
170 PRINT "THE RETURN KEY TO TEST IT";
220 PRINT CHR\$(125)
240 FOR B=1 TO 37
250 PRINT WNAME\$(A,A);

COMMODORE 64

Change: 140 PRINT CHR\$(147)
170 PRINT "THE RETURN KEY TO TEST IT";
220 PRINT CHR\$(147)

COMMODORE VIC-20

Change: 100 PRINT "ENTER YOUR WHOLE NAME"
140 PRINT CHR\$(147)
170 PRINT "THE RETURN KEY TO TEST IT"
190 FOR C=1 TO 22
220 PRINT CHR\$(147)
240 FOR B=1 TO 21

TEXAS INSTRUMENTS 99/4A

Change: 140 CALL CLEAR
150 PRINT "THIS COMPUTER IS ACTING UP AGAIN.
I THINK IT HAS SOME"
160 PRINT "STICKY KEYS. HIT THE ENTER"
170 PRINT "KEY TO TEST IT";
220 CALL CLEAR
240 FOR B=1 TO 28
250 PRINT SEG\$(WNAM\$,A,1);

TRS-80 COLOR COMPUTER

Change: 140 CLS
220 CLS
240 FOR B=1 TO 31

Make Me Laugh

Did you know your computer has a sense of humor? Maybe it's not exactly as funny as your favorite comedian — but it does enjoy a good laugh now and then. Tickle your computer's funny bone — and your own!

☐ **Sample Run**

I'LL BET YOU CAN'T MAKE ME LAUGH!
TYPE IN SOMETHING YOU THINK IS FUNNY
AND HIT ENTER WHEN YOU'RE DONE.
WELL? THE TIME WE PUT A FROG IN BETTY'S SOUP

HOHOHO CHUCKLE HOHOHO GIGGLE SNICKER
TEEHEE GIGGLE HAHA HAHA HOHOHO GIGGLE
YUK-YUK HEEHAW HEEHAW YUK-YUK GIGGLE
YUK-YUK HOHOHO SNICKER SNICKER TEEHEE
TEEHEE GIGGLE HEEHEE HAHA GIGGLE SNICKER
HUCKLE TEEHEE HOHOHO HAHA CHUCKLE GIGGLE
HAHA YUK-YUK HAHA HAHA SNICKER HOHOHO
HAHA HAHA CHUCKLE HEEHEE TEEHEE HAHA
HEEHAW SNICKER HEEHAW HOHOHO TEEHEE
SNICKER HEEHAW SNICKER CHUCKLE HOHOHO
HOHOHO HEEHEE YUK-YUK GIGGLE HOHOHO
HAHA CHUCKLE HOHOHO CHUCKLE GIGGLE
EEHAW CHUCKLE GIGGLE GIGGLE HEEHAW
HAHA SNICKER HEEHEE SNICKER HEEHAW
SNICKER HAHA SNICKER HEEHAW YUK-YUK
HAHA HAHA HOHOHO HOHOHO HOHOHO HEEHEE
HEEHAW YUK-YUK HEEHEE GIGGLE CHUCKLE
CHUCKLE HEEHAW HOHOHO YUK-YUK HEEHEE
HUCKLE SNICKER TEEHEE GIGGLE YUK-YUK
HEEHEE HAHA CHUCKLE HOHOHO HAHA TEEHEE
HAHA HAHA TEEHEE SNICKER HEEHEE CHUCKLE
GIGGLE HAHA TEEHEE HAHA TEEHEE HOHOHO
HEEHEE HOHOHO CHUCKLE YUK-YUK YUK-YUK
CHUCKLE HEEHAW GIGGLE HEEHAW GIGGLE
SNICKER HOHOHO HEEHEE SNICKER HOHOHO
TEEHEE HAHA HAHA CHUCKLE YUK-YUK HEEHEE

THAT'S SO FUNNY

I ALMOST FORGOT TO LAUGH.

☐ Program Listing

```
10 REM MAKE ME LAUGH
100 PRINT "I'LL BET YOU CAN'T MAKE ME LAUGH!"
110 PRINT "TYPE IN SOMETHING YOU THINK IS FUNNY"
120 PRINT "AND HIT ENTER WHEN YOU'RE DONE."
130 PRINT "WELL";
140 INPUT JOKE$
150 PRINT
160 IF LEN(JOKE$)<1 THEN 110
170 FOR I=1 TO 500
180 FOR J=1 TO INT(RND(1)*9)+1
190 READ A$
200 NEXT J
210 RESTORE
220 PRINT A$;" ";
230 NEXT I
240 FOR B=1 TO 25
250 PRINT
260 NEXT B
270 FOR C=1 TO 9
280 READ A$
290 NEXT C
300 FOR C=1 TO 7
310 READ R$
320 T$=R$+T$
330 NEXT C
340 PRINT TAB(12);MID$(T$,1,15)
350 PRINT
360 PRINT TAB(7);MID$(T$,16,25)
370 FOR D=1 TO 11
380 PRINT
390 NEXT D
400 DATA HAHA,CHUCKLE,YUK-YUK,HEEHAW,TEEHEE
410 DATA HOHOHO,SNICKER,GIGGLE,HEEHEE
420 DATA UGH.,O LA,RGOT T,MOST FO
430 DATA UNNYI AL,T'S SO F,THA
```

☐ If You Have . . .

APPLE II

Change: 120 PRINT "AND HIT RETURN KEY WHEN YOU'RE DONE."

ATARI

```
Add: 95 JOKE$(37),A$(8),R$(29),T$(33)
      264 AA=0
      265 RESTORE
```

(continued)

```

332 AA=AA+1
333 IF AA>3 THEN 347
335 IF AA>1 THEN 345
345 PRINT T$;
346 IF AA<4 THEN 265
347 IF AA>4 THEN 365
365 PRINT T$;
366 IF AA<7 THEN 265
431 END
445 FOR A=1 TO IDX
455 PRINT " ";:NEXT A:RETURN

```

Change: 120 PRINT "AND HIT THE RETURN KEY WHEN YOU'RE DONE."
 180 FOR J=1 TO INT(RND(0)*9)+1
 300 FOR C=1 TO 7-AA
 320 T\$=R\$
 340 IDX=12:GOSUB 445
 360 IDX=7:GOSUB 445
 420 DATA UGH.,O LA,RGOT T,I ALMOST FO
 430 DATA UNNY,T'S SO F,THA

COMMODORE 64

Change: 120 PRINT "AND HIT THE RETURN KEY WHEN YOU'RE DONE."

COMMODORE VIC-20

Change: 120 PRINT "AND HIT THE RETURN KEY WHEN YOU'RE DONE."
 130 PRINT "WELL"

TEXAS INSTRUMENTS 99/4A

Change: 100 PRINT "I'LL BET YOU CAN'T MAKE ME"
 110 PRINT "LAUGH! TYPE IN SOMETHING YOU THINK
 IS FUNNY AND"
 120 PRINT "HIT THE ENTER KEY WHEN YOU'RE DONE."
 180 FOR J=1 TO INT(RND*9)+1
 320 T\$=R\$&T\$

TRS-80 COLOR COMPUTER

Change: 180 FOR J=1 TO RND(9)

Design Your Dream House!

What kind of house would you like to have for your very own? Now you can custom-build your personal favorite — right on your computer screen!

☐ Sample Run

NOW YOU CAN DESIGN
YOUR VERY OWN DREAM HOUSE!

HOW MANY STORIES WILL YOUR HOUSE BE:

- 1--ONE STORY TALL
- 2--TWO STORIES TALL

YOUR CHOICE, PLEASE? 1

WHAT KIND OF ROOF WILL YOUR HOUSE HAVE:

- 1--FLAT ROOF
- 2--SLOPING ROOF

YOUR CHOICE, PLEASE? 1

WHICH SIDE WILL THE FRONT DOOR BE ON:

- 1--THE LEFT SIDE
- 2--THE RIGHT SIDE

YOUR CHOICE, PLEASE? 1

```
=====
=====
*                                     *
*                                     *
*      **      **      **      **      *
*      **      **      **      **      *
*                **                *
*                **                *
*****
```

WANT TO TRY ANOTHER (YES OR NO)? YES

(continued)

HOW MANY STORIES WILL YOUR HOUSE BE:

- 1--ONE STORY TALL
- 2--TWO STORIES TALL

YOUR CHOICE, PLEASE? 2

WHAT KIND OF ROOF WILL YOUR HOUSE HAVE:

- 1--FLAT ROOF
- 2--SLOPING ROOF

YOUR CHOICE, PLEASE? 2

WHICH SIDE WILL THE FRONT DOOR BE ON:

- 1--THE LEFT SIDE
- 2--THE RIGHT SIDE

YOUR CHOICE, PLEASE? 2

```
*****
*****
*****
*****
*****
*****
*                                     *
*                                     *
*      **      **      **      **      *
*      **      **      **      **      *
*                                     *
*                                     *
*****
*                                     *
*                                     *
*      **      **      **      **      *
*      **      **      **      **      *
*                                     *
*                                     *
*****
```

WANT TO TRY ANOTHER (YES OR NO)? NO

☐ Program Listing

```
10 REM DESIGN YOUR DREAM HOUSE
100 PRINT
110 PRINT "NOW YOU CAN DESIGN"
120 PRINT "YOUR VERY OWN DREAM HOUSE!"
130 PRINT
140 PRINT "HOW MANY STORIES WILL YOUR HOUSE BE:"
150 PRINT
160 PRINT TAB(5);"1--ONE STORY TALL"
```

```

170 PRINT TAB(5);"2--TWO STORIES TALL"
180 PRINT
190 PRINT "YOUR CHOICE, PLEASE";
200 INPUT S
210 IF S<1 OR S>2 THEN 190
220 PRINT
230 PRINT "WHAT KIND OF ROOF WILL YOUR HOUSE HAVE:"
240 PRINT
250 PRINT TAB(5);"1--FLAT ROOF"
260 PRINT TAB(5);"2--SLOPING ROOF"
270 PRINT
280 PRINT "YOUR CHOICE, PLEASE";
290 INPUT R
300 IF R<1 OR R>2 THEN 280
310 PRINT
320 PRINT "WHICH SIDE WILL THE FRONT DOOR BE ON:"
330 PRINT
340 PRINT TAB(5);"1--THE LEFT SIDE"
350 PRINT TAB(5);"2--THE RIGHT SIDE"
360 PRINT
370 PRINT "YOUR CHOICE, PLEASE";
380 INPUT D
390 IF D<1 OR D>2 THEN 370
400 PRINT
410 IF R=2 THEN 500
420 FOR A=1 TO 2
430 PRINT TAB(4);
440 FOR B=1 TO 31
450 PRINT "=";
460 NEXT B
470 PRINT "="
480 NEXT A
490 GOTO 570
500 FOR A=1 TO 6
510 PRINT TAB(8-A);
520 FOR B=1 TO 23+A*2
530 PRINT "*";
540 NEXT B
550 PRINT "*"
560 NEXT A
570 IF S=1 THEN 700
580 FOR C=1 TO 6
590 IF C>2 AND C<5 THEN 620
600 PRINT TAB(4);"*";TAB(35);"*"
610 GOTO 640
620 PRINT TAB(4);"*";TAB(10);"*";TAB(16);"*";
630 PRINT TAB(22);"*";TAB(28);"*";TAB(35);"*"
640 NEXT C
650 PRINT TAB(4);
660 FOR Z=1 TO 31
670 PRINT "*";
680 NEXT Z
690 PRINT "*"
700 FOR Z=1 TO 6
710 IF Z>2 AND Z<5 THEN 750

```

(continued)

```

720 IF Z>4 THEN 780
730 PRINT TAB(4);" ";TAB(35);" "
740 GOTO 820
750 PRINT TAB(4);" ";TAB(10);" ";TAB(16);" ";
760 PRINT TAB(22);" ";TAB(28);" ";TAB(35);" "
770 GOTO 820
780 IF D=2 THEN 810
790 PRINT TAB(4);" ";TAB(16);" ";TAB(35);" "
800 GOTO 820
810 PRINT TAB(4);" ";TAB(22);" ";TAB(35);" "
820 NEXT Z
830 PRINT TAB(4);
840 FOR Z=1 TO 31
850 PRINT " ";
860 NEXT Z
870 PRINT " "
880 PRINT
890 PRINT "WANT TO TRY ANOTHER (YES OR NO)";
900 INPUT Y$
910 IF MID$(Y$,1,1)<>"N" THEN 130

```

☐ If You Have . . .

APPLE II No Changes Required

ATARI

Add: 95 DIM Y\$(3)

```

625 IDX=4:GOSUB 925:PRINT "***";
635 IDX=5:GOSUB 925:PRINT " ";
755 IDX=4:GOSUB 925:PRINT "***";
765 IDX=5:GOSUB 925:PRINT " ";
795 IDX=17:GOSUB 925:PRINT " "
815 IDX=11:GOSUB 925:PRINT " "
911 END
925 FOR AA=1 TO IDX
935 PRINT " ";:NEXT AA:RETURN

```

Change:

```

160 IDX=5:GOSUB 925:PRINT "1--ONE STORY TALL"
170 GOSUB 925:PRINT "2--TWO STORIES TALL"
250 GOSUB 925:PRINT "1--FLAT ROOF"
260 GOSUB 925:PRINT "2--SLOPING ROOF"
340 GOSUB 925:PRINT "1--THE LEFT SIDE"
350 GOSUB 925:PRINT "2--THE RIGHT SIDE"
430 IDX=4:GOSUB 925
510 IDX=(7-A):GOSUB 925
600 IDX=3:GOSUB 925:PRINT "***";:
      IDX=30:GOSUB 925:PRINT " "
620 IDX=3:GOSUB 925:PRINT "***";:
      IDX=5:GOSUB 925:PRINT "***";
630 GOSUB 925:PRINT "***";:GOSUB 925:PRINT "***";

```

```

650 IDX=3:GOSUB 925
730 IDX=3:GOSUB 925:PRINT "*";:
  IDX=30:GOSUB 925:PRINT "*"
750 IDX=3:GOSUB 925:PRINT "*";:
  IDX=5:GOSUB 925:PRINT "***";
760 GOSUB 925:PRINT "***";:GOSUB 925:PRINT "***";
790 IDX=3:GOSUB 925:PRINT "*";:
  IDX=11:GOSUB 925:PRINT "***";
810 IDX=3:GOSUB 925:PRINT "*";:
  IDX=17:GOSUB 925:PRINT "***";
830 IDX=3:GOSUB 925
910 IF Y$(1,1)<>"N" THEN 130

```

COMMODORE 64 No Changes Required

COMMODORE VIC-20

Change:

```

160 PRINT TAB(3); "1--ONE STORY TALL"
170 PRINT TAB(3); "2--TWO STORIES TALL"
190 PRINT "YOUR CHOICE, PLEASE"
280 PRINT "YOUR CHOICE, PLEASE"
370 PRINT "YOUR CHOICE, PLEASE"
430 PRINT TAB(2);
440 FOR B=1 TO 19
500 FOR A=1 TO 3
510 PRINT TAB(6-A);
520 FOR B=1 TO 11+A*2
580 FOR C=1 TO 3
590 IF C>1 AND C<3 THEN 620
600 PRINT TAB(2);"*";TAB(21);"*"
620 PRINT TAB(2);"*";TAB(5);"***";TAB(9);"***";
630 PRINT TAB(13);"***";TAB(17);"***";TAB(21);"*"
650 PRINT TAB(2);
660 FOR Z=1 TO 19
700 FOR Z=1 TO 3
710 IF Z>1 AND Z<3 THEN 750
720 IF Z>2 THEN 780
730 PRINT TAB(2);"*";TAB(21);"*"
750 PRINT TAB(2);"*";TAB(5);"***";TAB(9);"***";
760 PRINT TAB(13);"***";TAB(17);"***";TAB(21);"*";
790 PRINT TAB(2);"*";TAB(9);"***";TAB(21);"*";
810 PRINT TAB(2);"*";TAB(13);"***";TAB(21);"*";
830 PRINT TAB(2);
840 FOR Z=1 TO 19
890 PRINT "TRY ANOTHER (Y OR N)"

```

TEXAS INSTRUMENTS 99/4A

Change:

```

140 PRINT "HOW MANY STORIES WILL YOUR HOUSE BE:"
210 IF (S<1)+(S>2) THEN 190
300 IF (R<1)+(R>2) THEN 280

```

(continued)

```

320 PRINT "WHICH SIDE WILL THE FRONT DOOR BE ON:"
390 IF (D<1)+(D>2) THEN 370
440 FOR B=1 TO 21
590 IF (C>2)*(C<5) THEN 620
600 PRINT TAB(4);"*";TAB(25);"*"
630 PRINT TAB(22);"***";TAB(25);"*"
660 FOR Z=1 TO 21
710 IF (Z=2)*(Z<5) THEN 750
730 PRINT TAB(4);"*";TAB(25);"*"
760 PRINT TAB(22);"***";TAB(25);"*"
790 PRINT TAB(4);"*";TAB(16);"***";TAB(25);"*"
810 PRINT TAB(4);"*";TAB(22);"***";TAB(25);"*"
840 FOR Z=1 TO 21
910 IF SEG$(Y$,1,1)<>"N" THEN 130

```

TRS-80 COLOR COMPUTER

Change:

```

510 PRINT TAB(9-A);
520 FOR B=1 TO 13+A*2
580 FOR C=2 TO 5
600 PRINT TAB(4);"*";TAB(27);"*"
620 PRINT TAB(4);"*";TAB(8);"***";TAB(13);"***";
630 PRINT TAB(18);"***";TAB(23);"***";TAB(27);"*"
660 FOR Z=1 TO 23
700 FOR Z=2 TO 6
730 PRINT TAB(4);"*";TAB(27);"*"
750 PRINT TAB(4);"*";TAB(8);"***";TAB(13);"***";
760 PRINT TAB(18);"***";TAB(23);"***";TAB(27);"*"
790 PRINT TAB(4);"*";TAB(13);"***";TAB(27);"*"
810 PRINT TAB(4);"*";TAB(18);"***";TAB(27);"*"
840 FOR Z=1 TO 23

```

Grouchy Computer

Sometimes computers can get awfully cranky. But you'll have to run this program to see just how much of a grouch your own computer can be!

☐ Sample Run

WHAT'S YOUR NAME? CRABBY APPLETON

CRABBY APPLETON? WHAT A DUMB NAME.
I'VE HEARD SOME STUPID NAMES BEFORE
BUT CRABBY APPLETON SURE TAKES THE CAKE.

HOW OLD ARE YOU? 10

10? ARE YOU OUT OF DIAPERS YET?
GO FIND SOMEONE OLDER TO USE ME, OKAY?

WHAT SCHOOL DO YOU GO TO? ARNSBARK SCHOOL

I SHOULD HAVE GUESSED...
ARNSBARK SCHOOL IS WORSE THAN MY SCHOOL.

HEY, CRABBY APPLETON, HOW TALL ARE YOU? 4 FEET 5 INCHES

OH, A MIDGET, HUH? WHY DO THEY GIVE ME
ALL THE MIDGETS?

HOW MANY POUNDS DO YOU WEIGH, CRABBY APPLETON? 118

WOW, YOU SURE HAVE A PROBLEM, CRABBY APPLETON.
I'M SURE GLAD I DON'T WEIGH 118 POUNDS!

AND BY THE WAY, THAT SURE IS AN UGLY
THING YOU'RE WEARING.

THINK YOU KNOW HOW TO USE A COMPUTER? YES

ARE YOU KIDDING, CRABBY APPLETON? HAH!
YOU COULDN'T EVEN USE AN ADDING MACHINE!

WHAT COMPUTER ARE YOU USING? TIGERPAW 64

I SHOULD'VE GUESSED. OF ALL THE KINDS
OF COMPUTERS, TIGERPAW 64 IS THE WORST!

WHAT JOB DO YOU WANT WHEN YOU GROW UP?
WELL? PROGRAMMER

(continued)

YOU'D MAKE A LOUSY PROGRAMMER, AND
AT 10 YEARS OLD, YOU HAVE A LONG WAIT!

DO YOU HAVE A PET (YES OR NO)? YES

WHAT KIND OF PET? (I BET IT'S A WORM!)
WELL? PARROT

THAT'S DISGUSTING! A PARROT? UGH!

WHAT'S ITS NAME? HYMIE

THAT HAS TO BE THE DUMBEST NAME EVER!
IMAGINE! A PARROT NAMED HYMIE!!

HERE, HYMIE, COME OVER HERE NOW!
AND BITE CRABBY APPLETON ON THE LEG!

WANT TO HEAR ALL ABOUT YOURSELF? YES

ALL ABOUT CRABBY APPLETON

AGE: 10

HEIGHT: 4 FEET 5 INCHES

WEIGHT: 118

ATTENDS: ARNSBARK SCHOOL

CRABBY APPLETON'S COMPUTER: TIGERPAW 64

HAS PARROT NAMED HYMIE

FUTURE OCCUPATION: PROGRAMMER

□ Program Listing

```
10 REM GROUCHY COMPUTER
100 PRINT CHR$(12)
110 SP=4
120 PRINT "WHAT'S YOUR NAME";
130 INPUT NAM$
140 IF NAM$="" THEN 120
150 GOSUB 1280
160 PRINT NAM$;"? WHAT A DUMB NAME."
170 PRINT "I'VE HEARD SOME STUPID NAMES BEFORE"
180 PRINT "BUT ";NAM$;" SURE TAKES THE CAKE."
190 GOSUB 1280
200 PRINT "HOW OLD ARE YOU";
210 INPUT OLD$
220 IF OLD$="" THEN 200
230 GOSUB 1280
240 PRINT OLD$;"? ARE YOU OUT OF DIAPERS YET?"
250 PRINT "GO FIND SOMEONE OLDER TO USE ME, OKAY?"
260 GOSUB 1280
270 PRINT "WHAT SCHOOL DO YOU GO TO";
280 INPUT SCHOOL$
290 IF SCHOOL$="" THEN 270
300 GOSUB 1280
310 PRINT "I SHOULD HAVE GUESSED..."
320 PRINT SCHOOL$;" IS WORSE THAN MY SCHOOL."
330 GOSUB 1280
340 PRINT "HEY, ";NAM$;", HOW TALL ARE YOU";
350 PRINT " ";
360 INPUT TALL$
370 IF TALL$="" THEN 340
380 GOSUB 1280
390 PRINT "OH, A MIDGET, HUH? WHY DO THEY GIVE ME"
400 PRINT "ALL THE MIDGETS?"
410 GOSUB 1280
420 PRINT "HOW MANY POUNDS DO YOU WEIGH, ";NAM$;
430 INPUT WEIGHT$
440 IF WEIGHT$="" THEN 420
450 GOSUB 1280
460 PRINT "WOW, YOU SURE HAVE A PROBLEM, ";NAM$;". "
470 PRINT "I'M SURE GLAD I DON'T WEIGH ";WEIGHT$; " POUNDS!"
480 GOSUB 1280
490 PRINT "AND BY THE WAY, THAT SURE IS AN UGLY"
500 PRINT "THING YOU'RE WEARING."
510 GOSUB 1280
520 PRINT "THINK YOU KNOW HOW TO USE A COMPUTER";
530 INPUT THINK$
540 IF THINK$="" THEN 520
550 GOSUB 1280
560 PRINT "ARE YOU KIDDING, ";NAM$;"? HAH!"
570 PRINT "YOU COULDN'T EVEN USE AN ADDING MACHINE!"
580 GOSUB 1280
590 PRINT "WHAT COMPUTER ARE YOU USING";
```

(continued)

```

600 INPUT CP$
610 IF CP$="" THEN 590
620 GOSUB 1280
630 PRINT "I SHOULD'VE GUESSED. OF ALL THE KINDS"
640 PRINT "OF COMPUTERS, ";CP$;" IS THE WORST!"
650 GOSUB 1280
660 PRINT "WHAT JOB DO YOU WANT WHEN YOU GROW UP?"
670 PRINT "WELL";
680 INPUT GROW$
690 IF GROW$="" THEN 660
700 GOSUB 1280
710 PRINT "YOU'D MAKE A LOUSY ";GROW$;" , AND"
720 PRINT "AT ";OLD$;" YEARS OLD, YOU HAVE A LONG WAIT!"
730 GOSUB 1280
740 PRINT "DO YOU HAVE A PET (YES OR NO)";
750 INPUT PET$
760 IF MID$(PET$,1,1)="N" THEN 970
770 IF MID$(PET$,1,1)="Y" THEN 790
780 GOTO 740
790 GOSUB 1280
800 PRINT "WHAT KIND OF PET? (I BET IT'S A WORM!)"
810 PRINT "WELL";
820 INPUT PK$
830 IF PK$="" THEN 800
840 GOSUB 1280
850 PRINT "THAT'S DISGUSTING! A ";PK$;"? UGH!"
860 GOSUB 1280
870 PRINT "WHAT'S ITS NAME";
880 INPUT PN$
890 IF PN$="" THEN 870
900 GOSUB 1280
910 PRINT "THAT HAS TO BE THE DUMBEST NAME EVER!"
920 PRINT "IMAGINE! A ";PK$;" NAMED ";PN$;"!!"
930 GOSUB 1280
940 PRINT "HERE, ";PN$;" , COME OVER HERE NOW!"
950 PRINT "AND BITE ";NAM$;" ON THE LEG!"
960 GOTO 980
970 PRINT "WELL, I'M GLAD I'M NOT YOUR PET.  WHEW!"
980 GOSUB 1280
990 PRINT "WANT TO HEAR ALL ABOUT YOURSELF";
1000 INPUT SELF$
1010 IF MID$(SELF$,1,1)="Y" THEN 1050
1020 GOSUB 1280
1030 PRINT "WELL, NEITHER WOULD ANYONE ELSE!"
1040 END
1050 SP=2
1060 A$="ALL ABOUT "+NAM$
1070 TBN=20-(LEN(A$)/2)
1080 PRINT TAB(TBN);A$
1090 GOSUB 1280
1100 PRINT "AGE: ";OLD$
1110 GOSUB 1280
1120 PRINT "HEIGHT: ";TALL$
1130 GOSUB 1280
1140 PRINT "WEIGHT: ";WEIGHT$

```

```

1150 GOSUB 1280
1160 PRINT "ATTENDS: ";SCHOOL$
1170 GOSUB 1280
1180 PRINT NAM$;"'S COMPUTER: ";CP$
1190 GOSUB 1280
1200 IF MID$(PET$,1,1)="Y" THEN 1230
1210 PRINT NAM$;" DOES NOT HAVE ANY PETS"
1220 GOTO 1240
1230 PRINT "HAS ";PK$;" NAMED ";PN$
1240 GOSUB 1280
1250 PRINT "FUTURE OCCUPATION: ";GROW$
1260 END
1270 REM --- SPACE SUBROUTINE ---
1280 FOR S=1 TO SP
1290 PRINT
1300 NEXT S
1310 RETURN

```

☐ If You Have . . .

APPLE II

```

Change: 100 HOME
        680 INPUT GW$
        690 IF GW$="" THEN 660
        710 PRINT "YOU'D MAKE A LOUSY ";GW$;" , AND"
        1250 PRINT "FUTURE OCCUPATION: ";GW$

```

ATARI

```

Add: 91 DIM TALL$(11),WEIGHT$(3),THINK$(11),CP$(17),
      PET$(3)
      92 DIM NAM$(40),OLD$(2),SCHOOL$(40),PK$(11)
      93 DIM GROW$(22),PN$(11),SELF$(3),A$(11)
      1075 IDX=TBN
      1325 FOR AA=1 TO IDX
      1335 PRINT " ";:NEXT AA:RETURN

Change: 100 PRINT CHR$(125)
        760 IF PET$(1,1)="N" THEN 970
        770 IF PET$(1,1)="Y" THEN 790
        1010 IF SELF$(1,1)="Y" THEN 1050
        1060 A$="ALL ABOUT "
        1070 TBN=20-((LEN(A$)+9/2))
        1080 GOSUB 1325:PRINT A$;NAM$
        1200 IF PET$(1,1)="Y" THEN 1230

```

COMMODORE 64

```

Change: 100 PRINT CHR$(147)

```

COMMODORE VIC-20

Delete: 1090,1110,1130,1150

Change: 100 PRINT CHR\$(147)
120 PRINT "WHAT'S YOUR NAME"
200 PRINT "HOW OLD ARE YOU"
270 PRINT "WHAT SCHOOL DO YOU GO TO"
350 PRINT ""
420 PRINT "HOW MANY POUNDS DO YOU WEIGH, ";NAM\$
520 PRINT "THINK YOU KNOW HOW TO USE A COMPUTER"
590 PRINT "WHAT COMPUTER ARE YOU USING"
670 PRINT "WELL"
740 PRINT "DO YOU HAVE A PET (YES OR NO)"
810 PRINT "WELL"
870 PRINT "WHAT'S ITS NAME"
990 PRINT "WANT TO HEAR ALL ABOUT YOURSELF"
1070 TBN=10-(LEN(A\$)/2)

TEXAS INSTRUMENTS 99/4A

Change: 100 CALL CLEAR
240 PRINT OLD\$;"? ARE YOU OUT OF DIAPERS YET?"
390 PRINT "OH, A MIDGET, HUH? WHY DO THEY GIVE ME"
660 PRINT "WHAT JOB DO YOU WANT WHEN YOU GROW UP?"
740 PRINT "DO YOU HAVE A PET (YES OR NO)";
760 IF SEG\$(PET\$,1,1)="N" THEN 970
770 IF SEG\$(PET\$,1,1)="Y" THEN 790
800 PRINT "WHAT KIND OF PET? (I BET IT'S A WORM)"
910 PRINT "THAT HAS TO BE THE DUMBEST NAME EVER!"
990 PRINT "WANT TO HEAR ALL ABOUT YOURSELF"
1010 IF SEG\$(SELF\$,1,1)="Y" THEN 1050
1030 PRINT "WELL, NEITHER WOULD ANYONE ELSE!"
1060 A\$="ALL ABOUT "&NAM\$
1200 IF SEG\$(PET\$,1,1)="Y" THEN 1230

TRS-80 COLOR COMPUTER

Delete: 480,1090

Change: 100 CLS
490 PRINT "AND BY THE WAY, THAT SURE IS AN UGLY";" ";

Optical Illusion

You may have seen something like this in books, but it's even better on your computer screen. Don't stare too long, or you'll be even dizzier than you are already!

Sample Run

I'LL BET YOU DIDN'T KNOW THAT YOUR COMPUTER IS CAPABLE OF PRINTING A FAMOUS OPTICAL ILLUSION. IT WILL BE MADE ENTIRELY OUT OF STRAIGHT ROWS OF CHARACTERS, BUT WILL LOOK VERY CROOKED.

HIT THE ENTER KEY TO SEE?

A large grid of alternating left-pointing and right-pointing chevrons (less-than and greater-than signs) arranged in a repeating pattern. The chevrons are black on a white background, forming a dense, textured field. The pattern consists of horizontal rows where each row is a mirror image of the one above it, with the chevrons pointing in opposite directions. This creates a visual effect of a continuous, undulating surface.

Break

☐ Program Listing

```
10 REM OPTICAL ILLUSION
100 PRINT "I'LL BET YOU DIDN'T KNOW THAT YOUR"
110 PRINT "COMPUTER IS CAPABLE OF PRINTING A"
120 PRINT "FAMOUS OPTICAL ILLUSION. IT WILL"
130 PRINT "BE MADE ENTIRELY OUT OF STRAIGHT"
140 PRINT "ROWS OF CHARACTERS, BUT WILL LOOK"
150 PRINT "VERY CROOKED."
160 PRINT
170 PRINT "HIT THE ENTER KEY TO SEE";
180 INPUT D$
190 PRINT CHR$(12)
200 A$=">"
210 IF A$="<" THEN 240
220 A$="<"
230 GOTO 250
240 A$=">"
250 FOR B=1 TO 38
260 PRINT A$;
270 NEXT B
280 PRINT A$
290 GOTO 210
```

☐ If You Have . . .

APPLE II

Change: 170 PRINT "HIT THE RETURN KEY TO SEE";
190 HOME

ATARI

Add: 95 DIM D\$(1),A\$(1)

Change: 170 PRINT "HIT THE RETURN KEY TO SEE";
190 PRINT CHR\$(125)
250 FOR B=1 TO 36

COMMODORE 64

Change: 170 PRINT "HIT THE RETURN KEY TO SEE";
190 PRINT CHR\$(147)

COMMODORE VIC-20

Change: 170 PRINT "HIT THE RETURN KEY TO SEE"
190 PRINT CHR\$(147)
250 FOR B=1 TO 20

TEXAS INSTRUMENTS 99/4A

Change: 100 PRINT "I'LL BET YOU DIDN'T KNOW THAT YOUR
 COMPUTER IS"
110 PRINT "CAPABLE OF PRINTING A"
120 PRINT "FAMOUS OPTICAL ILLUSION. IT WILL BE
 MADE ENTIRELY"
130 PRINT "OUT OF STRAIGHT ROWS OF"
140 PRINT "CHARACTERS, BUT WILL LOOK"
190 CALL CLEAR
250 FOR B=1 TO 26

TRS-80 COLOR COMPUTER

Change: 190 CLS
250 FOR B=1 TO 30

Yankee Doodle Computer

Here's the perfect way to celebrate any patriotic occasion.
It's a lot safer than fireworks!

☐ Sample Run

WE COMPUTERS ARE VERY PATRIOTIC. IF
YOU WANT TO SEE THE STARS AND STRIPES
JUST HIT THE ENTER KEY?

```
* * * * *
* * * * *
* * * * *
* * * * *
* * * * *
```

```
=====
=====
=====
=====
```

```
* * * * *
* * * * *
* * * * *
* * * * *
* * * * *
* * * * *
```

```
=====
=====
=====
=====
=====
=====
=====
=====
```

```
* * * * *
* * * * *
* * * * *
* * * * *
* * * * *
```

=====
=====
=====

* * * * *
* * * * *
* * * * *
* * * * *

Break

☐ Program Listing

```
10 REM YANKEE DOODLE COMPUTER
100 PRINT "WE COMPUTERS ARE VERY PATRIOTIC. IF"
110 PRINT "YOU WANT TO SEE THE STARS AND STRIPES"
120 PRINT "JUST HIT THE ENTER KEY";
130 INPUT Y$
140 PRINT CHR$(12)
150 K=0
160 GOSUB 430
170 FOR H=1 TO B/2
180 PRINT TAB(A);
190 FOR T=1 TO B+1 STEP 2
200 PRINT "* ";
210 NEXT T
220 NEXT H
230 FOR S=1 TO 5
240 PRINT
250 NEXT S
260 GOSUB 430
270 FOR J=1 TO B/2
280 PRINT TAB(A);
290 FOR U=1 TO B+1
300 IF K=1 THEN 330
310 PRINT "=";
320 GOTO 340
330 PRINT " ";
340 NEXT U
350 K=K+1
360 IF K=2 THEN K=0
370 NEXT J
380 FOR R=1 TO 5
390 PRINT
400 NEXT R
410 GOTO 160
420 REM --- RANDOM SUBROUTINE ---
430 A=INT(RND(1)*30)+1
440 B=INT(RND(1)*(30-A))+8
450 RETURN
```

☐ If You Have . . .

APPLE II

Add: 215 PRINT
365 PRINT

Change: 120 PRINT "JUST HIT THE RETURN KEY";
140 HOME

ATARI

Add: 95 DIM Y\$(1)
175 PRINT
275 PRINT
465 FOR AA=1 TO IDX
475 PRINT " ";:NEXT AA:RETURN

Change: 120 PRINT "JUST HIT THE RETURN KEY";
140 PRINT CHR\$(125)
180 IDX=A:GOSUB 465
280 IDX=A:GOSUB 465
430 A=INT(RND(0)*30)+1
440 B=INT(RND(0)*(30-A))+8

COMMODORE 64

Add: 215 PRINT
345 PRINT

Change: 120 PRINT "JUST HIT THE RETURN KEY";
140 PRINT CHR\$(147)

COMMODORE VIC-20

Add: 215 PRINT
345 PRINT

Change: 120 PRINT "JUST HIT THE RETURN KEY"
140 PRINT CHR\$(147)
430 A=INT(RND(0)*10)+1
440 B=INT(RND(0)*(10-A))+8

TEXAS INSTRUMENTS 99/4A

Add: 362 K=0

Change: 100 PRINT "WE COMPUTERS ARE VERY PATRI-OTIC. IF
YOU WANT TO SEE"
110 PRINT "THE STARS AND STRIPES"
140 CALL CLEAR
360 IF K<>2 THEN 370
430 A=INT(RND*19)+1
440 B=INT(RND*(19-A))+8

TRS-80 COLOR COMPUTER

Change: 215 PRINT
365 PRINT

Add: 140 CLS
430 A=RND(23)
440 B=RND(23-A)+7

Nutty Numbers

Know anybody with a femtobrain? Do you have two gigafriends? This program will tell you — it's a terabarrel of fun! (By the way — except for globlicks, all the terms in this program are for real!)

☐ Sample Run

IN OUTER SPACE THERE ARE ENORMOUS THINGS--AND TEENY, TINY ONES! LET ME SHOW YOU SOME YOU MIGHT NOT KNOW.

FIRST, WE NEED ONE OF SOMETHING-- A STANDARD UNIT. JUST FOR FUN, LET'S CALL IT A "GLOBLICK"!

A GLOBLICK IS NOT VERY BIG OR SMALL. (IN FACT, IT'S JUST THE RIGHT SIZE)

BUT PUT 10 GLOBLICKS TOGETHER IN ONE PLACE AND YOU HAVE A DEKAGLOBLICK. 100 OF THEM MAKE A HECTOGLICK.

YOU KNOW HOW IT IS WITH GLOBLICKS. YOU CAN'T GET MORE THAN 100 OF THEM TOGETHER, NO MATTER HOW HARD YOU TRY!

WE NEED A DIFFERENT STANDARD UNIT. TELL ME ONE OF YOUR OWN? RATNOSE

EXCELLENT! YOU CAN DEFINITELY GET MORE THAN 100 RATNOSES TOGETHER.

=====

RATNOSES COME IN DIFFERENT AMOUNTS:

- 1 DECIRATNOSE
- 2 CENTIRATNOSE
- 3 MILLIRATNOSE
- 4 MICRORATNOSE
- 5 NANORATNOSE
- 6 PICORATNOSE
- 7 FEMTORATNOSE
- 8 ATTORATNOSE
- 9 DEKARATNOSE
- 10 HECTORATNOSE

- 11 KILORATNOSE
- 12 MEGARATNOSE
- 13 GIGARATNOSE
- 14 TERARATNOSE

CHOOSE ONE (1-14)? 13

=====

ONE GIGARATNOSE IS EQUAL TO
1000000000 RATNOSES

=====

WANT TO TRY ANOTHER? YES

=====

RATNOSES COME IN DIFFERENT AMOUNTS:

- 1 DECIRATNOSE
- 2 CENTIRATNOSE
- 3 MILLIRATNOSE
- 4 MICRORATNOSE
- 5 NANORATNOSE
- 6 PICORATNOSE
- 7 FEMTORATNOSE
- 8 ATTORATNOSE
- 9 DEKARATNOSE
- 10 HECTORATNOSE
- 11 KILORATNOSE
- 12 MEGARATNOSE
- 13 GIGARATNOSE
- 14 TERARATNOSE

CHOOSE ONE (1-14)? 6

=====

ONE PICORATNOSE IS EQUAL TO
1/1000000000000TH OF ONE RATNOSE

=====

WANT TO TRY ANOTHER? NO

□ Program Listing

```
10 REM NUTTY NUMBERS
100 DIM Z(14),S(14)
110 G$="GLOBLICK"
120 PRINT "IN OUTER SPACE THERE ARE ENORMOUS"
130 PRINT "THINGS--AND TEENY, TINY ONES! LET ME"
140 PRINT "SHOW YOU SOME YOU MIGHT NOT KNOW."
150 PRINT
160 PRINT "FIRST, WE NEED ONE OF SOMETHING--"
170 PRINT "A STANDARD UNIT. JUST FOR FUN, LET'S"
180 PRINT "CALL IT A ";CHR$(34);G$;CHR$(34);"!"
190 PRINT
200 PRINT "A ";G$;" IS NOT VERY BIG OR SMALL."
210 PRINT "(IN FACT, IT'S JUST THE RIGHT SIZE)"
220 PRINT
230 PRINT "BUT PUT 10 ";G$;"S TOGETHER IN ONE"
240 PRINT "PLACE AND YOU HAVE A DEKA";G$;". "
250 PRINT "100 OF THEM MAKE A HECTO";G$;". "
260 PRINT
270 PRINT "YOU KNOW HOW IT IS WITH ";G$;"S."
280 PRINT "YOU CAN'T GET MORE THAN 100 OF THEM"
290 PRINT "TOGETHER, NO MATTER HOW HARD YOU TRY!"
300 PRINT
310 PRINT "WE NEED A DIFFERENT STANDARD UNIT."
320 PRINT "TELL ME ONE OF YOUR OWN";
330 INPUT U$
340 IF U$="" THEN 310
350 IF LEN(U$)>2 AND LEN(U$)<22 THEN 400
360 GOSUB 990
370 PRINT "WHAT A ROTTEN NAME! YOU CAN DO BETTER!"
380 GOSUB 990
390 GOTO 310
400 IF MID$(U$,LEN(U$),1)<>"S" THEN 480
410 GOSUB 990
420 PRINT "I DON'T LIKE UNITS THAT END IN ";CHR$(34);
    "S";CHR$(34)
430 PRINT "ESPECIALLY IF THEY ARE ALREADY PLURAL."
440 PRINT
450 PRINT "TRY AGAIN, AND MAKE SURE IT'S SINGULAR."
460 GOSUB 990
470 GOTO 300
480 PRINT
490 PRINT "EXCELLENT! YOU CAN DEFINITELY"
500 PRINT "GET MORE THAN 100 ";U$;"S TOGETHER."
510 IF LEN(U$)<11 THEN 540
520 GOSUB 1000
530 GOTO 550
540 GOSUB 990
550 PRINT U$;"S COME IN DIFFERENT AMOUNTS:"
560 PRINT
570 FOR A=1 TO 14
580 READ Z(A),S(A)
```

```

590 NEXT A
600 FOR B=1 TO 14
610 READ P$
620 PRINT TAB(9);B;TAB(14);P$;U$
630 NEXT B
640 RESTORE
650 PRINT
660 PRINT "CHOOSE ONE (1-14)";
670 INPUT N
680 IF N<1 OR N>14 THEN 660
690 PRINT
700 FOR C=1 TO 14
710 READ Z(C),S(C)
720 NEXT C
730 FOR D=1 TO N
740 READ P$
750 NEXT D
760 RESTORE
770 GOSUB 990
780 PRINT "ONE ";P$;U$;" IS EQUAL TO"
790 GOSUB 900
800 IF S(N)>0 THEN 830
810 PRINT "TH OF ONE ";U$
820 GOTO 840
830 PRINT CHR$(32);U$;"S"
840 GOSUB 990
850 PRINT "WANT TO TRY ANOTHER";
860 INPUT Y$
870 IF MID$(Y$,1,1)<>"N" THEN 540
880 END
890 REM --- ZERO ADDING SUBROUTINE ---
900 IF S(N)=0 THEN 930
910 PRINT "1";
920 GOTO 940
930 PRINT "1/1";
940 FOR E=1 TO Z(N)
950 PRINT "0";
960 NEXT E
970 RETURN
980 REM --- LINE SUBROUTINE ---
990 PRINT
1000 FOR L=1 TO 39
1010 PRINT "=";
1020 NEXT L
1030 PRINT
1040 PRINT
1050 RETURN
1060 DATA 1,0,2,0,3,0,6,0,9,0,12,0,15,0,18,0
1070 DATA 1,1,2,1,3,1,6,1,9,1,12,1
1080 DATA DECI,CENTI,MILLI,MICRO,NANO,PICO,FEMTO
1090 DATA ATTO,DEKA,HECTO,KILO,MEGA,GIGA,TERA

```

☐ If You Have . . .

APPLE II No Changes Required

ATARI

```
Add: 95 DIM G$(8),U$(29),P$(7),Y$(3)
      585 Z(A)=ZA:S(A)=SA
      624 IDX=5:IF B>=10 THEN IDX=4
      625 GOSUB 1105:PRINT P$;U$
      715 Z(C)=ZC:S(C)=SC
      1099 END
      1105 FOR AA=1 TO IDX
      1115 PRINT " ";:NEXT AA:RETURN
```

```
Change: 160 PRINT "FIRST, WE NEED ONE OF SOMETHING--"
      400 IF U$(LEN(U$),LEN(U$))<>"S" THEN 480
      580 READ ZA,SA
      620 IDX=9:GOSUB 1105:PRINT B;
      710 READ ZC,SC
      870 IF Y$(1,1)<>"N" THEN 540
      1000 FOR L=1 TO 37
```

COMMODORE 64 No Changes Required

COMMODORE VIC-20

Delete: 650

```
Add: 215 PRINT
      216 PRINT "HIT RETURN TO CONTINUE"
      217 INPUT X$
      505 PRINT
      506 PRINT "HIT RETURN TO CONTINUE"
      507 INPUT X$
```

```
Change: 320 PRINT "TELL ME ONE OF YOUR OWN"
      620 PRINT TAB(5);B;TAB(10);P$;U$
      660 PRINT "CHOOSE ONE (1-14)"
      850 PRINT "WANT TO TRY ANOTHER"
      1000 FOR L=1 TO 22
```

TEXAS INSTRUMENTS 99/4A

```
Change: 120 PRINT "IN OUTER SPACE THERE ARE      ENORMOUS
      THINGS--AND TEENY,"
      130 PRINT "TINY ONES! LET ME SHOW YOU"
      140 PRINT "SOME YOU MIGHT NOT KNOW."
```

```

160 PRINT "FIRST, WE NEED ONE OF SOME- THING--
    A STANDARD UNIT."
170 PRINT "JUST FOR FUN, LET'S"
210 PRINT "(IN FACT, IT'S JUST THE RIGHT SIZE)"
230 PRINT "BUT PUT 10 ";G$;"S";"TOGETHER IN ONE
    PLACE AND YOU HAVE"
240 PRINT "A DEKA";G$;"."; " 100 OF"
250 PRINT "THEM MAKE A HECTO";G$;"."
280 PRINT "YOU CAN'T GET MORE THAN 100 OF THEM
    TOGETHER,"
290 PRINT "NO MATTER HOW HARD YOU TRY!"
310 PRINT "WE NEED A DIFFERENT STANDARD"
320 PRINT "UNIT. TELL ME ONE OF YOUR OWN";
350 IF (LEN(U$)>2)*(LEN(U$)<22) THEN 400
400 IF SEG$(U$,LEN(U$),1)<>"S" THEN 480
430 PRINT "ESPECIALLY IF THEY ARE      ALREADY
    PLURAL."
450 PRINT "TRY AGAIN, AND MAKE SURE IT'S SINGULAR."
490 PRINT "EXCELLENT! YOU CAN DEFINITE-LY GET MORE
    THAN"
500 PRINT "100 ";U$;"S TOGETHER."
680 IF (N<1)+(N>14) THEN 660
870 IF SEG$(Y,1,1)<>"N" THEN 540
1000 FOR L=1 TO 28

```

TRS-80 COLOR COMPUTER

```

Add: 215 GOSUB 1115
    305 GOSUB 1115
    565 GOSUB 1115
    1115 PRINT "(PRESS ENTER FOR MORE)";
    1125 INPUT X$
    1135 CLS
    1145 RETURN

```

```

Change: 120 PRINT "IN OUTER SPACE THERE ARE ENORMOUS"; " ";
    130 PRINT "THINGS--AND TEENY, TINY ONES! LET ME"; " ";
    160 PRINT "FIRST, WE NEED ONE OF SOMETHING--"; " ";
    170 PRINT "A STANDARD UNIT. JUST FOR FUN, LET'S"; " ";
    1000 FOR L=1 TO 31

```

Lights! Computer! Action!

Here's a movie premiere — and you're the star of your very own production.
The professional credits will roll up your screen just the way they do
in movies and TV!

☐ Sample Run

ENTER YOUR FULL NAME? REBECCA MOONGLOW
ENTER A FRIEND'S NAME? HARVEY HICCUP
ENTER ANOTHER FRIEND'S NAME? NINA NOONAN
ENTER ANOTHER FRIEND'S NAME? ORSON MASON
DO YOU HAVE A PET (YES OR NO)? YES
ENTER WHAT KIND OF ANIMAL IT IS? DWARF PONY
ENTER YOUR PET'S NAME? MILTON
ENTER THE NAME OF A TEACHER? MR. MUNCHER
ENTER THE NAME OF YOUR TOWN? PARISVILLE

COMPUTER PICTURES PRESENTS

THE RESCUE OF MILTON

STARRING REBECCA MOONGLOW

DIRECTED BY REBECCA MOONGLOW

PRODUCED BY REBECCA MOONGLOW

ASSISTANT DIRECTOR--HARVEY HICCUP

ASSISTANT PRODUCER--NINA NOONAN

PHOTOGRAPHY BY ORSON MASON

EXECUTIVE PRODUCER--MR. MUNCHER

CO-STARRING THAT INCREDIBLE DWARF PONY

THE AMAZING MILTON

FILMED ON LOCATION IN PARISVILLE

IN GLORIOUS COMPUTER COLOR

LIGHTING BY HARVEY HICCUP

SOUND BY NINA NOONAN

EDITED BY ORSON MASON

ORIGINAL SCREENPLAY BY REBECCA MOONGLOW

Break

☐ Program Listing

```
10 REM LIGHTS! COMPUTER! ACTION!
100 PRINT "ENTER YOUR FULL NAME";
110 INPUT S$
120 IF S$="" THEN 100
130 PRINT "ENTER A FRIEND'S NAME";
140 INPUT FF$
150 IF FF$="" THEN 130
160 PRINT "ENTER ANOTHER FRIEND'S NAME";
170 INPUT SF$
180 IF SF$="" THEN 160
190 PRINT "ENTER ANOTHER FRIEND'S NAME";
200 INPUT TF$
210 IF TF$="" THEN 190
220 PRINT "DO YOU HAVE A PET (YES OR NO)";
230 INPUT PT$
240 IF PT$="" THEN 220
250 IF MID$(PT$,1,1)<>"Y" THEN 320
260 PRINT "ENTER WHAT KIND OF ANIMAL IT IS";
270 INPUT K$
280 IF K$="" THEN 260
290 PRINT "ENTER YOUR PET'S NAME";
300 INPUT P$
310 IF P$="" THEN 290
320 PRINT "ENTER THE NAME OF A TEACHER";
330 INPUT T$
340 IF T$="" THEN 320
```

(continued)

```

350 PRINT "ENTER THE NAME OF YOUR TOWN";
360 INPUT C$
370 IF C$="" THEN 350
380 FOR A=1 TO 24
390 PRINT
400 NEXT A
410 A$="COMPUTER PICTURES PRESENTS"
420 GOSUB 850
430 IF MID$(PT$,1,1)="Y" THEN 450
440 P$=T$
450 A$="THE RESCUE OF "+P$
460 GOSUB 850
470 A$="STARRING "+S$
480 GOSUB 850
490 A$="DIRECTED BY "+S$
500 GOSUB 850
510 A$="PRODUCED BY "+S$
520 GOSUB 850
530 A$="ASSISTANT DIRECTOR--"+FF$
540 GOSUB 850
550 A$="ASSISTANT PRODUCER--"+SF$
560 GOSUB 850
570 A$="PHOTOGRAPHY BY "+TF$
580 GOSUB 850
590 A$="EXECUTIVE PRODUCER--"+T$
600 GOSUB 850
610 IF MID$(PT$,1,1)="Y" THEN 630
620 K$="TEACHER"
630 A$="CO-STARRING THAT INCREDIBLE "+K$
640 GOSUB 850
650 A$="THE AMAZING "+P$
660 GOSUB 850
670 A$="FILMED ON LOCATION IN "+C$
680 GOSUB 850
690 A$="IN GLORIOUS COMPUTER COLOR"
700 GOSUB 850
710 A$="LIGHTING BY "+FF$
720 GOSUB 850
730 A$="SOUND BY "+SF$
740 GOSUB 850
750 A$="EDITED BY "+TF$
760 GOSUB 850
770 A$="ORIGINAL SCREENPLAY BY "+S$
780 GOSUB 850
790 FOR C=1 TO 7
800 A$=" "
810 GOSUB 850
820 NEXT C
830 END
840 REM ----- FIRST SUBROUTINE -----
850 TLE=20-(LEN(A$)/2)
860 PRINT TAB(TLE);A$
870 GOSUB 940
880 PRINT
890 GOSUB 940

```

```

900 PRINT
910 GOSUB 940
920 RETURN
930 REM ----- SECOND SUBROUTINE -----
940 FOR LAG=1 TO 400
950 NEXT LAG
960 RETURN

```

☐ If You Have . . .

APPLE II No Changes Required

ATARI

```

Add: 91 DIM S$(40),FF$(40),SF$(40),TF$(40),PT$(3),K$(11)
      92 DIM P$(40),T$(40),C$(40),A$(40),TEMP$(40)
      455 TEMP$=P$
      475 TEMP$=S$
      495 TEMP$=S$
      515 TEMP$=S$
      535 TEMP$=FF$
      555 TEMP$=SF$
      575 TEMP$=TF$
      595 TEMP$=T$
      635 TEMP$=K$
      655 TEMP$=P$
      675 TEMP$=C$
      695 TEMP$=" "
      715 TEMP$=FF$
      735 TEMP$=SF$
      755 TEMP$=TF$
      775 TEMP$=S$
      805 TEMP$=" "
      975 FOR AA=1 TO TLE
      985 PRINT " ";:NEXT AA:RETURN

```

```

Change: 250 IF PT$(1,1)<>"Y" THEN 320
        430 IF PT$(1,1)<>"Y" THEN 450
        450 A$="THE RESCUE OF "
        470 A$="STARRING "
        490 A$="DIRECTED BY "
        510 A$="PRODUCED BY "
        530 A$="ASSISTANT DIRECTOR--"
        550 A$="ASSISTANT PRODUCER--"
        570 A$="PHOTOGRAPHY BY "
        590 A$="EXECUTIVE PRODUCER--"
        610 IF PT$(1,1)="Y" THEN 630
        630 A$="CO-STARRING THAT INCREDIBLE "
        650 A$="THE AMAZING "
        670 A$="FILMED ON LOCATION IN "
        710 A$="LIGHTING BY "

```

(continued)

```

730 A$="SOUND BY "
750 A$="EDITED BY "
770 A$="ORIGINAL SCREENPLAY BY "
850 TLE=20-((LEN(A$)+LEN(TEMP$))/2)
860 GOSUB 975:PRINT A$;TEMP$

```

COMMODORE 64 No Changes Required

COMMODORE VIC-20

Delete: 850

Change:

```

100 PRINT "ENTER YOUR FULL NAME"
130 PRINT "ENTER A FRIEND'S NAME"
160 PRINT "ENTER ANOTHER FRIEND'S NAME"
190 PRINT "ENTER ANOTHER FRIEND'S NAME"
220 PRINT "DO YOU HAVE A PET (YES OR NO) "
260 PRINT "ENTER WHAT KIND OF ANIMAL IT IS"
290 PRINT "ENTER YOUR PET'S NAME"
320 PRINT "ENTER THE NAME OF A TEACHER"
350 PRINT "ENTER THE NAME OF YOUR TOWN"
860 PRINT A$

```

TEXAS INSTRUMENTS 99/4A

Change:

```

220 PRINT "DO YOU HAVE A PET(YES OR NO) ";
250 IF SEG$(PT$,1,1)<>"Y" THEN 320
430 IF SEG$(PT$,1,1)="Y" THEN 450
450 A$="THE RESCUE OF "&P$
470 A$="STARRING "&S$
490 A$="DIRECTED BY "&S$
510 A$="PRODUCED BY "&S$
530 A$="ASSISTANT DIRECTOR--"&FF$
550 A$="ASSISTANT PRODUCER--"&SF$
570 A$="PHOTOGRAPHY BY "&TF$
590 A$="EXECUTIVE PRODUCER--"&T$
610 IF SEG$(PT$,1,1)="Y" THEN 630
630 A$="CO-STARRING THAT INCREDIBLE "&K$
650 A$="THE AMAZING "&P$
670 A$="FILMED ON LOCATION IN "&C$
710 A$="LIGHTING BY "&FF$
730 A$="SOUND BY "&SF$
750 A$="EDITED BY "&TF$
770 A$="ORIGINAL SCREENPLAY BY "&S$
940 FOR LAG=1 TO 100

```

TRS-80 COLOR COMPUTER

Add: 855 IF TLE<1 THEN TLE=1

Change: 850 TLE=15-(LEN(A\$)/2)

Computer Decision-Maker

You'll never need to flip a coin again. Now your computer can take over
all your toughest decisions. Try it!

☐ Sample Run

NEED TO MAKE A DECISION? YOU'LL
FLIP OVER THIS PROGRAM!

YOUR COMPUTER WILL FLIP A COIN FOR YOU.
FIRST, ENTER A FRIEND'S NAME? BILL BUFFALO

WHAT HAPPENS IF THE COIN COMES UP
HEADS? BILL WILL SHAVE OFF HIS HAIR

WHAT HAPPENS IF THE COIN COMES UP
TAILS? BILL WILL MAIL HIS HAIR TO GERMANY

I'M FLIPPING THE COIN...
IT'S TAILS!

BILL WILL MAIL HIS HAIR TO GERMANY

WANT TO TRY AGAIN? YES

WHAT HAPPENS IF THE COIN COMES UP
HEADS? THIS COMPUTER WILL TURN GREEN

WHAT HAPPENS IF THE COIN COMES UP
TAILS? THE HOUSE WILL MELT

I'M FLIPPING THE COIN...
IT'S TAILS!

THE HOUSE WILL MELT

WANT TO TRY AGAIN? NO

☐ Program Listing

```
10 REM COMPUTER DECISION MAKER
100 PRINT "NEED TO MAKE A DECISION? YOU'LL"
110 PRINT "FLIP OVER THIS PROGRAM!"
120 PRINT
130 PRINT "YOUR COMPUTER WILL FLIP A COIN FOR YOU."
140 PRINT "FIRST, ENTER A FRIEND'S NAME";
150 INPUT A$
160 IF A$="" THEN 140
170 IF LEN(A$)<2 THEN 130
180 RANDOMIZE(ASC(MID$(A$,1,1))*ASC(MID$(A$,2,1)))
190 PRINT
200 PRINT "WHAT HAPPENS IF THE COIN COMES UP"
210 PRINT "HEADS";
220 INPUT H$
230 IF H$="" THEN 200
240 PRINT
250 PRINT "WHAT HAPPENS IF THE COIN COMES UP"
260 PRINT "TAILS";
270 INPUT T$
280 IF T$="" THEN 250
290 B=INT(RND(1)*2)
300 IF B=1 THEN C$="TAILS"
310 IF B=0 THEN C$="HEADS"
320 PRINT
330 PRINT "I'M FLIPPING THE COIN..."
340 FOR D=1 TO 900
350 NEXT D
360 PRINT "IT'S ";C$;"!"
370 PRINT
380 IF C$="TAILS" THEN PRINT T$
390 IF C$="HEADS" THEN PRINT H$
400 PRINT
410 PRINT "WANT TO TRY AGAIN";
420 INPUT Y$
430 IF MID$(Y$,1,1)<>"N" THEN 190
```

☐ If You Have . . .

APPLE II

Delete: 180

ATARI

Delete: 180

Add: 95 DIM A\$(40),H\$(100),T\$(100),C\$(7),Y\$(5)
 Change: 290 B=INT(RND(0)*2)
 430 IF Y\$(1,1)<>"N" THEN 190

COMMODORE 64

Delete: 180
 Change: 290 B=INT(RND(0)*2)

COMMODORE VIC-20

Delete: 180
 Change: 140 PRINT "FIRST, ENTER A FRIEND'S NAME"
 210 PRINT "HEADS"
 260 PRINT "TAILS"
 290 B=INT(RND(0)*2)
 410 PRINT "WANT TO TRY AGAIN"

TEXAS INSTRUMENTS 99/4A

Add: 302 IF B=0 THEN 320
 382 PRINT T\$
 392 PRINT H\$
 Change: 100 PRINT "NEED TO MAKE A DECISION? YOU'LL"
 130 PRINT "YOUR COMPUTER WILL FLIP A COIN FOR
 YOU."
 180 RANDOMIZE
 200 PRINT "WHAT HAPPENS IF THE COIN COMES UP"
 250 PRINT "WHAT HAPPENS IF THE COIN COMES UP"
 290 B=INT(RND*2)
 300 C\$="TAILS"
 310 C\$="HEADS"
 380 IF C\$<>"TAILS" THEN 390
 390 IF C\$<>"HEADS" THEN 400
 430 IF SEG\$(Y\$,1,1)<>"N" THEN 190

TRS-80 COLOR COMPUTER

Delete: 180
 Change: 290 B=RND(2)-1

Birthday Surprise

You've got a terrific computer, and you've got this program to go with it. Now you can celebrate your birthday anytime you feel like it. Lucky you!

☐ Sample Run

ENTER YOUR FIRST NAME? ROBIN

IT'S YOUR BIRTHDAY, AND ALL YOUR
FRIENDS HAVE GOTTEN TOGETHER AND
HIRED A SKY-WRITER TO CELEBRATE.

HIT THE ENTER KEY TO SEE HOW?

HAPPY

H A P P Y
H A P P Y
H A P P Y
H A P P Y
H A P P Y
H A P P Y
H A P P Y

BIRTHDAY

B I R T H D A Y
B I R T H D A Y
B I R T H D A Y
B I R T H D A Y

ROBIN

R O B I N
R O B I N
R O B I N
R O B I N
R O B I N
R O B I N
R O B I N

HAPPY

H A P P Y
H A P P Y
H A P P Y
H A P P Y
H A P P Y
H A P P Y
H A P P Y

BIRTHDAY
 B I R T H D A Y
 B I R T H D A Y
 B I R T H D A Y
 B I R T H D A Y

ROBIN
 R O B I N
 R O B I N
 R O B I N
 R O B I N
 R O B I N
 R O B I N
 R O B I N
 R O B I N

Break

☐ Program Listing

```

10 REM BIRTHDAY SURPRISE
100 R=1
110 PRINT "ENTER YOUR FIRST NAME";
120 INPUT FIRST$
130 IF FIRST$="" THEN 110
140 PRINT CHR$(12)
150 PRINT "IT'S YOUR BIRTHDAY, AND ALL YOUR"
160 PRINT "FRIENDS HAVE GOTTEN TOGETHER AND"
170 PRINT "HIRED A SKY-WRITER TO CELEBRATE."
180 PRINT
190 PRINT "HIT THE ENTER KEY TO SEE HOW";
200 INPUT D$
210 PRINT
220 ON R GOTO 230,250,270
230 W$="HAPPY"
240 GOTO 280
250 W$="BIRTHDAY"
260 GOTO 280
270 W$=FIRST$
280 FOR A=0 TO INT(40/LEN(W$))-1
290 FOR C=1 TO LEN(W$)
300 PRINT MID$(W$,C,1);
310 FOR D=1 TO A
320 PRINT CHR$(32);
330 NEXT D
340 NEXT C
350 PRINT
360 NEXT A
370 R=R+1
380 IF R=4 THEN R=1
390 GOTO 220

```

☐ If You Have . . .

APPLE II

Add: 305 IF A=0 THEN 340
365 PRINT:PRINT:PRINT

Change: 140 HOME
190 PRINT "HIT THE RETURN KEY TO SEE HOW";
280 FOR A=0 TO INT(39/LEN(W\$))-1

ATARI

Add: 95 DIM FIRST\$(17),D\$(1),W\$(17)
315 IF A=0 THEN 330

Change: 140 PRINT CHR\$(125)
190 PRINT "HIT THE RETURN KEY TO SEE HOW";
300 PRINT W\$(C,C);

COMMODORE 64

Change: 140 PRINT CHR\$(147)
190 PRINT "HIT THE RETURN KEY TO SEE HOW"

COMMODORE VIC-20

Change: 110 PRINT "ENTER YOUR FIRST NAME"
140 PRINT CHR\$(147)
190 PRINT "HIT THE RETURN KEY TO SEE HOW"

TEXAS INSTRUMENTS 99/4A

Add: 383 R=1

Change: 140 CALL CLEAR
160 PRINT "FRIENDS HAVE GOTTEN TOGETHER AND HIRED"
170 PRINT "A SKY-WRITER TO CELEBRATE."
280 FOR A=0 TO INT(28/LEN(W\$))-1
300 PRINT SEG\$(W\$,C,1);
380 IF R<>4 THEN 220

TRS-80 COLOR COMPUTER

Change: 140 CLS

Balancing Act

It's amazing! Your computer can do a stupendous balancing trick. Even the world's greatest jugglers can't match your machine at this!

☐ Sample Run

LADIES AND GENTLEMEN, STEP RIGHT
UP AND SEE THE GREATEST CIRCUS
ACT OF THE CENTURY!

THIS LITTLE COMPUTER
WILL PERFORM AN ACROBATIC MIRACLE
THAT DEFIES BELIEF. YOU'VE SEEN
INCREDIBLE JUGGLERS BEFORE, BUT
NOTHING LIKE THIS! THIS VERY
COMPUTER WILL BALANCE AN ENDLESS
SET OF SHARP-POINTED DIAMONDS ON
THE BRIDGE OF ITS LONG NOSE. HIT
THE ENTER KEY, THEN SIT BACK...?

```
=
^oe
mlr;9
cS176[q
9_AC?vuXP
BE0m8ptCOMB
kv9F@eOxewhxv
lhPH`bhT5u`7b7S
nv/pcW5TRZ7;VZ/;j
2o/<_>nHAMOBlodhS
[]s\fmUI<noHHGm
ZQ^2esgPUoC3N
9g/`KffNPP^
=25^]]vMK
T;RIDwm
:Rtbh
K:N
Y
v
:?A
mftn/
U2Q`M4V
<14y_9BDX
Kp`r:MmLpog
xBuc_se]MP;aU
S377FxDOPc42^AH
B\<1@^a3^OBHuE_
YECnfYFKU]OS0
j9xxc904AH=
Mnx5N0<gt
3TeU913
0UWgK
6sS
1
```

(continued)

```

T
/k7
wE4;T
mE:g`=Z
T[:e?@9Hy
J[XaUl;CfJ/
WjqcuI8Zi_;uT
EwPQK<kce/o?Z_\
=Fehg\2Soe]AL6w3V
:]wSMkhCi[mw=5SlIrF
rq3ul\l0p;;FKj6AYQYZS
v=3>=tqpxK@wY?;5iAH5V
30sSrSa4rUf\HFLmHSX
XkPj]Ty\804rnZsmy
`Hd6LIKbhv^Yj?5
nH\3\B]x;jWl>
DUnOswcTy4]
DC_wCY@wC
gJjNh?f
E8KmS
PyU
=
0
b=0
118\4
Nnuci?m
J2vIsNu=C
x\ITI`l9c4c
g8;Zm2@>h16JC
k9ApXShnFRD<pEQ
xg2J2Nlv^1E]\AF
9f;ofC_I_hmwU
>`QT@r/DlQT
V7mwSivCr
N73:ACu
\c[2H
d>Y
u
:
peC
h\5s6
;5<cLRw
a[dus`OE\
5HF_UD3rS
FCXvwOt
M3<y\
W\A
Y
n

```

Break

□ Program Listing

```
10 REM BALANCING ACT
100 C$="COMPUTER"
110 PRINT "LADIES AND GENTLEMEN, STEP RIGHT"
120 PRINT "UP AND SEE THE GREATEST CIRCUS"
130 PRINT "ACT OF THE CENTURY!"
140 PRINT
150 PRINT "THIS LITTLE ";
160 FOR J=1 TO 8
170 PRINT CHR$(7);
180 PRINT MID$(C$,J,1);
190 FOR D=1 TO 400
200 NEXT D
210 NEXT J
220 PRINT
230 PRINT
240 PRINT "WILL PERFORM AN ACROBATIC MIRACLE"
250 PRINT "THAT DEFIES BELIEF. YOU'VE SEEN"
260 PRINT "INCREDIBLE JUGGLERS BEFORE, BUT"
270 PRINT "NOTHING LIKE THIS! THIS VERY"
280 PRINT "COMPUTER WILL BALANCE AN ENDLESS"
290 PRINT "SET OF SHARP-POINTED DIAMONDS ON"
300 PRINT "THE BRIDGE OF ITS LONG NOSE. HIT"
310 PRINT "THE ENTER KEY, THEN SIT BACK...";
320 INPUT H$
330 A=INT(RND(1)*19)*2-1
340 IF A<3 THEN 330
350 FOR B=1 TO A STEP 2
360 PRINT TAB(20-B/2);
370 FOR F=1 TO B
380 PRINT CHR$((RND(1)*74)+47);
390 NEXT F
400 NEXT B
410 FOR C=A TO 1 STEP -2
420 PRINT TAB(20-C/2);
430 FOR G=1 TO C
440 PRINT CHR$((RND(1)*74)+47);
450 NEXT G
460 NEXT C
470 GOTO 330
```

☐ If You Have . . .

APPLE II

Add: 395 PRINT
455 PRINT

Change: 310 "THE RETURN KEY, THEN SIT BACK...";

ATARI

Add: 95 DIM C\$(8),H\$(1)
195 SOUND 2,0,0,0
395 PRINT
455 PRINT
485 FOR AA=1 TO IDX
495 PRINT " ";:NEXT AA:RETURN

Change: 170 SOUND 2,147,10,10
180 PRINT C\$(J,J);
310 PRINT "THE RETURN KEY, THEN SIT BACK...";
330 A=INT(RND(0)*19)*2-1
360 IDX=(20-B/2):GOSUB 485
380 PRINT CHR\$((RND(0)*74)+47);
420 IDX=(20-C/2):GOSUB 485
440 PRINT CHR\$((RND(0)*74)+47);

COMMODORE 64

Delete: 170

Add: 395 PRINT
455 PRINT

Change: 310 PRINT "THE RETURN KEY, THEN SIT BACK...";
330 A=INT(RND(0)*19)*2-1
380 PRINT CHR\$((RND(0)*74)+47);
440 PRINT CHR\$((RND(0)*74)+47);

COMMODORE VIC-20

Delete: 170

Add: 395 PRINT
455 PRINT

Change: 310 PRINT "THE RETURN KEY, THEN SIT BACK..."
330 A=INT(RND(0)*10)*2-1
360 PRINT TAB(11-B/2);
380 PRINT CHR\$((RND(0)*74)+47);
420 PRINT TAB(11-C/2);
440 PRINT CHR\$((RND(0)*74)+47);

TEXAS INSTRUMENTS 99/4A

Change: 110 PRINT "LADIES AND GENTLEMEN, STEP RIGHT UP
AND SEE"
120 PRINT "THE GREATEST CIRCUS"
170 CALL SOUND(020,400,2)
180 PRINT SEG\$(C\$,J,1)
240 PRINT "WILL PERFORM AN ACROBATIC MIRACLE"
280 PRINT "COMPUTER WILL BALANCE AN ENDLESS
SET"
290 PRINT "OF SHARP-POINTED DIAMONDS ON"
300 PRINT "THE BRIDGE OF ITS LONG NOSE. HIT THE"
310 PRINT "ENTER KEY, THEN SIT BACK...";
330 A=INT(RND*14)*2-1
360 PRINT TAB(14-B/2);
380 PRINT CHR\$((RND*74)+47);
420 PRINT TAB(14-C/2);
440 PRINT CHR\$((RND*74)+47);

TRS-80 COLOR COMPUTER

Add: 395 PRINT
455 PRINT

Change: 170 SOUND 173,2
330 A=RND(15)*2-1
360 PRINT TAB(16-B/2);
380 PRINT CHR\$(RND(74)+46);
420 PRINT TAB(16-C/2);
440 PRINT CHR\$(RND(74)+46);

Wise Old Elf

Got a burning question you need an answer to? So does the elf.
If you help him, he'll return the favor.

☐ Sample Run

NEAR THE SOUTH POLE IS A VERY OLD AND
COLD ELF WHO CAN TELL FORTUNES--BUT
ONLY IF YOU SOLVE A PROBLEM FOR HIM.

WANT THE ELF TO HELP YOU? YES

WHAT IS YOUR NAME, HUMAN? GRETTA

WELCOME TO MY ICE CAVE, GRETTA.

HERE IS MY PROBLEM.
I CANNOT SEEM TO SOLVE IT.

$$54 + 10 = ?$$

WHAT IS YOUR ANSWER? 64

RIGHT! VERY GOOD, GRETTA.
NOW PLEASE ASK ME A QUESTION--
ONE I CAN ANSWER YES OR NO.
? WILL I GROW ANOTHER FOOT TOMORROW?

NO, NOT REALLY.

DO YOU WISH TO CONTINUE? YES

HERE IS MY PROBLEM.
I CANNOT SEEM TO SOLVE IT.

$$68 + 96 = ?$$

WHAT IS YOUR ANSWER? 154

NO, HUMAN! $68 + 96 = 164$

DO YOU WISH TO TRY AGAIN? YES

HERE IS MY PROBLEM.
I CANNOT SEEM TO SOLVE IT.

$$92 + 54 = ?$$

WHAT IS YOUR ANSWER? 146

RIGHT! VERY GOOD, GRETTA.

NOW PLEASE ASK ME A QUESTION--

ONE I CAN ANSWER YES OR NO.

? WILL MY COMPUTER PROGRAM WIN THE PRIZE?

MOST DEFINITELY.

DO YOU WISH TO CONTINUE? NO

GOODBYE, HUMAN.

☐ Program Listing

```
10 REM WISE OLD ELF
100 PRINT "NEAR THE SOUTH POLE IS A VERY OLD AND"
110 PRINT "COLD ELF WHO CAN TELL FORTUNES--BUT"
120 PRINT "ONLY IF YOU SOLVE A PROBLEM FOR HIM."
130 PRINT
140 PRINT "WANT THE ELF TO HELP YOU";
150 INPUT LUCK$
160 IF MID$(LUCK$,1,1)="N" THEN END
170 PRINT
180 PRINT "WHAT IS YOUR NAME, HUMAN";
190 INPUT NAM$
200 IF LEN(NAM$)<2 THEN 180
210 RANDOMIZE(ASC(MID$(NAM$,1,1))*ASC(MID$(NAM$,2,1)))
220 PRINT
230 PRINT "WELCOME TO MY ICE CAVE, ";NAM$;". "
240 PRINT
250 A=INT(RND(1)*100)+1
260 B=INT(RND(1)*100)+1
270 PRINT "HERE IS MY PROBLEM."
280 PRINT "I CANNOT SEEM TO SOLVE IT."
290 PRINT
300 PRINT TAB(7);A;"+";B;"= ?"
310 PRINT
320 PRINT
330 PRINT "WHAT IS YOUR ANSWER";
340 INPUT ANSWER
350 IF ANSWER=0 THEN 330
360 IF ANSWER=A+B THEN 440
370 PRINT
380 PRINT "NO, HUMAN!";A;"+";B;"=";A+B
390 PRINT
400 PRINT "DO YOU WISH TO TRY AGAIN";
410 INPUT AGAIN$
420 IF MID$(AGAIN$,1,1)="N" THEN END
```

(continued)

```

430 GOTO 240
440 PRINT
450 PRINT
460 PRINT "RIGHT! VERY GOOD, ";NAM$;"."
470 PRINT "NOW PLEASE ASK ME A QUESTION--"
480 PRINT "ONE I CAN ANSWER YES OR NO."
490 INPUT Q$
500 IF LEN(Q$)=0 THEN 470
510 PRINT
520 ON INT(RND(1)*6)+1 GOTO 530,550,570,590,610,630
530 PRINT "YES, I BELIEVE SO."
540 GOTO 640
550 PRINT "NO, NOT REALLY."
560 GOTO 640
570 PRINT "MOST DEFINITELY."
580 GOTO 640
590 PRINT "NOT A CHANCE."
600 GOTO 640
610 PRINT "PROBABLY YES."
620 GOTO 640
630 PRINT "I HAVE TO SAY NO."
640 PRINT
650 PRINT
660 PRINT "DO YOU WISH TO CONTINUE";
670 INPUT C$
680 IF MID$(C$,1,1)<>"N" THEN 240
690 PRINT "GOODBYE, HUMAN."

```

☐ If You Have . . .

APPLE II

Delete: 210

ATARI

Delete: 210

Add: 95 DIM LUCK\$(3),AGAIN\$(3),Q\$(64),NAM\$(29),C\$(3)

Change: 160 IF LUCK\$(1,1)="N" THEN END
250 A=INT(RND(0)*100)+1
260 B=INT(RND(0)*100)+1
300 PRINT " ";A;"+";B;"= ?"
420 IF AGAIN\$(1,1)="N" THEN END
520 ON INT(RND(0)*6)+1 GOTO 530,550,570,590,610,630
680 IF C\$(1,1)<>"N" THEN 240

COMMODORE 64

Delete: 210

Change: 250 A=INT(RND(0)*100)+1
260 B=INT(RND(0)*100)+1
520 ON INT(RND(0)*6)+1 GOTO 530,550,570,590,610,630

COMMODORE VIC-20

Delete: 210

Change: 140 PRINT "WANT THE ELF TO HELP YOU"
180 PRINT "WHAT IS YOUR NAME, HUMAN"
250 A=INT(RND(0)*100)+1
260 B=INT(RND(0)*100)+1
520 ON INT(RND(0)*6)+1 GOTO 530,550,570,590,610,630
330 PRINT "WHAT IS YOUR ANSWER"
400 PRINT "DO YOU WISH TO TRY AGAIN"
660 PRINT "DO YOU WISH TO CONTINUE"

TEXAS INSTRUMENTS 99/4A

Add: 705 END

Change: 100 PRINT "NEAR THE SOUTH POLE IS A VERY OLD
AND COLD ELF"
110 PRINT "WHO CAN TELL FORTUNES--BUT"
160 IF SEG\$(LUCK\$,1,1)="N" THEN 705
210 RANDOMIZE
250 A=INT(RND*100)+1
260 B=INT(RND*100)+1
420 IF SEG\$(AGAIN\$,1,1)="N" THEN 705
520 ON INT(RND*6)+1 GOTO 530,550,570,590,610,630
680 IF SEG\$(C\$,1,1)<>"N" THEN 240

TRS-80 COLOR COMPUTER

Delete: 210

Change: 250 A=RND(100)
260 B=RND(100)
520 ON RND(6) GOTO 530,550,570,590,610,630

Your Name in Lights

Congratulations! You're a star! And now it's time to splash your name across the screen for the world to admire. Who knows? When your friends see this, they may ask for your autograph!

☐ Sample Run

YOU'VE FINALLY MADE IT TO THE TOP
OF THE MOVIE INDUSTRY. NOW THE
WHOLE WORLD CAN SEE YOUR NAME IN
LIGHTS ON EVERY MOVIE THEATRE SIGN.

HIT THE ENTER KEY TO TURN ON THE
SIGN AT THE OPENING OF YOUR LATEST
SMASH MOVIE HIT?

ENTER YOUR FIRST NAME? BO
ENTER YOUR LAST NAME? PEEPLEY

BO PEEPLEY BO PEEPLEY BO PEEPLEY BO
PEEPLEY BO PEEPLEY BO PEEPLEY BO PEEPLEY BO
PEEPLEY BO PEEPLEY BO PEEPLEY BO PEEPLEY BO
PEEPLEY BO PEEPLEY BO PEEPLEY BO PEEPLEY BO
PEEPLEY BO PEEPLEY BO PEEPLEY BO PEEPLEY BO
PEEPLEY BO PEEPLEY BO PEEPLEY BO PEEPLEY BO
PEEPLEY BO PEEPLEY BO PEEPLEY BO PEEPLEY BO
PEEPLEY BO PEEPLEY BO PEEPLEY BO PEEPLEY BO
PEEPLEY BO PEEPLEY BO PEEPLEY BO PEEPLEY BO
PEEPLEY BO PEEPLEY BO PEEPLEY BO PEEPLEY BO
PEEPLEY BO PEEPLEY BO PEEPLEY BO PEEPLEY BO
PEEPLEY BO PEEPLEY BO PEEPLEY BO PEEPLEY BO
PEEPLEY BO PEEPLEY BO PEEPLEY BO PEEPLEY BO

Break

☐ Program Listing

```
10 REM YOUR NAME IN LIGHTS
100 PRINT "YOU'VE FINALLY MADE IT TO THE TOP"
110 PRINT "OF THE MOVIE INDUSTRY. NOW THE"
120 PRINT "WHOLE WORLD CAN SEE YOUR NAME IN"
130 PRINT "LIGHTS ON EVERY MOVIE THEATRE SIGN."
140 PRINT
150 PRINT "HIT THE ENTER KEY TO TURN ON THE"
160 PRINT "SIGN AT THE OPENING OF YOUR LATEST"
170 PRINT "SMASH MOVIE HIT";
180 INPUT D$
190 PRINT
200 PRINT "ENTER YOUR FIRST NAME";
210 INPUT FIRST$
220 IF FIRST$="" THEN 200
230 PRINT "ENTER YOUR LAST NAME";
240 INPUT LAST$
250 IF LAST$="" THEN 230
260 PRINT
270 PRINT FIRST$;CHR$(32);LAST$;CHR$(32);
280 GOTO 270
```

☐ If You Have . . .

APPLE II

Change: 150 PRINT "HIT THE RETURN KEY TO TURN ON THE"

ATARI

Add: 95 DIM D\$(1),FIRST\$(40),LAST\$(40)

Change: 150 PRINT "HIT THE RETURN KEY TO TURN ON THE"

COMMODORE 64

Change: 150 PRINT "HIT THE RETURN KEY TO TURN ON THE"

COMMODORE VIC-20

Change: 150 PRINT "HIT THE RETURN KEY TO TURN ON THE"
200 PRINT "ENTER YOUR FIRST NAME"
230 PRINT "ENTER YOUR LAST NAME"

(continued)

TEXAS INSTRUMENTS 99/4A

Change: 100 PRINT "YOU'VE FINALLY MADE IT TO THE TOP"
110 PRINT "OF THE MOVIE INDUSTRY. NOW THE WHOLE
WORLD"
120 PRINT "CAN SEE YOUR NAME IN LIGHTS"
130 PRINT "ON EVERY MOVIE THEATRE SIGN."

TRS-80 COLOR COMPUTER

No Changes Required

Magic Words

Ready to read something no other human on earth has ever seen before? Run this program, and you'll create words that are totally new
— and utterly insane!

☐ Sample Run

I'LL TAKE A WORD YOU ENTER
AND CREATE MADE-UP WORDS BASED
ON YOUR WORD'S PATTERN OF
VOWELS AND CONSONANTS.

HOW MANY WORDS WOULD YOU LIKE
ME TO MAKE UP (FROM 1 TO 20)? 5

WHAT'S YOUR WORD? BOX

PAP
VOV
TUS
WAD
CIW

WANT TO TRY ANOTHER? YES

HOW MANY WORDS WOULD YOU LIKE
ME TO MAKE UP (FROM 1 TO 20)? 3

WHAT'S YOUR WORD? MAMA

DOGE
FUXI
LEHA

WANT TO TRY ANOTHER? YES

HOW MANY WORDS WOULD YOU LIKE
ME TO MAKE UP (FROM 1 TO 20)? 6

WHAT'S YOUR WORD? BEEP

DUUG
KIEN
DEES
KUOY
TUUT
LEOR

(continued)

WANT TO TRY ANOTHER? YES

HOW MANY WORDS WOULD YOU LIKE
ME TO MAKE UP (FROM 1 TO 20)? 5

WHAT'S YOUR WORD? ADAGE

OMAXO
ARAMU
UWONA
ILICA
IYUDI

WANT TO TRY ANOTHER? YES

HOW MANY WORDS WOULD YOU LIKE
ME TO MAKE UP (FROM 1 TO 20)? 5

WHAT'S YOUR WORD? CABLE

BERRI
WUTXU
DUTBI
JUNHA
BONDA

WANT TO TRY ANOTHER? NO

☐ Program Listing

```
10 REM MAGIC WORDS
100 PRINT "I'LL TAKE A WORD YOU ENTER"
110 PRINT "AND CREATE MADE-UP WORDS BASED"
120 PRINT "ON YOUR WORD'S PATTERN OF"
130 PRINT "VOWELS AND CONSONANTS."
140 PRINT
150 PRINT "HOW MANY WORDS WOULD YOU LIKE"
160 PRINT "ME TO MAKE UP (FROM 1 TO 20)";
170 INPUT M
180 IF M<1 OR M>20 THEN 150
190 PRINT
200 PRINT "WHAT'S YOUR WORD";
210 INPUT W$
220 IF W$="" THEN 200
230 IF LEN(W$)<36 THEN 270
240 PRINT "SORRY--YOUR WORD CAN'T BE LONGER"
250 PRINT "THAN 35 LETTERS. TRY AGAIN."
260 GOTO 200
270 PRINT
```

```

280 FOR B=1 TO M
290 PRINT TAB(5);
300 FOR C=1 TO LEN(W$)
310 D=ASC(MID$(W$,C,1))
320 IF D>64 AND D<91 THEN 350
330 PRINT CHR$(D);
340 GOTO 430
350 IF D=65 THEN 420
360 IF D=69 THEN 420
370 IF D=73 THEN 420
380 IF D=79 THEN 420
390 IF D=85 THEN 420
400 GOSUB 520
410 GOTO 430
420 GOSUB 620
430 NEXT C
440 PRINT
450 NEXT B
460 PRINT
470 PRINT "WANT TO TRY ANOTHER";
480 INPUT Y$
490 IF MID$(Y$,1,1)="N" THEN END
500 GOTO 140
510 REM --- CONSONANT SUBROUTINE ---
520 A=INT((RND(1)*25)+65)
530 IF A=65 THEN 520
540 IF A=69 THEN 520
550 IF A=73 THEN 520
560 IF A=79 THEN 520
570 IF A=85 THEN 520
580 IF A=81 THEN 520
590 PRINT CHR$(A);
600 RETURN
610 REM --- VOWEL SUBROUTINE ---
620 Z=INT(RND(1)*5)+1
630 IF Z=1 THEN PRINT CHR$(65);
640 IF Z=2 THEN PRINT CHR$(69);
650 IF Z=3 THEN PRINT CHR$(73);
660 IF Z=4 THEN PRINT CHR$(79);
670 IF Z=5 THEN PRINT CHR$(85);
680 RETURN

```

☐ If You Have . . .

APPLE II No Changes Required

ATARI

Add: 95 DIM W\$(46),Y\$(3)

Change: 290 PRINT " ";
310 D=ASC(W\$(C,C))
490 IF Y\$(1,1)="N" THEN END
520 A=INT((RND(0)*25)+65)

COMMODORE 64 No Changes Required

COMMODORE VIC-20

Change: 160 PRINT "ME TO MAKE UP (FROM 1 TO 10)"
200 PRINT "WHAT'S YOUR WORD"
470 PRINT "WANT TO TRY ANOTHER"

TEXAS INSTRUMENTS 99/4A

Add: 632 PRINT CHR\$(65)
642 PRINT CHR\$(69)
652 PRINT CHR\$(73)
662 PRINT CHR\$(79)
672 PRINT CHR\$(85)
705 END

Change: 150 PRINT "HOW MANY WORDS WOULD YOU LIKE"
180 IF (M<1)+(M>20) THEN 150
310 D=ASC(SEG\$(W\$,C,1))
320 IF (D>64)*(D<91) THEN 350
490 IF SEG\$(Y\$,1,1)="N" THEN 705
520 A=INT((RND*25)+65)
630 IF Z<>1 THEN 640
640 IF Z<>2 THEN 650
650 IF Z<>3 THEN 660
660 IF Z<>4 THEN 670
670 IF Z<>5 THEN 680

TRS-80 COLOR COMPUTER

Change: 520 A=RND(26)+64
620 Z=RND(5)

Lost in the Fun House

The fun house is fun, all right — but somehow you’ve managed to get lost inside. Try to find your way back out — just try!

□ Sample Run

WHAT CITY IS THE FUNHOUSE IN? HELSINKI

YOU'RE LOST IN THE FUNHOUSE! HOW CAN YOU POSSIBLY ESCAPE? ONLY BY FINDING THE SECRET PATH TO THE OUTSIDE WORLD. THERE ARE TEN ROOMS AND EACH HAS FOUR DOORS--BUT MOST ARE LOCKED. YOU'D BETTER START SOON, BEFORE THE OPEN DOORS ARE LOCKED TOO!!

HIT THE ENTER KEY TO BEGIN?

! ! ! ! ! ! ! ! ! ! ! ! ! ! ! ! ! ! ! !

[illegible]

YOU ARE ARE IN ROOM 1 AT DOOR 1
IT IS CLOSED. WHICH DOOR DO YOU
WANT TO TRY NOW (1-4)? 3

(continued)

□ Program Listing

```
10 REM LOST IN THE FUNHOUSE
100 DIM T(10)
110 PRINT CHR$(12)
120 W=1
130 FW=1
140 Q=1
150 V=0
160 ES=0
170 PRINT "WHAT CITY IS THE FUNHOUSE IN";
180 INPUT U$
190 IF LEN(U$)<2 THEN 170
200 RANDOMIZE (ASC(MID$(U$,1,1))*ASC(MID$(U$,2,1)))
210 PRINT
220 PRINT "YOU'RE LOST IN THE FUNHOUSE! HOW CAN"
230 PRINT "YOU POSSIBLY ESCAPE? ONLY BY FINDING"
240 PRINT "THE SECRET PATH TO THE OUTSIDE WORLD."
250 PRINT "THERE ARE TEN ROOMS AND EACH HAS FOUR"
260 PRINT "DOORS--BUT MOST ARE LOCKED. YOU'D"
270 PRINT "BETTER START SOON, BEFORE THE OPEN"
280 PRINT "DOORS ARE LOCKED TOO!!"
290 PRINT
300 PRINT "HIT THE ENTER KEY TO BEGIN";
310 INPUT D$
320 FOR R=1 TO 10
330 T(R)=INT(RND(1)*4)+1
340 NEXT R
350 GOSUB 580
360 PRINT TAB(3);"YOU ARE IN ROOM ";Q;" AT DOOR ";W
370 PRINT TAB(3);"IT IS CLOSED. WHICH DOOR DO YOU"
380 PRINT TAB(3);"WANT TO TRY NOW (1-4)";
390 INPUT W$
400 IF W$="" THEN 350
410 W=VAL(W$)
420 IF W>0 AND W<5 THEN 450
430 W=FW
440 GOTO 350
450 V=V+1
460 FW=W
470 IF W<>T(Q) THEN 350
480 Q=Q+1
490 IF Q<11 THEN 520
500 ES=1
510 GOSUB 840
520 W=1
530 GOSUB 840
540 PRINT TAB(3);"THE DOOR OUT OF ROOM ";Q-1;" IS OPEN!"
550 PRINT TAB(3);"HIT THE ENTER KEY TO CONTINUE";
560 INPUT D$
570 GOTO 350
580 REM --- CLOSED DOOR SUBROUTINE ---
590 PRINT CHR$(12)
600 FOR J=1 TO 5
```

(continued)

```

610 PRINT TAB(2);"!";
620 NEXT J
630 FOR K=1 TO 12
640 PRINT TAB(2);"!";TAB(18);
650 FOR L=1 TO 8
660 PRINT "M";
670 NEXT L
680 PRINT
690 NEXT K
700 PRINT CHR$(32);"!";
710 FOR M=1 TO 15
720 PRINT "-";
730 NEXT M
740 FOR N=1 TO 8
750 PRINT "M";
760 NEXT N
770 FOR P=1 TO 13
780 PRINT "-";
790 NEXT P
800 PRINT
810 PRINT "/"
820 RETURN
830 REM --- OPEN DOOR SUBROUTINE ---
840 PRINT CHR$(12)
850 FOR J=1 TO 5
860 PRINT TAB(2);"!";
870 IF J<5 THEN 920
880 PRINT TAB(17);
890 FOR Z=1 TO 10
900 PRINT "=";
910 NEXT Z
920 NEXT J
930 FOR K=1 TO 12
940 PRINT TAB(2);"!";TAB(10);
950 FOR L=1 TO 8
960 PRINT "M";
970 NEXT L
980 IF ES<>1 THEN 1000
990 GOTO 1080
1000 IF K>9 THEN 1040
1010 FOR X=1 TO 8
1020 PRINT ".";
1030 NEXT X
1040 IF K<>10 THEN 1080
1050 FOR Y=1 TO 8
1060 PRINT "-";
1070 NEXT Y
1080 PRINT TAB(26);"!";
1090 NEXT K
1100 PRINT CHR$(32);"!";
1110 FOR M=1 TO 7
1120 PRINT "-";
1130 NEXT M
1140 FOR N=1 TO 8
1150 PRINT "M";

```

```

1160 NEXT N
1170 PRINT TAB(26);"!";
1180 FOR P=1 TO 12
1190 PRINT "_";
1200 NEXT P
1210 PRINT
1220 PRINT "/"
1230 IF ES=1 THEN 1250
1240 RETURN
1250 PRINT TAB(15);"YOU DID IT!"
1260 PRINT TAB(5);"YOU ESCAPED IN ONLY ";V;" TRIES!!"
1270 PRINT TAB(12);"WANT TO TRY AGAIN";
1280 INPUT Y$
1290 IF MID$(Y$,1,1)="N" THEN END
1300 GOTO 110

```

☐ If You Have . . .

APPLE II

Delete: 200

Add: 915 PRINT

Change:

```

110 HOME
300 PRINT "HIT THE RETURN KEY TO BEGIN";
550 PRINT TAB(3);"HIT THE RETURN KEY TO CONTINUE";
590 HOME
720 PRINT CHR$(95);
780 PRINT CHR$(95);
840 HOME
870 IF J<5 THEN 915
1060 PRINT CHR$(95);
1120 PRINT CHR$(95);
1190 PRINT CHR$(95);

```

ATARI

Delete: 200

Add:

```

95 DIM U$(17),D$(1),W$(1),Y$(3)
1035 IF K>10 THEN 1085
1084 IF K<=10 THEN 1090
1085 IDX=8:GOSUB 1315:PRINT "!"
1315 FOR AA=1 TO IDX
1325 PRINT " ";:NEXT AA:RETURN

```

(continued)

Change: 110 PRINT CHR\$(125)
 300 PRINT "HIT THE RETURN KEY TO BEGIN";
 330 T(R)=INT(RND(0)*4)+1
 360 IDX=3:GOSUB 1315:PRINT "YOU ARE IN ROOM ";Q;
 "AT DOOR ";W
 370 GOSUB 1315:PRINT "IT IS CLOSED. WHICH DOOR
 DO YOU "
 380 GOSUB 1315:PRINT "WANT TO TRY NOW (1-4)";
 540 IDX=3:GOSUB 1315:PRINT "THE DOOR OUT OF ROOM ";
 Q-1;" IS OPEN!"
 550 GOSUB 1315:PRINT "HIT THE RETURN KEY TO CONTINUE";
 590 PRINT CHR\$(125)
 610 IDX=1:GOSUB 1315:PRINT "!"
 640 IDX=1:GOSUB 1315:PRINT "!";:IDX=15:GOSUB 1315
 700 PRINT CHR\$(32);"!"
 840 PRINT CHR\$(125)
 860 IDX=1:GOSUB 1315:PRINT "!"
 880 IDX=15:GOSUB 1315
 940 IDX=2:GOSUB 1315:PRINT "!";:IDX=8:GOSUB 1315
 990 GOTO 1085
 1080 PRINT "!"
 1170 IDX=8:GOSUB 1315:PRINT "!"
 1180 FOR P=1 TO 10
 1250 IDX=15:GOSUB 1315:PRINT "YOU DID IT!"
 1260 IDX=5:GOSUB 1315:PRINT "YOU ESCAPED IN ONLY ";
 V;" TRIES!!"
 1270 IDX=12:GOSUB 1315:PRINT "WANT TO TRY AGAIN";
 1290 IF Y\$(1,1)="N" THEN END

COMMODORE 64

Delete: 200

Add: 855 PRINT
 915 PRINT

Change: 110 PRINT CHR\$(147)
 300 PRINT "HIT THE RETURN KEY TO BEGIN";
 330 T(R)=INT(RND(0)*4)+1
 550 PRINT "HIT THE RETURN KEY TO CONTINUE";
 590 PRINT CHR\$(147)
 700 PRINT TAB(2);"!"
 810 PRINT CHR\$(32);"/"
 840 PRINT CHR\$(147)
 1100 PRINT TAB(2);"!"
 1220 PRINT CHR\$(32);"/"

COMMODORE VIC-20

Delete: 200,770,790,1110,1130,1180,1200,1210

Add: 855 PRINT
 915 PRINT

Change: 110 PRINT CHR\$(147)
 170 PRINT "WHAT CITY IS THE FUNHOUSE IN"

```

300 PRINT "HIT THE RETURN KEY TO BEGIN"
330 T(R)=INT(RND(0)*4)+1
380 PRINT TAB(3);"WANT TO TRY NOW (1-4)"
550 PRINT "HIT THE RETURN KEY TO CONTINUE"
590 PRINT CHR$(147)
640 PRINT TAB(2);"!";TAB(12);
700 PRINT TAB(2);"!";
710 FOR M=1 TO 9
810 PRINT CHR$(32);"/"
840 PRINT CHR$(147)
880 PRINT TAB(11);
940 PRINT TAB(2);"!";TAB(4);
1080 PRINT TAB(20);"!"
1100 PRINT TAB(2);"!";
1170 PRINT TAB(20);"!";
1220 PRINT CHR$(32);"/"
1270 PRINT TAB(12);"WANT TO TRY AGAIN"

```

TEXAS INSTRUMENTS 99/4A

Add: 1305 END

Change:

```

110 CALL CLEAR
200 RANDOMIZE
230 PRINT "POSSIBLY ESCAPE? ONLY BY FINDING"
240 PRINT "THE SECRET PATH TO THE OUT- SIDE WORLD."
270 PRINT "BETTER START SOON, BEFORE THE OPEN"
330 T(R)=INT(RND*4)+1
420 IF (W>0)*(W<5) THEN 450
590 CALL CLEAR
640 PRINT TAB(2);"!";TAB(12);
710 FOR M=1 TO 9
770 FOR P=1 TO 7
840 CALL CLEAR
880 PRINT TAB(11);
940 PRINT TAB(2);"!";TAB(4);
1080 PRINT TAB(20);"!"
1170 PRINT TAB(20);"!";
1180 FOR P=1 TO 8
1290 IF SEG$(Y$,1,1)="N" THEN 1305

```

TRS-80 COLOR COMPUTER

Change:

```

110 CLS
200 REM
330 T(R)=RND(4)
590 CLS
700 PRINT TAB(2);"!";
770 FOR P=1 TO 5
810 PRINT " /"
840 CLS
1180 FOR P=1 TO 4

```

Watch It Grow!

A computer is a terrific helper when it comes to solving complicated problems.
But the problem here is trickier than most. If you get it right,
you're pretty sharp!

☐ Sample Run

A FARMER HAS JUST CUT HIS HAYFIELD.
HE HAS PUT THE HAY INTO EIGHT
DIFFERENT STACKS. HE WANTS TO
TO KEEP A PART OF EACH STACK FOR
HIMSELF, AND GIVE THE REST TO HIS
WORKERS. HELP HIM WITH THIS PROBLEM:

THE FARMER KEEPS $6/7$ OF STACK #1
THE FARMER KEEPS $6/7$ OF STACK #2
THE FARMER KEEPS $1/8$ OF STACK #3
THE FARMER KEEPS $1/5$ OF STACK #4
THE FARMER KEEPS $1/7$ OF STACK #5
THE FARMER KEEPS $2/8$ OF STACK #6
THE FARMER KEEPS $3/4$ OF STACK #7
AND
THE FARMER KEEPS $1/8$ OF STACK #8

THEN HE PUTS ALL OF HIS HAY TOGETHER.
NOW HOW MANY HAYSTACKS DOES HE HAVE? 4 $3/4$

NOPE! THERE'S ONLY ONE STACK!

☐ Program Listing

```
10 REM WATCH IT GROW!
100 PRINT
110 PRINT "A FARMER HAS JUST CUT HIS HAYFIELD."
120 PRINT "HE HAS PUT THE HAY INTO EIGHT"
130 PRINT "DIFFERENT STACKS. HE WANTS TO"
140 PRINT "TO KEEP A PART OF EACH STACK FOR"
150 PRINT "HIMSELF, AND GIVE THE REST TO HIS"
160 PRINT "WORKERS. HELP HIM WITH THIS PROBLEM:"
```

```

170 PRINT
180 FOR A=1 TO 8
190 T=INT(RND(1)*8)+1
200 B=INT(RND(1)*8)+1
210 IF T>B THEN 190
220 IF T=B THEN 190
230 PRINT TAB(2);
240 PRINT "THE FARMER KEEPS ";T;"/";B;" OF STACK #";A
250 IF A=7 THEN PRINT TAB(13);"AND
260 NEXT A
270 PRINT
280 PRINT "THEN HE PUTS ALL OF HIS HAY TOGETHER."
290 PRINT "NOW HOW MANY HAYSTACKS DOES HE HAVE";
300 INPUT HAY$
310 PRINT
320 PRINT
330 IF HAY$="" THEN 290
340 IF MID$(HAY$,1,3)="ONE" THEN 430
350 IF LEN(HAY$)=1 AND MID$(HAY$,1,1)="1" THEN 430
360 PRINT
370 FOR I=1 TO 6
380 READ R$
390 J$=R$+J$
400 NEXT I
410 PRINT J$
420 END
430 PRINT "EITHER YOU'VE HEARD THIS BEFORE, OR"
440 PRINT "YOU'RE A PRETTY SMART COMPUTER USER!"
450 DATA ACK!,E ST,LY ON,ERE'S ON,PE! TH,NO

```

☐ If You Have . . .

APPLE II No Changes Required

ATARI

```

Add: 95 DIM HAY$(7),J$(34),R$(34)
      345 IF LEN(HAY$)=1 THEN 360
      365 AB=6
      405 AB=AB-1
      412 RESTORE
      415 IF AB>0 THEN 370
      455 END
      465 FOR AA=1 TO IDX
      475 PRINT " ";:NEXT AA:RETURN

```

Change: 190 T=INT(RND(0)*8)+1
 200 B=INT(RND(0)*8)+1

(continued)

```

230 IDX=1:GOSUB 465
250 IF A=7 THEN IDX=13:GOSUB 465:PRINT "AND"
340 IF LEN(HAY$)=1 AND HAY$(1,1)="1" THEN 430
370 FOR I=1 TO AB
390 J$=R$
410 PRINT J$;

```

COMMODORE 64 No Changes Required

COMMODORE VIC-20

Delete: 230

Add: 174 PRINT "HIT RETURN TO CONTINUE"
175 INPUT X\$

Change: 290 PRINT "NOW HOW MANY HAYSTACKS DOES HE HAVE"

TEXAS INSTRUMENTS 99/4A

Add: 252 PRINT TAB(13); "AND"

Change: 110 PRINT "A FARMER HAS JUST CUT HIS HAYFIELD.
HE"
120 PRINT "HAS PUT THE HAY INTO EIGHT"
150 PRINT "HIMSELF, AND GIVE THE REST TO HIS"
190 T=INT(RND*8)+1
200 B=INT(RND*8)+1
250 IF A<>7 THEN 260
340 IF SEG\$(HAY\$,1,3)="ONE" THEN 430
350 IF (LEN(HAY\$)=1)*(SEG\$(HAY\$,1,1)="1") THEN 430
390 J\$=R\$&J\$
430 PRINT "EITHER YOU'VE HEARD THIS BEFORE, OR
YOU'RE A"
440 PRINT "PRETTY SMART COMPUTER USER!"

TRS-80 COLOR COMPUTER

Add: 175 GOSUB 505
255 IF A=6 THEN GOSUB 505
445 END
505 PRINT "(PRESS ENTER FOR MORE)";
515 INPUT X\$
525 CLS
535 RETURN

Change: 190 T=RND(8)
200 B=RND(8)

Amaze Your Friends!

To have fun with this one, you have to know the secret. Here it is:
The computer will ask for a person's name. As soon as you enter that name,
the program will print nothing but insults about the person.

But there's a way to get the computer to print nothing but compliments —
especially about a wonderful person like you. The secret is to type in a period
— you know, a little dot (.) — right after the name and right before you hit the
ENTER key. If you do it fast, no one will ever notice.

Practice with this one before you try it out on your friends. They'll never figure
out why your computer likes you so much — and thinks they're so awful!

☐ Sample Run

ENTER SOMEONE'S FIRST NAME? HARRIET

HORRIBLE HARRIET!

AND YOU KNOW WHAT ELSE?

HARRIET IS ALSO ABOMINABLE AND BORING
AND CREEPY AND DUMB AND EVIL AND FREAKY
AND GOOFY AND IDIOTIC AND JERKY AND
KLUTZY AND LOUSY AND MEAN AND NASTY AND
ODD AND POMPOUS AND QUARRELSOME AND
RIDICULOUS AND STUPID AND TERRIBLE AND
UGLY AND VAIN AND WEIRD AND YUCKY AND
ZILCHY.

GOT ANY MORE NAMES FOR ME? YES

ENTER SOMEONE'S FIRST NAME? GEORGE

GOOFY GEORGE!

AND YOU KNOW WHAT ELSE?

GEORGE IS ALSO ABOMINABLE AND BORING
AND CREEPY AND DUMB AND EVIL AND FREAKY
AND HORRIBLE AND IDIOTIC AND JERKY AND
KLUTZY AND LOUSY AND MEAN AND NASTY
AND ODD AND POMPOUS AND QUARRELSOME
AND RIDICULOUS AND STUPID AND TERRIBLE
AND UGLY AND VAIN AND WEIRD AND YUCKY
AND ZILCHY.

GOT ANY MORE NAMES FOR ME? YES

(continued)

ENTER SOMEONE'S FIRST NAME? LARRY.

LOVING LARRY!

AND YOU KNOW WHAT ELSE?

LARRY IS ALSO ARTICULATE AND BEAUTIFUL
AND CLEVER AND DELIGHTFUL AND
EXCEPTIONAL AND FRIENDLY AND GENEROUS
AND HANDSOME AND INGENIOUS AND JOLLY
AND KIND AND MAGNIFICENT AND NOBLE AND
OUTSTANDING AND PERCEPTIVE AND QUICK
AND RELIABLE AND STUPENDOUS AND TERRIFIC
AND UNSELFISH AND VISIONARY AND
WONDERFUL AND YUMMY AND ZIPPY.

GOT ANY MORE NAMES FOR ME? NO

☐ Program Listing

```
10 REM AMAZE YOUR FRIENDS!
100 PRINT "ENTER SOMEONE'S FIRST NAME";
110 INPUT F$
120 IF LEN(F$)<2 THEN 100
130 I$=MID$(F$,1,1)
140 IF I$<>"X" THEN 200
150 PRINT
160 PRINT "UNLESS THE NAME'S KEXXES--WHICH I DOUBT"
170 PRINT "--YOU'RE TRYING TO FOOL ME. TRY AGAIN."
180 PRINT
190 GOTO 100
200 PRINT CHR$(12)
210 IF MID$(F$,LEN(F$),1)<>"." THEN 240
220 Z=26
230 F$=MID$(F$,1,LEN(F$)-1)
240 F=ASC(MID$(F$,1,LEN(F$)-1))-64
250 FOR C=1 TO F+Z
260 READ B$
270 NEXT C
280 RESTORE
290 PRINT
300 PRINT B$;" ";F$;"!"
310 PRINT
320 PRINT "AND YOU KNOW WHAT ELSE?"
330 PRINT F$;" IS ALSO ";
340 FOR I=1 TO 26+Z
350 READ K$
360 IF Z+1>I THEN 430
370 IF I=24 OR I=50 THEN 430
380 IF I=26+Z THEN 420
```

```

390 IF I=F OR I=F+26 THEN 430
400 PRINT K$;" AND ";
410 GOTO 430
420 PRINT K$;"."
430 NEXT I
440 RESTORE
450 PRINT
460 PRINT "GOT ANY MORE NAMES FOR ME";
470 INPUT Y$
480 IF MID$(Y$,1,1)="N" THEN END
490 Z=0
500 PRINT CHR$(12)
510 GOTO 100
520 DATA ABOMINABLE,BORING,CREEPY,DUMB,EVIL,FREAKY,GOOFY
530 DATA HORRIBLE,IDIOTIC,JERKY,KLUTZY,LOUSY,MEAN
540 DATA NASTY,ODD,POMPOUS,QUARRELSOME,RIDICULOUS,STUPID,
    TERRIBLE
550 DATA UGLY,VAIN,WEIRD,X,YUCKY,ZILCHY
560 DATA ARTICULATE,BEAUTIFUL,CLEVER,DELIGHTFUL,
    EXCEPTIONAL,FRIENDLY
570 DATA GENEROUS,HANDSOME,INGENIOUS,JOLLY,KIND,LOVING,
    MAGNIFICENT
580 DATA NOBLE,OUTSTANDING,PERCEPTIVE,QUICK,RELIABLE,
    STUPENDOUS
590 DATA TERRIFIC,UNSELFISH,VISIONARY,WONDERFUL,X,YUMMY,
    ZIPPY

```

☐ If You Have . . .

APPLE II

Change: 200 HOME
500 HOME

ATARI

Add: 95 DIM F\$(17),I\$(1),B\$(11),K\$(11),Y\$(3)

Change: 130 I\$=F\$(1,1)
200 PRINT CHR\$(125)
210 IF F\$(LEN(F\$),LEN(F\$))<>"." THEN 240
230 F\$=F\$(1,LEN(F\$)-1)
240 F=ASC(F\$(1,LEN(F\$)-1))-64
480 IF Y\$(1,1)="N" THEN END
500 PRINT CHR\$(125)

(continued)

COMMODORE 64

Change: 200 PRINT CHR\$(147)
500 PRINT CHR\$(147)

COMMODORE VIC-20

Change: 100 PRINT "ENTER SOMEONE'S FIRST NAME"
200 PRINT CHR\$(147)
460 PRINT "GOT ANY MORE NAMES FOR ME"
500 PRINT CHR\$(147)

TEXAS INSTRUMENTS 99/4A

Add: 605 END

Change: 130 I\$=SEG\$(F\$,1,1)
200 CALL CLEAR
210 IF SEG\$(F\$,LEN(F\$),1)<>"." THEN 240
230 F\$=SEG\$(F\$,1,LEN(F\$)-1)
240 F=ASC(SEG\$(F\$,1,LEN(F\$)-1))-64
370 IF (I=24)+(I=50) THEN 430
390 IF (I=F)+(I=F+26) THEN 430
480 IF SEG\$(Y\$,1,1)="N" THEN 605
500 CALL CLEAR

TRS-80 COLOR COMPUTER

Change: 200 CLS
500 CLS

Fast-Rising Star

Some stars seem to take forever to get to the top. But maybe you're different.
Run this program and find out!

□ Sample Run

ENTER YOUR WHOLE NAME? ROSE WITHERSPOON

ARE YOU A FAST-RISING STAR? HIT THE ENTER KEY AND FIND OUT?

[illegible]

Break

□ Program Listing

```
10 REM FAST-RISING STAR
100 PRINT "ENTER YOUR WHOLE NAME";
110 INPUT WNAM$
120 IF WNAM$="" THEN 100
130 PRINT
```

(continued)

```

140 PRINT "ARE YOU A FAST-RISING STAR?"
150 PRINT "HIT THE ENTER KEY AND FIND OUT";
160 INPUT D$
170 FOR A=1 TO 25
180 PRINT
190 NEXT A
200 LNAM=LEN(WNAM$)
210 SPOT=(40-LNAM)*RND(1)
220 PRINT TAB(SPOT);WNAM$
240 GOTO 210

```

☐ If You Have . . .

APPLE II

Change: 150 PRINT "HIT THE RETURN KEY AND FIND OUT";

ATARI

```

Add: 95 DIM WNAM$(40),D$(1)
      255 FOR AA=1 TO IDX
      265 PRINT " ";:NEXT AA:RETURN

```

Change: 150 PRINT "HIT THE RETURN KEY AND FIND OUT";
 210 SPOT=(38-LNAM)*RND(0)
 220 IDX=SPOT:GOSUB 255:RETURN

COMMODORE 64

Change: 150 PRINT "HIT THE RETURN KEY AND FIND OUT";

COMMODORE VIC-20

Change: 100 PRINT "ENTER YOUR WHOLE NAME"
 150 PRINT "HIT THE RETURN KEY AND FIND OUT"

TEXAS INSTRUMENTS 99/4A

Change: 210 SPOT=(40-LNAM)*RND

TRS-80 COLOR COMPUTER

Change: 210 SPOT=(32-LNAM)*RND(32)/32

Think You're So Smart?

Sometimes people get stuck on themselves. But sometimes it even happens to computers. Run this program and discover how conceited your computer can be!

☐ Sample Run

```
HMM...I FEEL SOMEONE TAPPING AT MY KEYS
=====
WHAT'S YOUR NAME, KEY-TAPPER? ANNIE OAKLEAF

OKAY, ANNIE OAKLEAF, LET'S SEE YOUR STUFF!

WHAT GRADE ARE YOU IN? 8

WELL, MY GRADE IS 9

HOW FAST CAN YOU TYPE THE ALPHABET?
TRY IT NOW? ACBDEFGHIJKLMNOPQRSTUVWXYZ

NOW HIT THE ENTER KEY AND WATCH ME!?

ABCDEFGHIJKLMNOPQRSTUVWXYZ
ABCDEFGHIJKLMNOPQRSTUVWXYZ
ABCDEFGHIJKLMNOPQRSTUVWXYZ
ABCDEFGHIJKLMNOPQRSTUVWXYZ
ABCDEFGHIJKLMNOPQRSTUVWXYZ

PRETTY FAST, EH? MUCH FASTER THAN YOU!

I DID IT 5 TIMES. YOU DID IT ONLY ONCE.
AND YOU DIDN'T EVEN GET IT RIGHT!

YOU PROBABLY THINK YOU'RE VERY GOOD
WITH NUMBERS. LET'S SEE HOW GOOD...

ENTER A NUMBER? 32

ENTER A DIFFERENT NUMBER? 5

ENTER A DIFFERENT NUMBER? 9

ENTER A DIFFERENT NUMBER? 12

OKAY. NOW MULTIPLY 32 BY 5,
THEN SUBTRACT 9 AND DIVIDE BY 12.

ENTER YOUR ANSWER ALREADY? 12

NOT EVEN CLOSE. THE ANSWER IS 12.58333
```

(continued)

NOW WATCH HOW FAST I CAN CALCULATE!
WHILE I PLUG THE SAME NUMBERS INTO
DIFFERENT PLACES IN THE SAME FORMULA!

HIT MY ENTER KEY WHENEVER YOU'RE READY?

(32 X 5 - 5) / 12 = 12.91667
(32 X 5 - 9) / 12 = 12.58333
(32 X 5 - 12) / 5 = 29.6
(32 X 9 - 5) / 9 = 31.44445
(32 X 9 - 9) / 32 = 8.71875
(32 X 9 - 12) / 12 = 23
(32 X 12 - 32) / 12 = 29.33334
(32 X 12 - 9) / 12 = 31.25
(5 X 32 - 5) / 5 = 31
(5 X 32 - 12) / 32 = 4.625
(5 X 5 - 32) / 12 = -.5833333
(5 X 5 - 12) / 12 = 1.083333
(5 X 9 - 32) / 12 = 1.083333
(5 X 9 - 5) / 12 = 3.333333
(5 X 9 - 12) / 12 = 2.75
(5 X 12 - 32) / 12 = 2.333333
(9 X 32 - 32) / 12 = 21.33334
(9 X 5 - 32) / 12 = 1.083333
(9 X 9 - 32) / 9 = 5.444445
(9 X 12 - 32) / 5 = 15.2
(12 X 32 - 32) / 5 = 70.4
(12 X 5 - 32) / 5 = 5.6
(12 X 9 - 12) / 5 = 19.2
(12 X 12 - 32) / 32 = 3.5
(12 X 12 - 12) / 5 = 26.4

THAT'S DOZENS OF CALCULATIONS IN THE
SAME TIME IT TOOK YOU TO DO JUST ONE.
NOT BAD FOR AN OLD COMPUTER CHIP!

NOW LET'S TRY ONE MORE ALPHABET TRICK.
YOU DO KNOW THE ALPHABET, DON'T YOU? YES

SEE HOW FAST YOU CAN TYPE THE ALPHABET
BACKWARDS, STARTING WITH Z AND ENDING
WITH A? ZYXWVUTSRQPONMLKJIHGFEDCBA

HMMMM. NOW IT'S MY TURN!

ZYXWVUTSRQPONMLKJIHGFEDCBA

MAYBE THAT WAS TOO SLOW FOR YOU.
HIT ENTER AND LET ME TRY AGAIN?

ZYXWVUTSRQPONMLKJIHGFEDCBA

COMPUTERS ARE FUN!

ANYTHING YOU CAN DO I CAN DO BETTER!

☐ Program Listing

```
10 REM THINK YOU'RE SO SMART?
100 DIM T(256)
110 PRINT CHR$(12)
120 PRINT "HMM...I FEEL SOMEONE TAPPING AT MY KEYS"
130 PRINT
140 FOR A=1 TO 40
150 PRINT "=";
160 NEXT A
170 PRINT
180 PRINT
190 PRINT "WHAT'S YOUR NAME, KEY-TAPPER";
200 INPUT NAM$
210 IF NAM$="" THEN 190
220 GOSUB 1470
230 PRINT "OKAY, ";NAM$;", LET'S SEE YOUR STUFF!"
240 GOSUB 1470
250 PRINT "WHAT GRADE ARE YOU IN";
260 INPUT G
270 IF G<1 THEN 250
280 GOSUB 1470
290 PRINT "WELL, MY GRADE IS ";G+1;
300 GOSUB 1470
310 PRINT "HOW FAST CAN YOU TYPE THE ALPHABET?"
320 PRINT "TRY IT NOW";
330 INPUT AL$
340 IF AL$="" THEN 310
350 TEST$="ABCDEFGHIJKLMNOPQRSTUVWXYZ"
360 GOSUB 1470
370 PRINT "NOW HIT THE ENTER KEY AND WATCH ME!";
380 INPUT X$
390 FOR A=1 TO 25
400 PRINT TEST$
410 NEXT A
420 GOSUB 1470
430 PRINT "PRETTY FAST, EH? MUCH FASTER THAN YOU!"
440 GOSUB 1470
450 PRINT "I DID IT 25 TIMES. YOU DID IT ONLY ONCE."
460 IF AL$=TEST$ THEN 480
470 PRINT "AND YOU DIDN'T EVEN GET IT RIGHT!"
480 GOSUB 1470
490 PRINT "YOU PROBABLY THINK YOU'RE VERY GOOD"
500 PRINT "WITH NUMBERS. LET'S SEE HOW GOOD..."
510 PRINT
520 PRINT "ENTER A NUMBER";
530 INPUT N(1)
540 IF N(1)<1 THEN 510
550 PRINT
560 FOR I=2 TO 4
570 PRINT "ENTER A DIFFERENT NUMBER";
580 GOTO 610
590 PRINT "I SAID A DIFFERENT NUMBER!"
```

(continued)

```

600 GOTO 570
610 INPUT N(I)
620 FOR J=I-1 TO 1 STEP -1
630 IF N(J)<1 OR N(J)=N(I) THEN 590
640 NEXT J
650 PRINT
660 NEXT I
670 GOSUB 1470
680 PRINT "OKAY. NOW MULTIPLY ";N(1); " BY "; N(2);", "
690 PRINT "THEN SUBTRACT "; N(3); " AND DIVIDE BY ";
    N(4);"."
700 GOSUB 1470
710 R=(N(1)*N(2)-N(3))/N(4)
720 PRINT "ENTER YOUR ANSWER ALREADY";
730 INPUT AN
740 IF AN<1 THEN 720
750 IF AN=R THEN 810
760 GOSUB 1470
770 PRINT "NOT EVEN CLOSE. THE ANSWER IS ";R
780 GOSUB 1470
790 PRINT "NOW WATCH HOW FAST I CAN CALCULATE!"
800 GOTO 830
810 GOSUB 1470
820 PRINT "WELL, YOU GOT IT RIGHT. BUT WATCH"
830 PRINT "WHILE I PLUG THE SAME NUMBERS INTO"
840 PRINT "DIFFERENT PLACES IN THE SAME FORMULA!"
850 PRINT
860 PRINT "HIT MY ENTER KEY WHENEVER YOU'RE READY";
870 INPUT X$
880 GOSUB 1470
890 K=0
900 FOR L=1 TO 4
910 FOR M=1 TO 4
920 FOR N=1 TO 4
930 FOR P=1 TO 4
940 K=K+1
950 T(K)=(N(L)*N(M)-N(N))/N(P)
960 PRINT "("; N(L); "X";N(M); "-";N(N);") /";N(P);"=";T(K)
970 NEXT P
980 NEXT N
990 NEXT M
1000 NEXT L
1010 GOSUB 1470
1020 PRINT "THAT'S 256 CALCULATIONS IN ABOUT THE"
1030 PRINT "SAME TIME IT TOOK YOU TO DO JUST ONE."
1040 PRINT "NOT BAD FOR AN OLD COMPUTER CHIP!"
1050 GOSUB 1470
1060 PRINT "NOW LET'S TRY ONE MORE ALPHABET TRICK."
1070 PRINT "YOU DO KNOW THE ALPHABET, DON'T YOU";
1080 INPUT X$
1090 IF X$="" THEN 1060
1100 GOSUB 1470
1110 PRINT "SEE HOW FAST YOU CAN TYPE THE ALPHABET"
1120 PRINT "BACKWARDS, STARTING WITH Z AND ENDING"
1130 PRINT "WITH A";

```

```

1140 INPUT X$
1150 IF X$="" THEN 1110
1160 GOSUB 1470
1170 PRINT "HMMM.  NOW IT'S MY TURN!";
1180 GOSUB 1470
1190 LAG=800
1200 L=26
1210 PRINT TAB(7);
1220 GOSUB 1530
1230 GOSUB 1470
1240 PRINT "MAYBE THAT WAS TOO SLOW FOR YOU."
1250 PRINT "HIT ENTER AND LET ME TRY AGAIN";
1260 INPUT X$
1270 GOSUB 1470
1280 LAG=0
1290 PRINT TAB(7);
1300 GOSUB 1530
1310 GOSUB 1470
1320 X$="!NUF5ERA5SRETUPMOC"
1330 Y$="!RETTEB5OD5NAC5I5OD5NAC5UOY5GNIHTYNA"
1340 TEST$=X$
1350 LAG=0
1360 L=LEN(X$)
1370 PRINT TAB(11);
1380 GOSUB 1530
1390 GOSUB 1470
1400 L=LEN(Y$)
1410 TEST$=Y$
1420 PRINT TAB(3);
1430 GOSUB 1530
1440 GOSUB 1470
1450 END
1460 REM ----- SUBROUTINE -----
1470 FOR S=1 TO 4
1480 PRINT
1490 NEXT S
1500 RETURN
1510 REM ----- SECOND SUBROUTINE -----
1520 GOSUB 1470
1530 FOR E=L TO 1 STEP -1
1540 IF MID$(TEST$,E,1)<>"5" THEN 1570
1550 PRINT CHR$(32);
1560 GOTO 1580
1570 PRINT MID$(TEST$,E,1);
1580 FOR Z=1 TO LAG
1590 NEXT Z
1600 NEXT E
1610 RETURN

```

☐ If You Have . . .

APPLE II

Add: 595 PRINT

Change: 110 HOME
370 PRINT "NOW HIT THE RETURN KEY AND WATCH ME!";
450 PRINT "I DID IT 25 TIMES. YOU DID IT ONLY
ONCE.";
860 PRINT "HIT MY RETURN KEY WHENEVER YOU'RE READY";
960 PRINT "("; N(L); " X "; N(M); " - "; N(N); ") / ";
N(P); " = "; T(K)
1250 PRINT "HIT RETURN AND LET ME TRY AGAIN";

ATARI

Add: 95 DIM NAM\$(17),AL\$(30),TEST\$(37),X\$(29),N(4),Y\$(37)
535 N(1)=N1
615 N(I)=N1
1625 FOR AA=1 TO IDX
1635 PRINT " ";:NEXT AA:RETURN

Change: 110 PRINT CHR\$(125)
140 FOR A=1 TO 38
370 PRINT "NOW HIT THE RETURN KEY AND WATCH ME!"
530 INPUT N1
610 INPUT N1
860 PRINT "HIT MY RETURN KEY WHENEVER YOU'RE READY";
1210 IDX=7:GOSUB 1625
1250 PRINT "HIT RETURN AND LET ME TRY AGAIN";
1290 IDX=7:GOSUB 1625
1370 IDX=11:GOSUB 1625
1420 IDX=2:GOSUB 1625
1540 IF TEST\$(E,E)<>"5" THEN 1570
1570 PRINT TEST\$(E,E);

COMMODORE 64

Change: 110 PRINT CHR\$(147)
370 PRINT "NOW HIT THE RETURN KEY AND WATCH ME!";
860 PRINT "HIT MY RETURN KEY WHENEVER YOU'RE READY";
1250 PRINT "HIT RETURN AND LET ME TRY AGAIN";

COMMODORE VIC-20

Delete: 100,130,170,180,510,550,650,850,1210,1290,1370,
1420,1460,1510

Change: 110 PRINT CHR\$(147)
140 FOR A=1 TO 22
190 PRINT "WHAT'S YOUR NAME, KEY-TAPPER"
250 PRINT "WHAT GRADE ARE YOU IN"
320 PRINT "TRY IT NOW"
370 PRINT "NOW HIT THE RETURN KEY AND WATCH ME!"
520 PRINT "ENTER A NUMBER"
570 PRINT "ENTER A DIFFERENT NUMBER"
720 PRINT "ENTER YOUR ANSWER ALREADY"
860 PRINT "HIT MY RETURN KEY WHENEVER YOU'RE READY"
950 T=(N(L)*N(M)-N(N))/N(P)
960 PRINT "(";N(L);"X";N(M);"-";N(N);")/";N(P);
 "=";T
1070 PRINT "YOU DO KNOW THE ALPHABET, DON'T YOU"
1130 PRINT "WITH A"
1250 PRINT "HIT RETURN AND LET ME TRY AGAIN"

TEXAS INSTRUMENTS 99/4A

Change: 110 CALL CLEAR
140 FOR A=1 TO 28
310 PRINT "HOW FAST CAN YOU TYPE THE ALPHABET?"
370 PRINT "NOW HIT THE ENTER KEY AND WATCH ME!";
430 PRINT "PRETTY FAST, EH? MUCH FASTER THAN YOU!"
450 PRINT "I DID IT 25 TIMES. YOU DID IT ONLY ONCE."
470 PRINT "AND YOU DIDN'T EVEN GET IT RIGHT!"
490 PRINT "YOU PROBABLY THINK YOU'RE VERY GOOD"
630 IF (N(J)<1)+(N(J)=N(I)) THEN 590
860 PRINT "HIT MY ENTER KEY WHENEVER YOU'RE READY";
920 FOR Z=1 TO 4
950 T(K)=(N(L)*N(M)-N(Z))/N(P)
960 PRINT "(";N(L);"X";N(M);"-";N(Z);") /";N(P);
 "=";T(K)
980 NEXT Z
1070 PRINT "YOU DO KNOW THE ALPHABET, DON'T YOU";
1190 LAG=500
1250 PRINT "HIT ENTER AND LET ME TRY AGAIN";
1540 IF SEG\$(TEST\$,E,1)<>"5" THEN 1570
1570 PRINT SEG\$(TEST\$,E,1);

TRS-80 COLOR COMPUTER

Change: 110 CLS
140 FOR A=1 TO 31
1210 PRINT TAB(2);
1290 PRINT TAB(2);
1370 PRINT TAB(6);

Report Card Fun

How come kids always get report cards instead of giving them out?

Now you can change that forever!

□ Sample Run

IT'S REPORT CARD DAY.

**BUT THIS TIME YOU GIVE OUT THE REPORT
CARD INSTEAD OF GETTING IT!**

WHAT'S YOUR FIRST NAME? HANK HENRIES

ENTER THE FULL NAME OF THE PERSON WHO
WILL GET THE REPORT CARD? HORACE HENRIES

DO YOU WANT TWO PEOPLE TO GET THIS
REPORT CARD? YES

ENTER THE SECOND PERSON'S FULL NAME? HILDA
HENRIES

WHAT COMPUTER ARE YOU USING? HILLHUGGER-20

DO NOT USE COMMAS IN YOUR NEXT ANSWER:
WHAT IS THE DATE TODAY? 10-10-86

WHAT GRADE DO YOU WANT
HORACE HENRIES AND HILDA HENRIES TO BE IN? 5TH

GOOD...

FOR THE NEXT QUESTIONS, GIVE HORACE HENRIES
AND HILDA HENRIES
THE CORRECT GRADES (FROM A TO F ONLY)

GIVING OUT ALLOWANCE	? B
KEEPING PROMISES	? A
HELPING WITH HOMEWORK	? C
SPENDING ENOUGH TIME WITH HANK	? B
BEING THERE WHEN IT COUNTS	? A
UNDERSTANDING HANK HENRIES	? C
TREATING HANK HENRIES LIKE AN ADULT	? B
BUYING COMPUTER SUPPLIES	? D
NOT BEING TOO STRICT	? C

OKAY. NOW WHILE THE COMPUTER GETS THE
REPORT CARD FOR HORACE HENRIES AND HILDA
HENRIES READY,
GO BRING HORACE HENRIES AND HILDA HENRIES
OVER TO YOUR HILLHUGGER-20.

BUT FIRST, HIT ENTER TO HIDE ALL THIS?

ATTENTION: HORACE HENRIES AND HILDA HENRIES
HIT ENTER TO SEE SOMETHING IMPORTANT?

HILLHUGGER-20 UNIVERSITY
REPORT CARD FOR
HORACE HENRIES AND HILDA HENRIES

GRADE 5TH DATE:10-10-86

GIVING OUT ALLOWANCE	B
KEEPING PROMISES	A
HELPING WITH HOMEWORK	C
SPENDING ENOUGH TIME WITH HANK	B
BEING THERE WHEN IT COUNTS	A
UNDERSTANDING HANK HENRIES	C
TREATING HANK HENRIES LIKE AN ADULT	B
BUYING COMPUTER SUPPLIES	D
NOT BEING TOO STRICT	C

YOUR GRADE-POINT AVERAGE IS 2.666667

HAVE THIS REPORT CARD SIGNED BY HANK

☐ Program Listing

```
10 REM REPORT CARD FUN
100 GOSUB 1150
110 PRINT TAB(10);"IT'S REPORT CARD DAY."
120 GOSUB 1150
130 PRINT
140 PRINT "BUT THIS TIME YOU GIVE OUT THE REPORT"
150 PRINT "CARD INSTEAD OF GETTING IT!"
160 PRINT
```

(continued)

```

170 PRINT "WHAT'S YOUR FIRST NAME";
180 INPUT NAM$
190 IF NAM$="" THEN 170
200 PRINT
210 PRINT "ENTER THE FULL NAME OF THE PERSON WHO"
220 PRINT "WILL GET THE REPORT CARD";
230 INPUT PN$
240 IF PN$="" THEN 210
250 PRINT
260 PRINT "DO YOU WANT TWO PEOPLE TO GET THIS"
270 PRINT "REPORT CARD";
280 INPUT Y$
290 IF Y$="" THEN 260
300 PRINT
310 IF MID$(Y$,1,1)<>"Y" THEN 360
320 PRINT "ENTER THE SECOND PERSON'S FULL NAME";
330 INPUT T$
340 IF T$="" THEN 320
350 PN$=PN$+" AND "+T$
360 PRINT
370 PRINT "WHAT COMPUTER ARE YOU USING";
380 INPUT C$
390 IF C$="" THEN 370
400 PRINT
410 PRINT "DO NOT USE COMMAS IN YOUR NEXT ANSWER:"
420 PRINT "WHAT IS THE DATE TODAY";
430 INPUT D$
440 IF D$="" THEN 410
450 PRINT
460 PRINT "WHAT GRADE DO YOU WANT "
470 PRINT PN$;" TO BE IN";
480 INPUT G$
490 IF G$="" THEN 460
500 FOR D=1 TO 25
510 PRINT
520 NEXT D
530 PRINT "GOOD..."
540 FOR D=1 TO 11
550 PRINT
560 NEXT D
570 PRINT "FOR THE NEXT QUESTIONS, GIVE ";PN$
580 PRINT "THE CORRECT GRADES (FROM A TO F ONLY)"
590 GOSUB 1150
600 FOR A=1 TO 9
610 READ A$
620 IF A=4 OR A=6 OR A=7 THEN A$=A$+" "+NAM$
630 IF A=7 THEN READ R$
640 IF A=7 THEN A$=A$+" "+R$
650 PRINT A$;TAB(35);
660 INPUT S$
670 IF S$="" THEN 650
680 IF ASC(S$)<65 THEN 650
690 IF ASC(S$)>70 THEN 650
700 G(A)=ASC(S$)-64
710 NEXT A

```

```

720 PRINT
730 PRINT "OKAY. NOW WHILE THE COMPUTER GETS THE"
740 PRINT "REPORT CARD FOR ";PN$;" READY,"
750 PRINT "GO BRING ";PN$;" OVER TO YOUR ";C$;". "
760 PRINT
770 PRINT "BUT FIRST, HIT ENTER TO HIDE ALL THIS";
780 INPUT X$
790 FOR A=1 TO 25
800 PRINT
810 NEXT A
820 PRINT "ATTENTION: ";PN$
830 PRINT "HIT ENTER TO SEE SOMETHING IMPORTANT";
840 INPUT X$
850 PRINT
860 GOSUB 1150
870 SCH$=C$+" UNIVERSITY"
880 PRINT TAB(20-LEN(SCH$)/2);SCH$
890 PRINT TAB(13);"REPORT CARD FOR"
900 PRINT TAB(20-LEN(PN$)/2);PN$
910 PRINT
920 GOSUB 1150
930 PRINT "GRADE ";G$;TAB(20);"DATE: ";D$
940 GOSUB 1150
950 GPA=0
960 RESTORE
970 FOR A=1 TO 9
980 J=5-G(A)
990 IF J<0 THEN J=0
1000 GPA=GPA+J
1010 READ A$
1020 IF A=4 OR A=6 OR A=7 THEN A$=A$+" "+NAM$
1030 IF A=7 THEN READ R$
1040 IF A=7 THEN A$=A$+" "+R$
1050 S$=CHR$(G(A)+64)
1060 PRINT A$;TAB(37);S$
1070 NEXT A
1080 GPA=GPA/9
1090 GOSUB 1150
1100 PRINT "YOUR GRADE-POINT AVERAGE IS";GPA
1110 GOSUB 1150
1120 PRINT "HAVE THIS REPORT CARD SIGNED BY ";NAM$
1130 END
1140 REM --- LINE SUBROUTINE ---
1150 FOR L=1 TO 38
1160 PRINT "-";
1170 NEXT L
1180 PRINT "-"
1190 RETURN
1200 DATA GIVING OUT ALLOWANCE
1210 DATA KEEPING PROMISES
1220 DATA HELPING WITH HOMEWORK
1230 DATA SPENDING ENOUGH TIME WITH
1240 DATA BEING THERE WHEN IT COUNTS
1250 DATA UNDERSTANDING

```

(continued)

```

1260 DATA TREATING,LIKE AN ADULT
1270 DATA BUYING COMPUTER SUPPLIES
1280 DATA NOT BEING TOO STRICT

```

☐ If You Have . . .

APPLE II

Change: 770 PRINT "BUT FIRST, HIT RETURN TO HIDE ALL THIS";
 830 PRINT "HIT RETURN TO SEE SOMETHING IMPORTANT";

ATARI

Delete: 350,620,1020,1040

Add: 91 DIM NAM\$(40),PN\$(40),Y\$(3),T\$(40),C\$(40),D\$(11)
 92 DIM G\$(7),A\$(29),R\$(17),S\$(2),X\$(1),G(9),SCH\$(11)
 651 IDX=35:GOSUB 1295
 652 GOTO 660
 655 PRINT A\$;" ";NAM\$;" ";R\$;
 656 GOTO 651
 901 PRINT PN\$;
 902 IF LEN(T\$)>2 THEN PRINT " AND " ;T\$;
 903 PRINT
 975 R\$=" "
 1055 IF A=4 OR A=6 OR A=7 THEN 1065
 1061 IDX=5:GOSUB 1295:PRINT S\$
 1062 GOTO 1070
 1065 PRINT A\$;" ";NAM\$;" ";R\$;
 1066 GOTO 1061
 1285 END
 1295 FOR AA=1 TO IDX
 1305 PRINT " ";:NEXT AA:RETURN

Change: 110 IDX=10:GOSUB 1295:PRINT "IT'S REPORT CARD DAY."
 310 IF Y\$(1,1)<>"Y" THEN 360
 640 IF A=4 OR A=6 OR A=7 THEN 655
 650 PRINT A\$;
 770 PRINT "BUT FIRST, HIT RETURN TO HIDE ALL THIS";
 830 PRINT "HIT RETURN TO SEE SOMETHING IMPORTANT";
 870 SCH\$=" UNIVERSITY"
 880 IDX=(20-(LEN(SCH\$)+LEN(C\$))/2):GOSUB 1295:PRINT
 C\$;SCH\$
 890 IDX=13:GOSUB 1295:PRINT "REPORT CARD FOR"
 900 IDX=(20-(LEN(PN\$)+LEN(T\$)+5)/2):GOSUB 1295
 930 PRINT "GRADE ";G\$;:IDX=17:GOSUB 1295:PRINT
 "DATE: ";D\$
 1060 PRINT A\$;
 1100 PRINT "YOUR GRADE-POINT AVERAGE IS ";GPA
 1150 FOR L=1 TO 36

COMMODORE 64

Change: 770 PRINT "BUT FIRST, HIT RETURN TO HIDE ALL THIS";
830 PRINT "HIT RETURN TO SEE SOMETHING IMPORTANT";

COMMODORE VIC-20

Add: 944 PRINT
945 PRINT "HIT RETURN TO SEE"
946 INPUT X\$
1084 PRINT
1085 PRINT "HIT RETURN TO CONTINUE"
1086 INPUT X\$

Change: 170 PRINT "WHAT'S YOUR FIRST NAME"
220 PRINT "WILL GET THE REPORT CARD"
270 PRINT "REPORT CARD"
320 PRINT "ENTER THE SECOND PERSON'S FULL NAME"
370 PRINT "WHAT COMPUTER ARE YOU USING"
420 PRINT "WHAT IS THE DATE TODAY"
470 PRINT PN\$;" TO BE IN"
650 PRINT A\$
770 PRINT "BUT FIRST, HIT RETURN TO HIDE ALL THIS"
830 PRINT "HIT RETURN TO SEE SOMETHING IMPORTANT"
880 PRINT TAB(11-LEN(SCH\$)/2);SCH\$
890 PRINT TAB(3);"REPORT CARD FOR"
900 PRINT TAB(11-LEN(PN\$)/2);PN\$
930 PRINT "GRADE ";G\$;TAB(9);"DATE: ";D\$
1060 PRINT A\$;" ";S\$
1150 FOR L=1 TO 21

TEXAS INSTRUMENTS 99/4A

Delete: 640,1040

Add: 622 A\$=A\$&" "&NAM\$
632 READ R\$
634 A\$=A\$&" "&R\$
992 J=0
1022 A\$=A\$&" "&NAM\$
1032 READ R\$
1034 A\$=A\$&" "&R\$

Change: 140 PRINT "BUT THIS TIME YOU GIVE OUT THE REPORT"
210 PRINT "ENTER THE FULL NAME OF THE PERSON WHO"
260 PRINT "DO YOU WANT TWO PEOPLE TO GET THIS"
310 IF SEG\$(Y\$,1,1)<>"Y" THEN 360
320 PRINT "ENTER THE SECOND PERSON'S FULL NAME";
350 PN\$=PN\$&" AND "&T\$

(continued)

```

410 PRINT "DO NOT USE COMMAS IN YOUR      NEXT ANSWER:"
620 IF (A<>4)*(A<>6)*(A<>7) THEN 630
630 IF A<>7 THEN 650
650 PRINT A$;TAB(23);
730 PRINT "OKAY. NOW WHILE THE COMPUTER GETS THE"
830 PRINT "HIT ENTER TO SEE SOMETHING  IMPORTANT";
870 SCH$=C$&" UNIVERSITY"
880 PRINT TAB(13-LEN(SCH$)/2);SCH$
890 PRINT TAB(6);"REPORT CARD FOR"
900 PRINT TAB(13-LEN(PN$)/2);PN$
1020 IF (A<>4)*(A<>6)*(A<>7) THEN 1030
1030 IF A<>7 THEN 1050
1060 PRINT A$;TAB(25);S$
1150 FOR L=1 TO 27

```

TRS-80 COLOR COMPUTER

Add:

```

944 PRINT
945 PRINT "HIT ENTER TO SEE"
946 INPUT X$
1085 PRINT
1086 PRINT "HIT ENTER TO CONTINUE"
1087 INPUT X$

```

Change:

```

880 PRINT TAB(15-LEN(SCH$)/2);SCH$
890 PRINT TAB(8);"REPORT CARD FOR"
900 PRINT TAB(15-LEN(PN$)/2);PN$
930 PRINT "GRADE ";G$;TAB(18);"DATE: ";D$
1060 PRINT A$;TAB(30);S$
1150 FOR L=1 TO 31

```

Uphill Racer

How fast can you climb a hill? Your computer and this program may make you the master of the Alps!

☐ Sample Run

ENTER YOUR WHOLE NAME? BRUTUS PALOOKA

IT'S EASY TO RACE DOWN A MOUNTAIN VERY QUICKLY. THE TRICK IS TO RACE UP JUST AS FAST. HIT THE ENTER KEY TO SEE?

B
R
U
T
U
S

P
A
L
O
O
K
A

B
R
U
T
U
S

P
A
L
O
O
K
A

(continued)

B
R
U
T
U
S

P
A
L
O
O
K
A

Break

☐ Program Listing

```
10 REM UPHILL RACER
100 PRINT "ENTER YOUR WHOLE NAME";
110 INPUT WNAME$
120 LNAME=LEN(WNAME$)
130 IF LNAME=0 THEN 100
140 PRINT CHR$(12)
150 PRINT "IT'S EASY TO RACE DOWN A MOUNTAIN VERY"
160 PRINT "QUICKLY. THE TRICK IS TO RACE UP JUST"
170 PRINT "AS FAST. HIT THE ENTER KEY TO SEE";
180 INPUT D$
190 FOR A=1 TO 25
200 PRINT
210 NEXT A
220 FOR B=1 TO LNAME
230 PRINT TAB(2*B);MID$(WNAME$,B,1)
240 NEXT B
250 PRINT
260 GOTO 220
```

☐ If You Have . . .

APPLE II

Change: 140 HOME
170 PRINT "AS FAST. HIT THE RETURN KEY TO SEE";

ATARI

Add: 95 DIM WNAME\$(40),D\$(1)
275 FOR AA=1 TO IDX
285 PRINT " ";:NEXT AA:RETURN

Change: 140 PRINT CHR\$(125)
170 PRINT "AS FAST. HIT THE RETURN KEY TO SEE";
230 IDX=(2*B):GOSUB 275:PRINT WNAME\$(B,B)

COMMODORE 64

Change: 140 PRINT CHR\$(147)
170 PRINT "AS FAST. HIT THE RETURN KEY TO SEE";

COMMODORE VIC-20

Change: 100 PRINT "ENTER YOUR WHOLE NAME"
140 PRINT CHR\$(147)
170 PRINT "AS FAST. HIT THE RETURN KEY TO SEE"

TEXAS INSTRUMENTS 99/4A

Change: 140 CALL CLEAR
150 PRINT "IT'S EASY TO RACE DOWN A MOUNTAIN
VERY QUICKLY."
160 PRINT "THE TRICK IS TO RACE UP JUST"
230 PRINT TAB(2*B);SEG\$(WNAME\$,B,1)

TRS-80 COLOR COMPUTER

Change: 140 CLS

Funny Friends

Your computer knows a lot of people you've never met. If you're nice to it, maybe it'll introduce you to them!

☐ Sample Run

WHAT'S YOUR FIRST NAME? LINDA

LINDA? YOU CALL THAT UNUSUAL?

WE HAVE SOME FUNNY FRIENDS NAMED
BINDA AND CINDA AND DINDA AND FINDA AND
GINDA AND HINDA AND JINDA AND KINDA AND
MINDA AND NINDA AND PINDA AND QINDA AND
RINDA AND SINDA AND TINDA AND VINDA AND
WINDA AND XINDA AND YINDA AND ZINDA AND
BLINDA AND BRINDA AND CHINDA AND CLINDA
AND CRINDA AND FLINDA AND FRINDA AND
GRINDA AND SHINDA AND SLINDA AND STINDA
AND THINDA AND TRINDA AND EVEN STRINDA!

WANT TO TRY ANY OTHER NAMES? YES

WHAT'S YOUR FIRST NAME? FRANK

FRANK? YOU CALL THAT UNUSUAL?

WE HAVE SOME FUNNY FRIENDS NAMED
BANK AND CANK AND DANK AND FANK AND
GANK AND HANK AND JANK AND KANK AND
LANK AND MANK AND NANK AND PANK AND
QANK AND RANK AND SANK AND TANK AND
VANK AND WANK AND XANK AND YANK AND
ZANK AND BLANK AND BRANK AND CHANK
AND CLANK AND CRANK AND FLANK AND
GRANK AND SHANK AND SLANK AND STANK
AND THANK AND TRANK AND EVEN STRANK!

WANT TO TRY ANY OTHER NAMES? NO

□ Program Listing

```
10 REM FUNNY FRIENDS
100 A$="A"
110 E$="E"
120 I$="I"
130 O$="O"
140 U$="U"
150 PRINT "WHAT'S YOUR FIRST NAME";
160 INPUT NAM$
170 IF LEN(NAM$)<1 THEN 150
180 PRINT
190 PRINT NAM$;"? YOU CALL THAT UNUSUAL?"
191 PRINT
200 PRINT "WE HAVE SOME FUNNY FRIENDS NAMED"
210 GOSUB 440
220 FOR A=1 TO 26
230 B$=CHR$(A+64)
240 IF B$=A$ OR B$=E$ OR B$=I$ OR B$=O$ OR B$=U$ OR B$=P$
    THEN 260
250 PRINT B$+REST$;" "; "AND"; " ";
260 NEXT A
270 FOR I=1 TO 14
280 READ B$
290 IF B$=P$ THEN 340
300 IF I=14 THEN 330
310 PRINT B$+REST$;" "; "AND"; " ";
320 GOTO 340
330 PRINT "EVEN ";B$+REST$;"!"
340 NEXT I
350 PRINT
360 PRINT "WANT TO TRY ANY OTHER NAMES";
370 INPUT NAM$
380 IF MID$(NAM$,1,1)="N" THEN END
390 PRINT
400 RESTORE
410 GOTO 150
420 END
430 REM --- VOWEL CHECKER ---
440 FOR I=1 TO LEN(NAM$)
450 F$=MID$(NAM$,I,1)
460 IF F$=A$ OR F$=E$ OR F$=I$ OR F$=O$ OR F$=U$ THEN 480
470 NEXT I
480 REST$=MID$(NAM$,I,LEN(NAM$))
490 P$=MID$(NAM$,1,I-1)
500 RETURN
510 DATA BL,BR,CH,CL,CR,FL,FR,GR,SH,SL,ST,TH,TR,STR
```

☐ If You Have . . .

APPLE II No Changes Required

ATARI

Add: 91 DIM A\$(1),E\$(1),I\$(1),O\$(1),B\$(3),U\$(1)
92 DIM NAM\$(40),P\$(40),F\$(40),REST\$(40)

Change: 250 PRINT B\$;REST\$;" "; "AND"; " ";
310 PRINT B\$;REST\$;" "; "AND"; " ";
330 PRINT "EVEN ";B\$;REST\$;"!"
380 IF NAM\$(1,1)="N" THEN END
450 F\$=NAM\$(I,I)
480 REST\$=NAM\$(1,LEN(NAM\$))
490 P\$=NAM\$(1,I-1)

COMMODORE 64 No Changes Required

COMMODORE VIC-20

Add: 205 PRINT
206 PRINT "(HIT RETURN TO SEE)"
207 INPUT X\$

Change: 150 PRINT "WHAT'S YOUR FIRST NAME"
360 PRINT "WANT TO TRY ANY OTHER NAMES"

TEXAS INSTRUMENTS 99/4A

Add: 525 END

Change: 200 PRINT "WE HAVE SOME FUNNY FRIENDS NAMED"
240 IF (B\$=A\$)+(B\$=E\$)+(B\$=I\$)+(B\$=O\$)+(B\$=U\$)+(B\$=P\$) THEN 260
250 PRINT B\$&REST\$;" "; "AND"; " ";
310 PRINT B\$&REST\$;" "; "AND"; " ";
330 PRINT "EVEN ";B\$&REST\$;"!"
380 IF SEG\$(NAM\$,1,1)="N" THEN 520
450 F\$=SEG\$(NAM\$,I,1)
480 REST\$=SEG\$(NAM\$,I,LEN(NAM\$))
490 P\$=SEG\$(NAM\$,1,I-1)

TRS-80 COLOR COMPUTER

Add: 205 PRINT
206 PRINT "HIT ENTER TO SEE"
207 INPUT X\$

A Winner Every Time

Step right up! Press the ENTER key! With this program,
you're bound to be a winner!

☐ Sample Run

HERE WE ARE AT THE WORLD'S SMARTEST
COMPUTERIZED SLOT MACHINE. BUT WITH
THIS ONE YOU WIN EVERY TIME. HIT
THE ENTER KEY AND WATCH WHAT IT DOES?

WHAT'S YOUR FIRST NAME? RALPH

```
* <7><7><7><R><A><L><7><7><7><7><7><7>*
* <8><8><8><R><A><L><8><8><8><8><8><8>*
* <9><9><9><R><A><L><P><9><9><9><9><9>*
* <:><:><:><R><A><L><P><:><:><:><:><:>*
* <:><:><:><R><A><L><P><:><:><:><:><:>*
* <=><=><=><R><A><L><P><=><=><=><=><=>*
* <?><?><?><R><A><L><P><?><?><?><?><?>*
* <@><@><@><R><A><L><P><@><@><@><@><@>*
* <A><A><A><R><A><L><P><A><A><A><A><A>*
* <B><B><B><R><A><L><P><B><B><B><B><B>*
* <C><C><C><R><A><L><P><C><C><C><C><C>*
* <D><D><D><R><A><L><P><D><D><D><D><D>*
* <E><E><E><R><A><L><P><E><E><E><E><E>*
* <F><F><F><R><A><L><P><F><F><F><F><F>*
* <G><G><G><R><A><L><P><G><G><G><G><G>*
* <H><H><H><R><A><L><P><H><H><H><H><H>*
* <I><I><I><R><A><L><P><I><I><I><I><I>*
* <J><J><J><R><A><L><P><J><J><J><J><J>*
* <K><K><K><R><A><L><P><K><K><K><K><K>*
* <L><L><L><R><A><L><P><L><L><L><L><L>*
* <M><M><M><R><A><L><P><M><M><M><M><M>*
* <N><N><N><R><A><L><P><N><N><N><N><N>*
* <O><O><O><R><A><L><P><O><O><O><O><O>*
* <P><P><P><R><A><L><P><P><P><P><P><P>*
* <Q><Q><Q><R><A><L><P><Q><Q><Q><Q><Q>*
* <R><R><R><R><A><L><P><H><R><R><R><R>*
* <S><S><S><R><A><L><P><H><S><S><S><S>*
* <T><T><T><R><A><L><P><H><T><T><T><T>*
* <U><U><U><R><A><L><P><H><U><U><U><U>*
* <V><V><V><R><A><L><P><H><V><V><V><V>*
* <W><W><W><R><A><L><P><H><W><W><W><W>*
* <X><X><X><R><A><L><P><H><X><X><X><X>*
* <Y><Y><Y><R><A><L><P><H><Y><Y><Y><Y>*
* <Z><Z><Z><R><A><L><P><H><Z><Z><Z><Z>*
* <5><5><5><R><A><L><P><H><5><5><5><5>*
```

Break

□ Program Listing

```
10 REM A WINNER EVERY TIME
100 PRINT "HERE WE ARE AT THE WORLD'S SMARTEST"
110 PRINT "COMPUTERIZED SLOT MACHINE. BUT WITH"
120 PRINT "THIS ONE YOU WIN EVERY TIME. HIT"
130 PRINT "THE ENTER KEY AND WATCH WHAT IT DOES";
140 INPUT D$
150 DIM N(12)
160 PRINT
170 PRINT "WHAT'S YOUR FIRST NAME";
180 INPUT NAM$
190 L=LEN(NAM$)
200 IF L<2 THEN 170
210 IF L<11 THEN 240
220 PRINT "--NAME MUST BE 10 OR FEWER LETTERS--"
230 GOTO 170
240 PRINT CHR$(12)
270 START=6-L/2
280 IF L=3 OR L=5 OR L=7 OR L=9 THEN 310
290 ODDEVEN=12
300 GOTO 320
310 ODDEVEN=13
320 K=48
330 PRINT TAB(2);"*";
340 FOR A=1 TO START
350 A$="<"+CHR$(K)+">"
360 PRINT A$;
370 NEXT A
380 FOR Z=1 TO J
390 PRINT "<";CHR$(N(Z));">";
400 NEXT Z
410 FOR A=J+1 TO ODDEVEN-START
420 A$="<"+CHR$(K)+">"
430 PRINT A$;
440 NEXT A
450 PRINT "*"
460 K=K+1
470 IF K<48 THEN 460
480 IF K>90 THEN K=48
490 L=L+1
500 IF L<>25 THEN 530
510 J=J+1
520 L=0
530 IF J>LEN(NAM$) THEN 580
540 FOR B=1 TO J
550 N(B)=ASC(MID$(NAM$,B,1))
560 NEXT B
570 GOTO 330
580 J=L
590 GOTO 330
```

☐ If You Have . . .

APPLE II

Add: 375 IF J=0 THEN 410

Change: 130 PRINT "THE RETURN KEY AND WATCH WHAT IT DOES";
240 HOME

ATARI

Add: 95 DIM NAM\$(11),D\$(1),A\$(3)
435 IF J+1=ODDEVEN-START THEN PRINT "<";A\$;">";

Change: 130 PRINT "THE RETURN KEY AND WATCH WHAT IT DOES";
240 PRINT CHR\$(125)
290 ODDEVEN=10
310 ODDEVEN=11
330 PRINT " ";"*";
350 A\$=CHR\$(K)
360 PRINT "<";A\$;">";
380 FOR Z=0 TO J
420 A\$=CHR\$(K)
430 PRINT "<";A\$;">";
550 N(B)=ASC(NAM\$(B,B))

COMMODORE 64

Change: 130 PRINT "THE RETURN KEY AND WATCH WHAT IT DOES";
240 PRINT CHR\$(147)
270 ART=6-L/2
330 PRINT "*";
340 FOR A=1 TO ART
380 FOR Z=0 TO J
410 FOR A=J+1 TO ODDEVEN-ART

COMMODORE VIC-20

Change: 130 PRINT "THE RETURN KEY AND WATCH WHAT IT DOES"
200 PRINT "NAME MUST BE 4 OR FEWER LETTERS--"
210 IF L<5 THEN 240
240 PRINT CHR\$(147)
270 ART=6-L/2
280 IF L=3 THEN 310
290 ODDEVEN=6
310 ODDEVEN=7
330 PRINT "*";
340 FOR A=1 TO ART
380 FOR Z=0 TO J
410 FOR A=J+1 TO ODDEVEN-ART

(continued)

TEXAS INSTRUMENTS 99/4A

Add: 485 K=48

Change: 100 PRINT "HERE WE ARE AT THE WORLD'S SMARTEST
COMPUTERIZED"
110 PRINT "SLOT MACHINE. BUT WITH"
220 PRINT "--NAME MUST BE 10 OR FEWER LETTERS--"
240 CALL CLEAR
280 IF (L=3)+(L=5)+(L=7)+(L=9) THEN 310
350 A\$=" "&CHR\$(K)&">"
420 A\$=" "&CHR\$(K)&">"
480 IF K<=90 THEN 490
550 N(B)=ASC(SEG\$(NAM\$,B,1))

TRS-80 COLOR COMPUTER

Add: 445 IF J=0 THEN PRINT A\$;

Change: 210 IF L<7 THEN 240
220 PRINT "--NAME MUST BE 6 OR FEWER LETTERS--"
240 CLS
290 ODDEVEN=8
310 ODDEVEN=9
330 PRINT "*";
410 FOR A=J+1 TO ODDEVEN-START-1

Computer Charmer

At last! A program for lazy people — folks who'd rather let others do their work.
Just type this in, and your most hated chores won't be a problem anymore.

☐ Sample Run

WHAT'S YOUR NAME? GARY

ENTER THREE JOBS YOU HAVE TO DO
INSIDE YOUR HOUSE THAT YOU HATE TO DO:
1:? CLEANING UP THE BASEMENT
2:? SWEEPING THE FLOORS
3:? MAKING THE BEDS

NOW ENTER TWO JOBS YOU HAVE TO DO
OUTSIDE YOUR HOUSE THAT YOU HATE TO DO:
1:? WASHING THE CAR
2:? MOWING THE LAWN

ENTER THE NAMES OF FIVE PEOPLE YOU'D
LIKE TO DO ALL YOUR WORK FOR YOU:
1:? UNCLE DAVE
2:? BARNEY NEWFUSS
3:? AUNT SALLY
4:? DAD
5:? MOM

OKAY. NOW HIT THE ENTER KEY AND SAY
GOODBYE TO ALL THAT WORK FOREVER!?

FOREVER IN THE FUTURE, AUNT SALLY
WILL BE HANDLING THE FOLLOWING JOB
FOR GARY: MOWING THE LAWN

FOREVER IN THE FUTURE, MOM
WILL BE HANDLING THE FOLLOWING JOB
FOR GARY: WASHING THE CAR

FOREVER IN THE FUTURE, UNCLE DAVE
WILL BE HANDLING THE FOLLOWING JOB
FOR GARY: MAKING THE BEDS

FOREVER IN THE FUTURE, DAD
WILL BE HANDLING THE FOLLOWING JOB
FOR GARY: SWEEPING THE FLOORS

(continued)

FOREVER IN THE FUTURE, BARNEY NEWFUSS
WILL BE HANDLING THE FOLLOWING JOB
FOR GARY: CLEANING UP THE BASEMENT

(ALL IT TAKES IS A LITTLE CHARM!)

☐ Program Listing

```
10 REM COMPUTER CHARMER
100 PRINT "WHAT'S YOUR NAME";
110 INPUT NAM$
111 IF NAM$="" THEN 100
120 PRINT
130 PRINT "ENTER THREE JOBS YOU HAVE TO DO"
140 PRINT "INSIDE YOUR HOUSE THAT YOU HATE TO DO:"
150 PRINT "1:";
160 INPUT A$
170 IF A$="" THEN 150
180 PRINT "2:";
190 INPUT B$
200 IF B$="" THEN 180
210 PRINT "3:";
220 INPUT C$
230 IF C$="" THEN 210
240 PRINT
250 PRINT "NOW ENTER TWO JOBS YOU HAVE TO DO"
260 PRINT "OUTSIDE YOUR HOUSE THAT YOU HATE TO DO:"
270 PRINT "1:";
280 INPUT D$
290 IF D$="" THEN 270
300 PRINT "2:";
310 INPUT E$
320 IF E$="" THEN 300
330 PRINT
340 PRINT "ENTER THE NAMES OF FIVE PEOPLE YOU'D"
350 PRINT "LIKE TO DO ALL YOUR WORK FOR YOU:"
360 PRINT "1:";
370 INPUT JA$
380 IF JA$="" THEN 360
390 PRINT "2:";
400 INPUT JB$
410 IF JB$="" THEN 390
420 PRINT "3:";
430 INPUT JC$
440 IF JC$="" THEN 420
450 PRINT "4:";
460 INPUT JD$
470 IF JD$="" THEN 450
480 PRINT "5:";
490 INPUT JE$
500 IF JE$="" THEN 480
```

```

510 PRINT
520 PRINT "OKAY.  NOW HIT THE ENTER KEY AND SAY"
530 PRINT "GOODBYE TO ALL THAT WORK FOREVER!";
540 INPUT X$
550 PRINT
560 FOR A=1 TO 5
570 PRINT "FOREVER IN THE FUTURE, ";
580 IF A=1 THEN PRINT JC$
590 IF A=2 THEN PRINT JE$
600 IF A=3 THEN PRINT JA$
610 IF A=4 THEN PRINT JD$
620 IF A=5 THEN PRINT JB$
630 PRINT "WILL BE HANDLING THE FOLLOWING JOB"
640 PRINT "FOR ";NAM$;": ";
650 IF A=1 THEN PRINT E$
660 IF A=2 THEN PRINT D$
670 IF A=3 THEN PRINT C$
680 IF A=4 THEN PRINT B$
690 IF A=5 THEN PRINT A$
700 PRINT
710 NEXT A
720 PRINT "(ALL IT TAKES IS A LITTLE CHARM!)"

```

☐ If You Have . . .

APPLE II

Change: 520 PRINT "OKAY. NOW HIT THE RETURN KEY AND SAY"

ATARI

Add: 95 DIM NAM\$(40),A\$(90),B\$(90),C\$(90),D\$(90),E\$(90)
 96 DIM JA\$(40),JB\$(40),JC\$(40),JD\$(40),JE\$(40),X\$(1)

Change: 520 PRINT "OKAY. NOW HIT THE RETURN KEY AND SAY"

COMMODORE 64

Change: 520 PRINT "OKAY. NOW HIT THE RETURN KEY AND SAY"

COMMODORE VIC-20

Add: 565 IF A=3 THEN PRINT "HIT RETURN TO SEE MORE"
 566 IF A=3 THEN INPUT X\$

Change: 100 PRINT "WHAT'S YOUR NAME"
 150 PRINT "1:"
 180 PRINT "2:"
 210 PRINT "3:"

(continued)

```

270 PRINT "1:"
300 PRINT "2:"
360 PRINT "1:"
390 PRINT "2:"
420 PRINT "3:"
450 PRINT "4:"
480 PRINT "5:"
520 PRINT "OKAY.  NOW HIT THE RETURN KEY AND SAY"

```

TEXAS INSTRUMENTS 99/4A

```

Add: 582 PRINT JC$
      592 PRINT JE$
      602 PRINT JA$
      612 PRINT JD$
      622 PRINT JB$
      652 PRINT E$
      662 PRINT D$
      672 PRINT C$
      682 PRINT B$
      692 PRINT A$
      702 IF A<>4 THEN 710
      704 PRINT "HIT ENTER TO SEE MORE";
      706 INPUT A$
      708 PRINT

```

```

Change: 140 PRINT "INSIDE YOUR HOUSE THAT YOU  HATE TO DO:"
        340 PRINT "ENTER THE NAMES OF FIVE      PEOPLE YOU'D"
        350 PRINT "LIKE TO DO ALL YOUR        WORK FOR YOU:"
        530 PRINT "GOODBYE TO ALL THAT WORK    FOREVER!"
        580 IF A<>1 THEN 590
        590 IF A<>2 THEN 600
        600 IF A<>3 THEN 610
        610 IF A<>4 THEN 620
        620 IF A<>5 THEN 630
        630 PRINT "WILL BE HANDLING THE";"FOLLOWING JOB"
        650 IF A<>1 THEN 660
        660 IF A<>2 THEN 670
        670 IF A<>3 THEN 680
        680 IF A<>4 THEN 690
        690 IF A<>5 THEN 700
        720 PRINT "(ALL IT TAKES IS A LITTLE  CHARM!)"

```

TRS-80 COLOR COMPUTER

```

Add: 704 IF A=4 THEN 707
      705 IF A>4 THEN 710
      706 IF A<>2 THEN 710
      707 PRINT
      708 PRINT "HIT ENTER TO CONTINUE"
      709 INPUT X$

```

Tricky Toy

You're not a baby anymore! You're a top-notch computer programmer. You've outgrown all those childish toys from when you were little . . . or have you?

☐ Sample Run

AREN'T YOU GLAD YOU'RE OLD ENOUGH
THAT YOU CAN HAVE FUN WITH YOUR
COMPUTER, INSTEAD OF HAVING TO PLAY
WITH LITTLE KID'S TOYS LIKE ALPHABET
BLOCKS? YES

AAAAAAA
AAAAAAA
AAAAAAA
AAAAAAA

BBB
BBB
BBB

CCCCCCC
CCCCCCC
CCCCCCC
CCCCCCC
CCCCCCC
CCCCCCC

DDDDDDDD
DDDDDDDD
DDDDDDDD
DDDDDDDD
DDDDDDDD

EEEEEEEE
EEEEEEEE
EEEEEEEE
EEEEEEEE
EEEEEEEE
EEEEEEEE
EEEEEEEE
EEEEEEEE

FFFFFFFF
FFFFFFFF
FFFFFFFF
FFFFFFFF
FFFFFFFF
FFFFFFFF
FFFFFFFF
FFFFFFFF

Break

☐ Program Listing

```
10 REM TRICKY TOY
100 PRINT
110 PRINT "AREN'T YOU GLAD YOU'RE OLD ENOUGH"
120 PRINT "THAT YOU CAN HAVE FUN WITH YOUR"
130 PRINT "COMPUTER, INSTEAD OF HAVING TO PLAY"
140 PRINT "WITH LITTLE KID'S TOYS LIKE ALPHABET"
150 PRINT "BLOCKS";
160 INPUT D$
170 PRINT CHR$(12)
180 Z=65
190 A=INT(RND(1)*10)+2
200 FOR B=1 TO A
210 PRINT TAB(A);
220 FOR C=1 TO A
230 PRINT CHR$(Z);
240 NEXT C
250 PRINT
260 NEXT B
270 PRINT
280 Z=Z+1
290 IF Z>90 THEN Z=65
300 GOTO 190
```

☐ If You Have . . .

APPLE II

Change: 170 HOME

ATARI

```
Add: 95 DIM D$(3)
      315 FOR AA=1 TO IDX
      325 PRINT " ";:NEXT AA:RETURN
```

Change: 170 PRINT CHR\$(125)
190 A=INT(RND(0)*10)+2
210 IDX=A:GOSUB 315

COMMODORE 64

Change: 170 PRINT CHR\$(147)
190 A=INT(RND(0)*10)+2

COMMODORE VIC-20

Change: 150 PRINT "BLOCKS"
170 PRINT CHR\$(147)
190 A=INT(RND(0)*10)+2

TEXAS INSTRUMENTS 99/4A

Add: 292 Z=65

Change: 110 PRINT "AREN'T YOU GLAD YOU'RE OLD ENOUGH THAT"
120 PRINT "YOU CAN HAVE FUN WITH YOUR"
170 CALL CLEAR
190 A=INT(RND*10)+2
290 IF Z<91 THEN 300

TRS-80 COLOR COMPUTER

Change: 170 CLS
190 A=INT(RND(10))+2

Sidewords

You've probably written sentences backwards just for fun. But have you ever seen a sentence written sideways? With this program, you will!

☐ Sample Run

TYPE IN A SENTENCE WITHOUT ANY COMMAS
AND I'LL SHOW IT TO YOU SIDeways!

WHAT'S THE SENTENCE? I LOVE MY COMPUTER!

I I I I I I I I I I I I I I I I I
LOVE LOVE LOVE LOVE LOVE LOVE LOVE
MY MY MY MY MY MY MY MY MY MY MY MY MY
COMPUTER COMPUTER COMPUTER COMPUTER
! ! ! ! ! ! ! ! ! ! ! ! ! ! ! ! !

WANT TO TRY ANOTHER? YES

TYPE IN A SENTENCE WITHOUT ANY COMMAS
AND I'LL SHOW IT TO YOU SIDeways!

WHAT'S THE SENTENCE? IF MY FRIEND SALLY
WERE A HOUSE SHE'D BE CONDEMNED!

IF IF IF IF IF IF IF IF IF IF IF IF IF
MY MY MY MY MY MY MY MY MY MY MY MY MY
FRIEND FRIEND FRIEND FRIEND FRIEND
SALLY SALLY SALLY SALLY SALLY SALLY
WERE WERE WERE WERE WERE WERE WERE
A A A A A A A A A A A A A A A A A
HOUSE HOUSE HOUSE HOUSE HOUSE HOUSE
SHE'D SHE'D SHE'D SHE'D SHE'D SHE'D
BE BE BE BE BE BE BE BE BE BE BE BE BE
CONDEMNED CONDEMNED CONDEMNED
! ! ! ! ! ! ! ! ! ! ! ! ! ! ! ! !

WANT TO TRY ANOTHER? YES

TYPE IN A SENTENCE WITHOUT ANY COMMAS
AND I'LL SHOW IT TO YOU SIDeways!

WHAT'S THE SENTENCE? I ONCE ALMOST
SHAVED MY COLLIE.

I I I I I I I I I I I I I I I I I
 ONCE ONCE ONCE ONCE ONCE ONCE ONCE
 ALMOST ALMOST ALMOST ALMOST ALMOST
 SHAVED SHAVED SHAVED SHAVED SHAVED
 MY MY MY MY MY MY MY MY MY MY MY MY
 COLLIE COLLIE COLLIE COLLIE COLLIE

WANT TO TRY ANOTHER? NO

☐ Program Listing

```

10 REM SIDEWORDS
100 PRINT CHR$(12)
110 B=0
120 PRINT "TYPE IN A SENTENCE WITHOUT ANY COMMAS"
130 PRINT "AND I'LL SHOW IT TO YOU SIDEWAYS!"
140 PRINT
150 PRINT "WHAT'S THE SENTENCE";
160 INPUT A$
170 PRINT
180 IF A$="" THEN 160
190 L$=MID$(A$,LEN(A$),1)
200 IF L$<>"." AND L$<>"?" AND L$<>"!" THEN 230
210 A$=MID$(A$,1,LEN(A$)-1)+CHR$(32)
220 GOTO 240
230 A$=A$+CHR$(32)
240 FOR A=1 TO LEN(A$)+1
250 M$=MID$(A$,A,1)
260 IF M$=" " AND MID$(A$,A+1,1)=" " THEN 340
270 IF M$<>" " AND M$<>"." THEN 340
280 N$=MID$(A$,B+1,A-B)
290 FOR C=1 TO 39/LEN(N$)
300 PRINT N$;
310 NEXT C
320 PRINT
330 B=A
340 NEXT A
350 IF ASC(L$)>64 AND ASC(L$)<91 THEN 390
360 FOR C=1 TO 19
370 PRINT L$;" ";
380 NEXT C
390 PRINT
400 PRINT
410 PRINT "WANT TO TRY ANOTHER";
420 INPUT Y$
430 IF MID$(Y$,1,1)<>"N" THEN 100

```

☐ If You Have . . .

APPLE II

Change: 100 HOME

ATARI

Delete: 260

```
Add: 95 DIM A$(100),L$(1),M$(1),N$(100),Y$(3)
      344 N$=A$(B+1,LEN(A$))
      345 FOR A=1 TO 28/LEN(N$):PRINT N$;" ";:NEXT A
```

```
Change: 100 PRINT CHR$(125)
      190 L$=A$(LEN(A$),LEN(A$))
      210 REM
      230 REM
      240 FOR A=1 TO LEN(A$)
      250 M$=A$(A,A)
      280 N$=A$(B+1,B+(A-B))
      290 FOR C=1 TO 28/LEN(N$)
      430 IF Y$(1,1)<>"N" THEN 100
```

COMMODORE 64

Change: 100 PRINT CHR\$(147)

COMMODORE VIC-20

```
Change: 100 PRINT CHR$(147)
      150 PRINT "WHAT'S THE SENTENCE"
      290 FOR C=1 TO 21/LEN(N$)
      410 PRINT "WANT TO TRY ANOTHER"
```

TEXAS INSTRUMENTS 99/4A

```
Change: 100 CALL CLEAR
      120 PRINT "TYPE IN A SENTENCE WITHOUT ANY COMMAS"
      190 L$=SEG$(A$,LEN(A$),1)
      200 IF (L$<>".")*(L$<>"?")*(L$<>"!") THEN 230
      210 A$=SEG$(A$,1,LEN(A$)-1)&CHR$(32)
      230 A$=A$&CHR$(32)
      250 M$=SEG$(A$,A,1)
      260 IF (M$=" ")*(SEG$(A$,A+1,1)=" ") THEN 340
      270 IF (M$<>" ")*(M$<>".") THEN 340
      280 N$=SEG$(A$,B+1,A-B)
      290 FOR C=1 TO 28/LEN(N$)
      350 IF (ASC(L$)>64)*(ASC(L$)<91) THEN 390
      430 IF SEG$(Y$,1,1)<>"N" THEN 100
```

TRS-80 COLOR COMPUTER

Change: 100 CLS

290 FOR C=1 TO 30/LEN(N\$)

____ Reward Yourself ____

You've come a long way in this book. Now it's time for a little break. Toast yourself with your favorite liquid refreshment. It's right up there on your screen!

☐ Sample Run

NOW THAT YOU'VE GOTTEN THIS FAR, DON'T
YOU THINK IT'S TIME TO CELEBRATE A BIT?

HIT THE ENTER KEY TO SEE?

[illegible]

CHEERS!

□ Program Listing

```
10 REM REWARD YOURSELF
100 PRINT "NOW THAT YOU'VE GOTTEN THIS FAR, DON'T"
110 PRINT "YOU THINK IT'S TIME TO CELEBRATE A BIT?"
120 PRINT
130 PRINT "HIT THE ENTER KEY TO SEE";
140 INPUT D$
150 FOR A=1 TO 4
160 PRINT
170 NEXT A
180 PRINT TAB(8); "*****"
190 FOR B=1 TO 6
200 PRINT TAB(9); "*****"
210 NEXT B
220 FOR C=1 TO 5
230 PRINT TAB(9-C);
240 FOR D=1 TO 4+(2*C)
250 PRINT "**";
260 NEXT D
270 PRINT TAB(24+C)
280 FOR Z=1 TO 2*(7-C)
290 PRINT "**";
300 NEXT Z
310 PRINT "**"
320 NEXT C
330 FOR E=1 TO 7
340 PRINT TAB(4);
350 FOR F=1 TO 14
360 PRINT "**";
370 NEXT F
380 IF E=7 THEN 410
390 PRINT TAB(30); "****"
400 GOTO 450
410 PRINT TAB(26);
420 FOR G=1 TO 11
430 PRINT "**";
440 NEXT G
450 NEXT E
460 PRINT
470 PRINT
480 PRINT TAB(17); "CHEERS!"
```

☐ If You Have . . .

APPLE II

Change: 130 PRINT "HIT THE RETURN KEY TO SEE";

ATARI

```
Add: 95 DIM D$(1)
      265 R=7-C
      485 END
      495 FOR AA=1 TO IDX
      505 PRINT " ";:NEXT AA:RETURN
```

Change: 130 "HIT THE RETURN KEY TO SEE";
180 IDX=7:GOSUB 495:PRINT "*****"
200 IDX=8:GOSUB 495:PRINT "*****"
230 IDX=(8-C):GOSUB 495
270 IDX=(R+C):GOSUB 495
340 IDX=3:GOSUB 495
390 IDX=8:GOSUB 495:PRINT "****"
410 IDX=4:GOSUB 495
480 IDX=16:GOSUB 495:PRINT "CHEERS!"

COMMODORE 64

Change: 130 PRINT "HIT THE RETURN KEY TO SEE";

COMMODORE VIC-20

Change: 130 PRINT "HIT THE RETURN KEY TO SEE"
180 PRINT TAB(4);"*****"
190 FOR B=1 TO 5
200 PRINT TAB(5);"****"
220 FOR C=1 TO 3
230 PRINT TAB(6-C);
240 FOR D=1 TO 1+(2*C)
270 PRINT TAB(10+C)
280 FOR Z=1 TO 2*(5-C)
330 FOR E=1 TO 8
340 PRINT TAB(3);
350 FOR F=1 TO 7
390 PRINT TAB(14);"****"
410 PRINT TAB(12);
420 FOR G=1 TO 7
480 PRINT TAB(8);"CHEERS!"

TEXAS INSTRUMENTS 99/4A

Change: 110 PRINT "YOU THINK IT'S TIME TO CELEBRATE A BIT?"
180 PRINT TAB(5); "*****"
200 PRINT TAB(6); "*****"
230 PRINT TAB(6-C);
270 PRINT TAB(14+C);
340 PRINT TAB(1);
390 PRINT TAB(20); "****"
410 PRINT TAB(18);
420 FOR G=1 TO 7

TRS-80 COLOR COMPUTER

Delete: 150, 170, 470

Change: 180 PRINT TAB(4); "*****"
190 FOR B=1 TO 4
200 PRINT TAB(5); "*****"
220 FOR C=1 TO 3
230 PRINT TAB(6-C);
240 D=1 TO 1 TO 2+(2*C)
270 PRINT TAB(15+C)
280 FOR Z=1 TO 2*(5-C)
330 FOR E=1 TO 5
340 PRINT TAB(3);
350 FOR F=1 TO 8
380 IF E=5 THEN 410
390 PRINT TAB(19); "****"
410 PRINT TAB(17);
420 FOR G=1 TO 7
480 PRINT TAB(11); "CHEERS!"

Sign on the Dotted Line

You think it's easy being a successful businessperson? With all the important contracts you have to sign, you may get writer's cramp. Maybe your computer can help spare your fingers.

□ Sample Run

ENTER YOUR WHOLE NAME? GOLDIE NOVA

GOLDIE NOVA, YOU ARE SUCH A BIG
BUSINESSPERSON, I HAVE 200 CONTRACTS
FOR YOU TO SIGN. I'LL BET YOU CAN'T
SIGN THEM ALL ON THE DOTTED LINE
BEFORE YOUR HAND GETS TIRED!

MAYBE YOUR COMPUTER CAN HELP. HIT THE ENTER KEY TO FIND OUT?

[illegible]

Break

☐ Program Listing

```
10 REM SIGN ON THE DOTTED LINE
100 K=0
110 PRINT "ENTER YOUR WHOLE NAME";
120 INPUT WNAME$
130 LNAME=LEN(WNAME$)
140 IF LNAME=0 THEN 110
150 PRINT CHR$(12)
160 PRINT WNAME$;"", YOU ARE SUCH A BIG"
170 PRINT "BUSINESSPERSON, I HAVE 200 CONTRACTS"
180 PRINT "FOR YOU TO SIGN. I'LL BET YOU CAN'T"
190 PRINT "SIGN THEM ALL ON THE DOTTED LINE"
200 PRINT "BEFORE YOUR HAND GETS TIRED!"
210 PRINT
220 PRINT "MAYBE YOUR COMPUTER CAN HELP. HIT THE"
230 PRINT "ENTER KEY TO FIND OUT";
240 INPUT D$
250 FOR A=1 TO 25
260 PRINT
270 NEXT A
280 LFTSIDE=INT((40-LNAME)*RND(1))
290 RGHTSIDE=39-LNAME-LFTSIDE
300 FOR B=1 TO LFTSIDE
310 PRINT ".";
320 NEXT B
330 PRINT WNAME$;
340 FOR C=1 TO RGHTSIDE
350 PRINT ".";
360 NEXT C
370 PRINT
380 K=K+1
390 IF K=200 THEN 410
400 GOTO 280
410 PRINT
420 PRINT "WOW, THAT WAS EASY!"
```

☐ If You Have . . .

APPLE II

Change: 150 HOME
230 PRINT "RETURN KEY TO FIND OUT";

ATARI

Add: 95 DIM WNAME\$(40),D\$(1)

(continued)

```
150 PRINT CHR$(125)
230 PRINT "RETURN KEY TO FIND OUT";
280 LFTSIDE=INT((36-LNAM)*RND(0))
290 RGHTSIDE=35-LNAM-LFTSIDE
```

COMMODORE 64

Change: 150 PRINT CHR\$(147)
230 PRINT "RETURN KEY TO FIND OUT";
280 LFTSIDE=INT((40-LNAM)*RND(0))

COMMODORE VIC-20

Add: 295 IF LFTSIDE=0 THEN RGHTSIDE=RGHTSIDE-1

Change: 150 PRINT CHR\$(147)
230 PRINT "RETURN KEY TO FIND OUT"
280 LFTSIDE=INT((20-LNAM)*RND(0))
290 RGHTSIDE=21-LNAM-LFTSIDE

TEXAS INSTRUMENTS 99/4A

Change: 150 CALL CLEAR
170 PRINT "BUSINESSPERSON, I HAVE 200 CONTRACTS"
180 PRINT "FOR YOU TO SIGN. I'LL BET YOU CAN'T"
280 LFTSIDE=INT((40-LNAM)*RND)

TRS-80 COLOR COMPUTER

Change: 150 CLS
280 LFTSIDE=RND(30-LNAM)
290 RGHTSIDE=30-LNAM-LFTSIDE

Magic Wishing Well

Now you don't have to travel to your lucky pond and toss your hard-earned pennies into the drink. Your computer wishing well will grant your every wish
— well, almost!

☐ Sample Run

WHAT'S YOUR NAME? RHODA

OKAY, RHODA.

YOU HAVE A \$1.00 CREDIT WITH
THE FIRST COMPUTER BANK & TRUST.
RIGHT OUTSIDE THE BANK
IS A WISHING WELL.

WHY DON'T YOU TRY YOUR LUCK?
HIT THE ENTER KEY TO LOOK AT THE SIGN
THAT SITS OVER THE WISHING WELL?

```
XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX
X                                     X
X                                     X
X                                     X
X      COMPUTER WISHING WELL        X
X    WISHES:  5, 10, OR 25 CENTS    X
X      (EXACT CHANGE ONLY)          X
X                                     X
X                                     X
X                                     X
X                                     X
XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX
```

YOU HAVE \$1.00 AVAILABLE FOR WISHES.

HOW MANY CENTS WOULD YOU LIKE TO TOSS
IN THE WELL: 5, 10, OR 25? 25

OKAY. NOW TYPE IN A 25 CENT WISH
? I WANT TO GO TO NORWAY

SORRY! THAT WISH DIDN'T QUITE MAKE IT.

(continued)

```

XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX
X                                     X
X                                     X
X                                     X
X      COMPUTER WISHING WELL        X
X    WISHES:  5, 10, OR 25 CENTS    X
X      (EXACT CHANGE ONLY)          X
X                                     X
X                                     X
X                                     X
XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX

```

YOU HAVE 75 CENTS LEFT FOR MORE WISHES

HOW MANY CENTS WOULD YOU LIKE TO TOSS
IN THE WELL: 5, 10, OR 25? 25

OKAY. NOW TYPE IN A 25 CENT WISH
? I WANT A MAINFRAME COMPUTER

YOU'LL GET YOUR WISH! HOW ABOUT THAT!

```

XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX
X                                     X
X                                     X
X                                     X
X      COMPUTER WISHING WELL        X
X    WISHES:  5, 10, OR 25 CENTS    X
X      (EXACT CHANGE ONLY)          X
X                                     X
X                                     X
X                                     X
XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX

```

YOU HAVE 50 CENTS LEFT FOR MORE WISHES

HOW MANY CENTS WOULD YOU LIKE TO TOSS
IN THE WELL: 5, 10, OR 25? 25

OKAY. NOW TYPE IN A 25 CENT WISH
? I WOULD LIKE A ROLLS ROYCE

SORRY! THAT WISH DIDN'T QUITE MAKE IT.

```

XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX
X                                     X
X                                     X
X                                     X
X      COMPUTER WISHING WELL        X
X    WISHES:  5, 10, OR 25 CENTS    X
X      (EXACT CHANGE ONLY)          X
X                                     X
X                                     X
X                                     X
XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX

```

YOU HAVE 25 CENTS LEFT FOR MORE WISHES

HOW MANY CENTS WOULD YOU LIKE TO TOSS
IN THE WELL: 5, 10, OR 25? 15

READ THE SIGN--EXACT CHANGE ONLY!
HOW MANY CENTS WOULD YOU LIKE TO TOSS
IN THE WELL: 5, 10, OR 25? 10

OKAY. NOW TYPE IN A 10 CENT WISH
? I WANT TO LEARN HOW TO FLY

SORRY! THAT WISH DIDN'T QUITE MAKE IT.

```

XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX
X                                     X
X                                     X
X                                     X
X      COMPUTER WISHING WELL        X
X    WISHES:  5, 10, OR 25 CENTS    X
X      (EXACT CHANGE ONLY)          X
X                                     X
X                                     X
X                                     X
XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX

```

YOU HAVE 15 CENTS LEFT FOR MORE WISHES

HOW MANY CENTS WOULD YOU LIKE TO TOSS
IN THE WELL: 5, 10, OR 25? 25

SORRY! YOUR BANK ACCOUNT IS TOO LOW!

HOW MANY CENTS WOULD YOU LIKE TO TOSS
IN THE WELL: 5, 10, OR 25? 10

OKAY. NOW TYPE IN A 10 CENT WISH
? I WISH FOR A ROOM OF MY OWN

(continued)

SORRY! THAT WISH DIDN'T QUITE MAKE IT.

```
XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX
X                                     X
X                                     X
X                                     X
X      COMPUTER WISHING WELL        X
X  WISHES:  5, 10, OR 25 CENTS      X
X      (EXACT CHANGE ONLY)          X
X                                     X
X                                     X
X                                     X
X                                     X
XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX
```

YOU HAVE 5 CENTS LEFT FOR MORE WISHES

HOW MANY CENTS WOULD YOU LIKE TO TOSS
IN THE WELL: 5, 10, OR 25? 5

OKAY. NOW TYPE IN A 5 CENT WISH
? I WANT MY OWN COUNTRY

SORRY! THAT WISH DIDN'T QUITE MAKE IT.

```
XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX
X                                     X
X                                     X
X                                     X
X      COMPUTER WISHING WELL        X
X  WISHES:  5, 10, OR 25 CENTS      X
X      (EXACT CHANGE ONLY)          X
X                                     X
X                                     X
X                                     X
X                                     X
XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX
```

YOU'RE BROKE--ALL OUT OF WISHES!

☐ Program Listing

```
10 REM MAGIC WISHING WELL
100 AMT=100
110 PRINT "WHAT'S YOUR NAME";
120 INPUT NAM$
130 IF NAM$="" THEN 110
140 PRINT
150 PRINT "OKAY, ";NAM$;". "
160 PRINT "YOU HAVE A $1.00 CREDIT WITH"
170 PRINT "THE FIRST COMPUTER BANK & TRUST."
180 PRINT "RIGHT OUTSIDE THE BANK"
190 PRINT "IS A WISHING WELL."
200 PRINT
210 PRINT "WHY DON'T YOU TRY YOUR LUCK?"
220 PRINT "HIT THE ENTER KEY TO LOOK AT THE SIGN"
230 PRINT "THAT SITS OVER THE WISHING WELL";
240 INPUT X$
250 PRINT
260 GOSUB 650
270 PRINT "HOW MANY CENTS WOULD YOU LIKE TO TOSS"
280 PRINT "IN THE WELL: 5, 10, OR 25";
290 INPUT CENTS
300 IF CENTS<=AMT THEN 350
310 PRINT
320 PRINT "SORRY! YOUR BANK ACCOUNT IS TOO LOW!"
330 PRINT
340 GOTO 270
350 IF CENTS>0 THEN 400
360 PRINT
370 PRINT "CHEAPSKATE! TRY AGAIN!"
380 PRINT
390 GOTO 270
400 IF CENTS=5 OR CENTS=10 OR CENTS=25 THEN 440
410 PRINT
420 PRINT "READ THE SIGN--EXACT CHANGE ONLY!"
430 GOTO 270
440 PRINT
450 PRINT "OKAY. NOW TYPE IN A ";CENTS;" CENT WISH:"
460 INPUT WISH$
470 IF WISH$<>"" THEN 510
480 PRINT
490 PRINT "WHAT KIND OF WISH IS THAT? TRY AGAIN."
500 GOTO 460
510 PRINT
520 GRANT=INT(RND(1)*(30-CENTS))+1
530 IF GRANT<3 THEN 580
540 PRINT
550 PRINT " SORRY! THAT WISH DIDN'T QUITE MAKE IT."
560 PRINT
570 GOTO 610
580 PRINT
590 PRINT " YOU'LL GET YOUR WISH! HOW ABOUT THAT!"
600 PRINT
```

(continued)

```

610 GOSUB 650
620 GOTO 270
630 END
640 REM ----- SUBROUTINE -----
650 PRINT " ";
660 FOR A=1 TO 37
670 PRINT "X";
680 NEXT A
690 PRINT "X"
700 FOR A=1 TO 3
710 PRINT " "; "X"; TAB(39); "X"
720 NEXT A
730 PRINT " X          COMPUTER WISHING WELL          X"
740 PRINT " X      WISHES:  5, 10, OR 25 CENTS      X"
750 PRINT " X          (EXACT CHANGE ONLY)          X"
760 FOR A=1 TO 3
770 PRINT " "; "X"; TAB(39); "X"
780 NEXT A
790 PRINT " ";
800 FOR A=1 TO 37
810 PRINT "X";
820 NEXT A
830 PRINT "X"
840 PRINT
850 AMT=AMT-CENTS
860 IF AMT>4 THEN 890
870 PRINT "      YOU'RE BROKE--ALL OUT OF WISHES!"
880 END
890 IF AMT<100 THEN 920
900 PRINT "      YOU HAVE $1.00 AVAILABLE FOR WISHES."
910 GOTO 930
920 PRINT "      YOU HAVE";AMT;" CENTS LEFT FOR MORE WISHES"
930 PRINT
940 PRINT
950 PRINT
960 RETURN

```

☐ If You Have . . .

APPLE II

```

Change: 220 PRINT "HIT THE RETURN KEY TO LOOK AT THE SIGN"
        520 G=INT(RND(1)*(30-CENTS))+1
        530 IF G<3 THEN 580
        920 PRINT "YOU HAVE ";AMT;" CENTS LEFT FOR MORE
              WISHES"

```

ATARI

Add: 95 DIM NAM\$(40),X\$(1),WISH\$(40)
975 FOR AA=1 TO IDX
985 PRINT " ";:NEXT AA:RETURN

Change: 220 PRINT "HIT THE RETURN KEY TO LOOK AT THE SIGN"
520 GRANT=INT(RND(0)*(30-CENTS))+1
660 FOR A=1 TO 35
710 PRINT " "; "X";:IDX=34:GOSUB 975:PRINT "X"
770 PRINT " "; "X";:IDX=34:GOSUB 975:PRINT "X"
800 FOR A=1 TO 35
920 PRINT "YOU HAVE ";AMT;" CENTS LEFT FOR MORE WISHES"

COMMODORE 64

Change: 220 PRINT "HIT THE RETURN KEY TO LOOK AT THE SIGN"
520 GRANT=INT(RND(0)*(30-CENTS))+1
710 PRINT " "; "X";TAB(38);"X"
770 PRINT " "; "X";TAB(38);"X"

COMMODORE VIC-20

Delete: 560,600,840,940,950

Add: 735 PRINT " X WISHING WELL X"
745 PRINT " X 5,10,OR 25 CENTS X"
755 PRINT " X CHANGE ONLY X"

Change: 110 PRINT "WHAT'S YOUR NAME"
220 PRINT "HIT THE RETURN KEY TO LOOK AT THE SIGN"
280 PRINT "IN THE WELL: 5,10,OR 25"
520 GRANT=INT(RND(0)*(30-CENTS))+1
660 FOR A=1 TO 19
700 FOR A=1 TO 2
710 PRINT " "; "X";TAB(20);"X"
730 PRINT " X COMPUTER X"
740 PRINT " X WISHES: X"
750 PRINT " X EXACT X"
760 FOR A=1 TO 2
770 PRINT " "; "X";TAB(20);"X"
800 FOR A=1 TO 19

TEXAS INSTRUMENTS 99/4A

Add: 742 PRINT " X 25 CENTS X"

Change: 170 PRINT "THE FIRST COMPUTER BANK & TRUST."
230 PRINT "THAT SITS OVER THE WISHING WELL";

(continued)

```

270 PRINT "HOW MANY CENTS WOULD YOU      LIKE TO TOSS"
400 IF (CENTS=5)+(CENTS=10)+(CENTS=25) THEN 440
520 GRANT=INT(RND*(30-CENTS))+1
660 FOR A=1 TO 25
710 PRINT " "; "X"; TAB(27); "X"
730 PRINT " X COMPUTER WISHING WELL X"
740 PRINT " X   WISHES: 5, 10, OR   X"
750 PRINT " X (EXACT CHANGE ONLY)  X"
770 PRINT " "; "X"; TAB(27); "X"
800 FOR A=1 TO 25
900 PRINT "   YOU HAVE $1.00 AVAILABLE  FOR WISHES."

```

TRS-80 COLOR COMPUTER

Delete: 560,600,700,760,940,950

```

Change: 520 GRANT=INT(RND(30-CENTS))+1
660 FOR A=1 TO 29
710 PRINT " "; "X"; TAB(30); "X";
720 PRINT " ";
730 PRINT " X   COMPUTER WISHING WELL      X"
740 PRINT " X WISHES 5, 10, OR 25 CENTS    X"
750 PRINT " X (EXACT CHANGE ONLY)         X"
770 PRINT " "; "X"; TAB(30); "X";
780 PRINT " ";
790 PRINT " ";
800 FOR A=1 TO 29

```

The Perfect Computer

Computers hardly ever make mistakes. But once in a while, even your machine can have a bad day. Until you run this program, you'll have no idea how confused your machine can get!

☐ Sample Run

I JUST WANT TO SEE IF I'M OKAY.
PLEASE HIT THE ENTER KEY?

I'M FEELING A LITTLE MIXED UP TODAY
P.T'MLLSING A LITTLE MIXED UP TODAY
P.T'MLLSPUN'A LITTLE MIXED UP TODAY
P.T'MLLSPUN'H'SP[[LE MIXED UP TODAY
P.T'MLLSPUN'H'SP[[SL'MIXED UP TODAY
P.T'MLLSPUN'H'SP[[SL'TP_LK'UP TODAY
P.T'MLLSPUN'H'SP[[SL'TP_LK'\W' [VDAY
P.T'MLLSPUN'H'SP[[SL'TP_LK'\W' [ODAY
P.T'MLLSPUN'H'SP[[SL'TP_LK'\W TODAY
P.T'MLLSPUN'H'SP[[SL'TIXED UP TODAY
P.T'MLLSPUN'H'SP[TLE MIXED UP TODAY
P.T'MLLSPUN'H'LITTLE MIXED UP TODAY
P.T'MLLSING A LITTLE MIXED UP TODAY
P.T'FEELING A LITTLE MIXED UP TODAY

WHEW, THAT WAS A CLOSE ONE...
I ALMOST LOST ALL COMMUNICATION
WITH YOU. BUT I FEEL FINE NOW.

LET'S SEE...WHERE WERE WE?
OH, YES. WHAT'S YOUR FIRST NAME? CINDY.

OKAY, CINDY, WATCH THIS TRICK--

I BET I CAN GUESS YOUR LAST NAME.
HIT THE ENTER KEY AND SEE?

YOUR LAST NAME IS FQWSVBXE--RIGHT? NO
YOUR LAST NAME IS DRMVDDPW--RIGHT? NO
YOUR LAST NAME IS DQGHFY YO--RIGHT? NO
YOUR LAST NAME IS LGHAVDWX--RIGHT? NO
YOUR LAST NAME IS HLKGUNDI--RIGHT? NO

(continued)

WHAT'S YOUR REAL LAST NAME? SIMPSON

SIMPSON? THAT WAS MY NEXT GUESS!!
YOU DIDN'T GIVE ME ENOUGH TIME!

CINDY NOSPMIS--
THAT'S A BACKWARDS-SOUNDING
NAME IF EVER I'VE HEARD ONE!

NOW LET'S TRY SOMETHING I KNOW
I'LL GET RIGHT--SOME EASY MATH!

PICK A NUMBER FROM 1 TO 10? 5

PICK ANOTHER NUMBER FROM 1 TO 10? 4

I'M GREAT AT MATH...WATCH ME...

5	X	4	=	27
5	/	4	=	13.25
5	+	4	=	10.25
5	-	4	=	15.25

AM I RIGHT? NO

WELL, I GIVE UP, THEN. EX-CUSE ME!

☐ Program Listing

```
10 REM THE PERFECT COMPUTER
100 PRINT
110 PRINT "...HMMMM"
120 PRINT
130 PRINT "I JUST WANT TO SEE IF I'M OKAY."
140 PRINT "PLEASE HIT THE ENTER KEY";
150 INPUT X$
160 PRINT
170 R=(RND(1)*10)+1
180 A$="I'M FEELING A LITTLE MIXED UP TODAY"
190 PRINT A$
200 FOR A=1 TO LEN(A$)
210 MID$(A$,A,1)=CHR$(ASC(MID$(A$,A,1))+R)
220 PRINT A$
230 NEXT A
240 FOR B=LEN(A$) TO 1 STEP -1
250 MID$(A$,B,1)=CHR$(ASC(MID$(A$,B,1))-R)
260 PRINT A$
270 NEXT B
```

```

280 PRINT
290 PRINT "WHEW, THAT WAS A CLOSE ONE..."
300 PRINT "I ALMOST LOST ALL COMMUNICATION"
310 PRINT "WITH YOU. BUT I FEEL FINE NOW."
320 PRINT
330 PRINT "LET'S SEE...WHERE WERE WE?"
340 PRINT "OH, YES. WHAT'S YOUR FIRST NAME";
350 INPUT F$
360 IF F$="" THEN 340
370 PRINT
380 PRINT "OKAY, ";F$;", WATCH THIS TRICK--"
390 PRINT
400 PRINT "I BET I CAN GUESS YOUR LAST NAME."
410 PRINT "HIT THE ENTER KEY AND SEE";
420 INPUT X$
430 PRINT
440 FOR C=1 TO 5
450 PRINT
460 PRINT "YOUR LAST NAME IS ";
470 FOR D=1 TO 8
480 PRINT CHR$((RND(1)*25)+65);
490 NEXT D
500 PRINT "--RIGHT";
510 INPUT R$
520 PRINT
530 IF R$="" THEN 500
540 IF MID$(R$,1,1)<>"Y" THEN 590
550 PRINT
560 PRINT "WOW--I DON'T KNOW WHAT'S STRANGER--"
570 PRINT "YOUR NAME OR MY GUESSING ABILITY!"
580 GOTO 600
590 NEXT C
600 PRINT
610 PRINT "WHAT'S YOUR REAL LAST NAME";
620 INPUT L$
630 IF L$="" THEN 610
640 PRINT
650 PRINT L$;"? THAT WAS MY NEXT GUESS!!"
660 PRINT "YOU DIDN'T GIVE ME ENOUGH TIME!"
670 PRINT
680 PRINT F$;CHR$(32);
690 FOR E=LEN(L$) TO 1 STEP -1
700 PRINT MID$(L$,E,1);
710 NEXT E
720 PRINT "--"
730 PRINT "THAT'S A BACKWARDS-SOUNDING"
740 PRINT "NAME IF EVER I'VE HEARD ONE!"
750 PRINT
760 PRINT "NOW LET'S TRY SOMETHING I KNOW"
770 PRINT "I'LL GET RIGHT--SOME EASY MATH!"
780 PRINT
790 PRINT "PICK A NUMBER FROM 1 TO 10";
800 INPUT N
810 IF N<1 OR N>10 THEN 790
820 PRINT

```

(continued)

```

830 PRINT "PICK ANOTHER NUMBER FROM 1 TO 10";
840 INPUT M
850 IF M<1 OR M>10 THEN 830
860 PRINT
870 PRINT "I'M GREAT AT MATH...WATCH ME..."
880 PRINT
890 PRINT N;" X ";M;" = ";N*M+(INT(RND(1)*10)+5)
900 PRINT N;" / ";M;" = ";N/M+(INT(RND(1)*10)+5)
910 PRINT N;" + ";M;" = ";N/M+(INT(RND(1)*10)+5)
920 PRINT N;" - ";M;" = ";N/M+(INT(RND(1)*10)+5)
930 PRINT
940 PRINT "AM I RIGHT";
950 INPUT Y$
960 IF MID$(Y$,1,1)="Y" THEN 1000
970 PRINT
980 PRINT "WELL, I GIVE UP, THEN. EX-CUSE ME!"
990 END
1000 PRINT
1010 PRINT "SEE? I TOLD YOU I WAS FINE!"
1020 PRINT "YOU'RE THE ONE WHO'S MIXED UP!"

```

☐ If You Have . . .

APPLE II

```

Add: 175 R=INT(R)
      205 AA$=MID$(A$,1,A-1)+CHR$(ASC(MID$(A$,A,1))+R)
      245 AA$=MID$(A$,1,B-1)+CHR$(ASC(MID$(A$,B,1))-R)

```

```

Change: 140 PRINT "PLEASE HIT THE RETURN KEY";
        210 A$=AA$+MID$(A$,A+1,LEN(A$)-A)
        250 A$=AA$+MID$(A$,B+1,LEN(A$)-B)
        410 PRINT "HIT THE RETURN KEY AND SEE";

```

ATARI

```

Add: 95 DIM X$(1),A$(40),F$(17),R$(3),L$(17),Y$(3)

```

```

Change: 140 PRINT "PLEASE HIT THE RETURN KEY";
        170 R=(RND(0)*10)+1
        210 A$(A,A)=CHR$(ASC(A$(A,A))+R)
        250 A$(B,B)=CHR$(ASC(A$(B,B))-R)
        410 PRINT "HIT THE RETURN KEY AND SEE";
        480 PRINT CHR$((RND(0)*25)+65);
        540 IF R$(1,1)<>"Y" THEN 590
        700 PRINT L$(E,E);
        890 PRINT N;" X ";M;" = ";N*M+(INT(RND(0)*10)+5)
        900 PRINT N;" / ";M;" = ";N/M+(INT(RND(0)*10)+5)

```

```

910 PRINT N;" + ";M;" = ";N/M+(INT(RND(0)*10)+5)
920 PRINT N;" - ";M;" = ";N/M+(INT(RND(0)*10)+5)
960 IF Y$(1,1)="Y" THEN 1000

```

COMMODORE 64

```

Add:  205 XY=ASC(MID$(A$,A,1))+R
      245 XYZ=ASC(MID$(A$,B,1))-R+1

```

```

Change: 140 PRINT "PLEASE HIT THE RETURN KEY";
        170 R=(RND(0)*10)+1
        210 A$=LEFT$(A$,A-1)+CHR$(XY)+MID$(A$,A+1)
        250 A$=LEFT$(A$,B-1)+CHR$(XYZ)+MID$(A$,B+1)
        410 PRINT "HIT THE RETURN KEY AND SEE";
        890 PRINT N;" X ";M;" = ";N*M+(INT(RND(0)*10)+5)
        900 PRINT N;" / ";M;" = ";N/M+(INT(RND(0)*10)+5)
        910 PRINT N;" + ";M;" = ";N/M+(INT(RND(0)*10)+5)
        920 PRINT N;" - ";M;" = ";N/M+(INT(RND(0)*10)+5)

```

COMMODORE VIC-20

```

Add:  205 XY=ASC(MID$(A$,A,1))+R
      245 XYZ=ASC(MID$(A$,B,1))-R+1

```

```

Change: 140 PRINT "PLEASE HIT THE RETURN KEY"
        170 R=(RND(1)*10)+1
        210 A$=LEFT$(A$,A-1)+CHR$(XY)+MID$(A$,A+1)
        250 A$=LEFT$(A$,B-1)+CHR$(XYZ)+MID$(A$,B+1)
        340 PRINT "OH, YES.  WHAT'S YOUR FIRST NAME"
        410 PRINT "HIT THE RETURN KEY AND SEE"
        500 PRINT "--RIGHT"
        610 PRINT "WHAT'S YOUR REAL LAST NAME"
        790 PRINT "PICK A NUMBER FROM 1 TO 10"
        830 PRINT "PICK ANOTHER NUMBER FROM 1 TO 10"
        890 PRINT N;" X ";M;" = ";N*M+(INT(RND(0)*10)+5)
        900 PRINT N;" / ";M;" = ";N/M+(INT(RND(0)*10)+5)
        910 PRINT N;" + ";M;" = ";N/M+(INT(RND(0)*10)+5)
        920 PRINT N;" - ";M;" = ";N/M+(INT(RND(0)*10)+5)
        940 PRINT "AM I RIGHT"

```

TEXAS INSTRUMENTS 99/4A

```

Add:  232 A$=AA$
      272 A$=AA$

```

```

Change: 130 PRINT "I JUST WANT TO SEE IF I'M  OKAY."
        170 R=(RND*10)+1
        210 AA$=AA$&CHR$(ASC(SEG$(A$,A,1))+R)
        220 PRINT AA$;SEG$(A$,A+1,LEN(A$)-A)
        250 AA$=AA$&CHR$(ASC(SEG$(A$,B,1))-R)
        260 PRINT AA$;SEG$(A$,B+1,LEN(A$)-B)

```

(continued)

```

290 PRINT "WHEW, THAT WAS A CLOSE ONE.."
300 PRINT "I ALMOST LOST ALL COMMUNICA-TION"
310 PRINT "WITH YOU. BUT I FEEL FINE NOW."
340 PRINT "OH, YES. WHAT'S YOUR FIRST NAME";
480 PRINT CHR$( (RND*25)+65 );
540 IF SEG$(R$,1,1)<>"Y" THEN 590
560 PRINT "WOW--I DON'T KNOW WHAT'S STRANGER--"
570 PRINT "YOUR NAME OR MY GUESSING ABILITY!"
660 PRINT "YOU DIDN'T GIVE ME ENOUGH TIME!"
700 PRINT SEG$(L$,E,1);
770 PRINT "I'LL GET RIGHT--SOME EASY MATH!"
810 IF (N<1)+(N>10) THEN 790
830 PRINT "PICK ANOTHER NUMBER FROM 1 TO 10";
850 IF (M<1)+(M>10) THEN 830
890 PRINT N;" X ";M;" = ";N*M+(INT(RND*10)+5)
900 PRINT N;" / ";M;" = ";N/M+(INT(RND*10)+5)
910 PRINT N;" + ";M;" = ";N/M+(INT(RND*10)+5)
920 PRINT N;" - ";M;" = ";N/M+(INT(RND*10)+5)
960 IF SEG$(Y$,1,1)="Y" THEN 1000
980 PRINT "WELL I GIVE UP, THEN. EX- CUSE ME!"
1020 PRINT "YOU'RE THE ONE WHO'S MIXED UP!"

```

TRS-80 COLOR COMPUTER

Change: 170 R=RND(10)

```

480 PRINT CHR$(RND(25)+64);
760 PRINT "NOW LET'S TRY SOMETHING I KNOW"
890 PRINT N;" X ";M;" = ";N*M+RND(10)+5
900 PRINT N;" / ";M;" = ";N/M+RND(10)+5
910 PRINT N;" + ";M;" = ";N/M+RND(10)+5
920 PRINT N;" - ";M;" = ";N/M+RND(10)+5

```

Computer Magician

Introducing the miraculous, the fantastic, the stupendous Computo — world's greatest magician! Computo can perform six — count 'em, six — astounding tricks that will have you gasping with amazement.
No magician alive can rival the great Computo!

☐ Sample Run

I CAN PERFORM INCREDIBLE FEATS
OF MAGIC! JUST WATCH!

=====
HERE ARE THE TRICKS I CAN DO:

- 1 TURN DIRT INTO GOLD
- 2 WALK ON WATER
- 3 CUT A WOMAN IN HALF
- 4 MAKE A TWO-TON CAR VANISH
- 5 MAKE AN ELEPHANT APPEAR
- 6 PULL A RABBIT OUT OF A HAT
- 7 QUIT

=====

CHOOSE ONE? 1

WATCH ME TURN DIRT INTO GOLD!

HIT THE ENTER KEY TO CONTINUE?

HERE IT IS: PLAIN OLD D I R T

HIT THE ENTER KEY TO CONTINUE?

PRESTO! G O L D!

=====
THAT WAS FANTASTICALLY TERRIFIC!
=====

HIT THE ENTER KEY TO CONTINUE?

=====
HERE ARE THE TRICKS I CAN DO:

- 1 TURN DIRT INTO GOLD
- 2 WALK ON WATER
- 3 CUT A WOMAN IN HALF

(continued)

- 4 MAKE A TWO-TON CAR VANISH
- 5 MAKE AN ELEPHANT APPEAR
- 6 PULL A RABBIT OUT OF A HAT
- 7 QUIT

=====

CHOOSE ONE? 2

WATCH ME WALK ON WATER!

HIT THE ENTER KEY TO CONTINUE?

WALK

WATER WATER WATER WATER WATER WATER

WALK

WATER WATER WATER WATER WATER WATER

WALK

WATER WATER WATER WATER WATER WATER

WALK

WATER WATER WATER WATER WATER WATER

WALK

WATER WATER WATER WATER WATER WATER

WALK

WATER WATER WATER WATER WATER WATER

WALK

WATER WATER WATER WATER WATER WATER

=====

THAT WAS INCREDIBLY WONDERFUL!

=====

=====

HERE ARE THE TRICKS I CAN DO:

- 1 TURN DIRT INTO GOLD
- 2 WALK ON WATER
- 3 CUT A WOMAN IN HALF
- 4 MAKE A TWO-TON CAR VANISH
- 5 MAKE AN ELEPHANT APPEAR
- 6 PULL A RABBIT OUT OF A HAT
- 7 QUIT

=====

CHOOSE ONE? 3

WATCH ME CUT A WOMAN IN HALF!

HIT THE ENTER KEY TO CONTINUE?

149

=====
HERE ARE THE TRICKS I CAN DO:

- 1 TURN DIRT INTO GOLD
- 2 WALK ON WATER
- 3 CUT A WOMAN IN HALF
- 4 MAKE A TWO-TON CAR VANISH
- 5 MAKE AN ELEPHANT APPEAR
- 6 PULL A RABBIT OUT OF A HAT
- 7 QUIT

=====
CHOOSE ONE? 5

WATCH ME MAKE AN ELEPHANT APPEAR!

HIT THE ENTER KEY TO CONTINUE?

AN ELEPHANT

SORRY THAT WAS SO SLOW, BUT THESE BIG
ANIMALS ARE HARD TO MOVE QUICKLY.

=====
THAT WAS TOO AMAZING FOR WORDS!
=====

=====
HERE ARE THE TRICKS I CAN DO:

- 1 TURN DIRT INTO GOLD
- 2 WALK ON WATER
- 3 CUT A WOMAN IN HALF
- 4 MAKE A TWO-TON CAR VANISH
- 5 MAKE AN ELEPHANT APPEAR
- 6 PULL A RABBIT OUT OF A HAT
- 7 QUIT

=====
CHOOSE ONE? 6

WATCH ME PULL A RABBIT OUT OF A HAT!

HIT THE ENTER KEY TO CONTINUE?

THE HAT JUST FELL ON THE RABBIT!

OOPS! THE RABBIT ESCAPED!

THE RABBIT IS LOOSE IN YOUR COMPUTER!

ONE SEC! THE RABBIT'S WEARING THE HAT!

THE RABBIT IS LOOSE IN YOUR COMPUTER!

HEY! STOP! THE RABBIT'S EATING THE HAT!

THE HAT JUST FELL ON THE RABBIT!

HOLD ON! I LOST THE HAT!

OKAY: NOW WE'RE SET!

HIT THE ENTER KEY TO CONTINUE?

MMMMMMMM

MMMM

MMMM

MMMM

MMMM

MMMM

MMMM

A

MMMMMMMM

MMMM

MMMM

MMMM

MMMM

MMMM

MMMM

A

MMMMMMMM

MMMM

MMMM

MMMM

MMMM

MMMM

MMMM

MMMM

A R

MMMMMMMM

MMMM

MMMM

MMMM

MMMM

MMMM

MMMM

(continued)

A RA
MMMMMMMM
MMMM
MMMM
MMMM
MMMM
MMMM
MMMM

A RAB
MMMMMMMM
MMMM
MMMM
MMMM
MMMM
MMMM
MMMM

A RABB
MMMMMMMM
MMMM
MMMM
MMMM
MMMM
MMMM
MMMM

A RABBI
MMMMMMMM
MMMM
MMMM
MMMM
MMMM
MMMM
MMMM

A RABBIT
MMMMMMMM
MMMM
MMMM
MMMM
MMMM
MMMM
MMMM

=====
THAT WAS AWESOME--JUST AWESOME!
=====

HIT THE ENTER KEY TO CONTINUE?

=====
HERE ARE THE TRICKS I CAN DO:

- 1 TURN DIRT INTO GOLD
- 2 WALK ON WATER
- 3 CUT A WOMAN IN HALF
- 4 MAKE A TWO-TON CAR VANISH
- 5 MAKE AN ELEPHANT APPEAR
- 6 PULL A RABBIT OUT OF A HAT
- 7 QUIT

=====
CHOOSE ONE? 7

I QUIT!

=====
THAT WAS INCREDIBLY WONDERFUL!
=====

☐ Program Listing

```
10 REM COMPUTER MAGICIAN
100 PRINT CHR$(12)
110 PRINT "I CAN PERFORM INCREDIBLE FEATS"
120 PRINT "OF MAGIC! JUST WATCH!"
130 PRINT
140 GOSUB 1850
150 PRINT
160 PRINT "HERE ARE THE TRICKS I CAN DO:"
170 PRINT
180 A$="TURN DIRT INTO GOLD"
190 B$="WALK ON WATER"
200 C$="CUT A WOMAN IN HALF"
210 D$="MAKE A TWO-TON CAR VANISH"
220 E$="MAKE AN ELEPHANT APPEAR"
230 F$="PULL A RABBIT OUT OF A HAT"
240 W$="WATER"+CHR$(32)
250 PRINT TAB(5);"1";TAB(8);A$
260 PRINT TAB(5);"2";TAB(8);B$
270 PRINT TAB(5);"3";TAB(8);C$
280 PRINT TAB(5);"4";TAB(8);D$
290 PRINT TAB(5);"5";TAB(8);E$
```

(continued)

```

300 PRINT TAB(5);"6";TAB(8);F$
310 PRINT TAB(5);"7";TAB(8);"QUIT"
320 PRINT
330 GOSUB 1850
340 PRINT
350 PRINT "CHOOSE ONE";
360 INPUT K
370 IF K<1 OR K>7 THEN 350
380 IF K<7 THEN 430
390 PRINT
400 PRINT "I QUIT!"
410 GOSUB 1680
420 END
430 PRINT
440 PRINT "WATCH ME ";
450 ON K GOTO 460,560,720,830,1160,1300
460 PRINT A$;"!"
470 GOSUB 1620
480 PRINT CHR$(12)
490 PRINT "HERE IT IS: PLAIN OLD D I R T"
500 GOSUB 1620
510 PRINT CHR$(12)
520 PRINT "PRESTO!";TAB(23);"G O L D!"
530 GOSUB 1680
540 GOSUB 1620
550 GOTO 130
560 PRINT B$;"!"
570 PRINT
580 GOSUB 1620
590 FOR M=1 TO 32
600 PRINT CHR$(12)
610 PRINT TAB(M);"WALK"
620 FOR NN=1 TO 6
630 PRINT W$;
640 NEXT NN
650 PRINT
660 FOR MM=1 TO 100
670 NEXT MM
680 NEXT M
690 PRINT
700 GOSUB 1680
710 GOTO 130
720 PRINT C$;"!"
730 GOSUB 1620
740 PRINT CHR$(12)
750 PRINT "OKAY: HERE'S ";CHR$(34);"A WOMAN";CHR$(34)
760 GOSUB 1620
770 PRINT CHR$(12)
780 PRINT TAB(14);CHR$(34);"A WO";CHR$(32);"MAN";CHR$(34)
790 PRINT
800 PRINT TAB(6);"SEE? I CUT A WOMAN IN HALF!"
810 GOSUB 1680
820 GOTO 130
830 PRINT D$;"!"
840 GOSUB 1620

```

```

850 PRINT "OKAY!  HERE IT IS:"
860 PRINT
870 FOR T=1 TO 4
880 PRINT TAB(9);
890 FOR U=1 TO 13-(5-T/2)
900 PRINT "M";
910 NEXT U
920 PRINT
930 NEXT T
940 FOR V=1 TO 5
950 PRINT TAB(2);
960 FOR W=1 TO 30-(6-V/2)
970 IF V>3 THEN 1010
980 IF W>1 THEN 1010
990 PRINT CHR$(32);
1000 GOTO 1020
1010 PRINT "M";
1020 NEXT W
1030 PRINT
1040 NEXT V
1050 PRINT TAB(5);"(( )";TAB(19);"(( )"
1060 PRINT TAB(6);"---";TAB(20);"---"
1070 PRINT
1080 GOSUB 1620
1090 PRINT CHR$(12)
1100 FOR Y=1 TO 10
1110 PRINT
1120 NEXT Y
1130 PRINT TAB(10);"POOF!  GONE!"
1140 GOSUB 1680
1150 GOTO 130
1160 PRINT E$;"!"
1170 GOSUB 1620
1180 PRINT TAB(9);
1190 FOR X=6 TO 16
1200 PRINT MID$(E$,X,1);
1210 FOR Q=1 TO 600
1220 NEXT Q
1230 NEXT X
1240 PRINT
1241 PRINT
1250 PRINT "SORRY THAT WAS SO SLOW, BUT THESE BIG"
1260 PRINT "ANIMALS ARE HARD TO MOVE QUICKLY."
1280 GOSUB 1680
1290 GOTO 130
1300 PRINT F$;"!"
1310 S=0
1320 GOSUB 1620
1330 FOR I=1 TO 20
1340 ON INT(RND(1)*6)+1 GOTO 1350,1370,1390,1410,1430,1450
1350 PRINT "OOPS!  THE RABBIT ESCAPED!"
1360 GOTO 1460
1370 PRINT "HOLD ON!  I LOST THE HAT!"
1380 GOTO 1460

```

(continued)

```

1390 PRINT "ONE SEC! THE RABBIT'S WEARING THE HAT!"
1400 GOTO 1460
1410 PRINT "HEY! STOP! THE RABBIT'S EATING THE HAT!"
1420 GOTO 1460
1430 PRINT "THE RABBIT IS LOOSE IN YOUR COMPUTER!"
1440 GOTO 1460
1450 PRINT "THE HAT JUST FELL ON THE RABBIT!"
1460 FOR R=1 TO 400
1470 NEXT R
1480 PRINT
1490 NEXT I
1500 PRINT "OKAY: NOW WE'RE SET!"
1510 GOSUB 1620
1520 GOSUB 1910
1530 S=1
1540 FOR U=6 TO 13
1550 GOSUB 1910
1560 NEXT U
1570 PRINT
1580 GOSUB 1680
1590 GOSUB 1620
1600 GOTO 130
1610 REM --- NEXT LINE SUBROUTINE ---
1620 PRINT
1630 PRINT "HIT THE ENTER KEY TO CONTINUE";
1640 INPUT L$
1650 PRINT
1660 RETURN
1670 REM --- BOAST SUBROUTINE ---
1680 PRINT
1690 GOSUB 1850
1700 PRINT TAB(5); "THAT WAS ";
1710 ON (RND(1)*5)+1 GOTO 1720,1740,1760,1780,1800
1720 PRINT "ABSOLUTELY SENSATIONAL!"
1730 GOTO 1810
1740 PRINT "INCREDIBLY WONDERFUL!"
1750 GOTO 1810
1760 PRINT "AWESOME--JUST AWESOME!"
1770 GOTO 1810
1780 PRINT "FANTASTICALLY TERRIFIC!"
1790 GOTO 1810
1800 PRINT "TOO AMAZING FOR WORDS!"
1810 GOSUB 1850
1820 PRINT
1830 RETURN
1840 REM --- LINE SUBROUTINE ---
1850 FOR Z=1 TO 39
1860 PRINT "=";
1870 NEXT Z
1880 PRINT
1890 RETURN
1900 REM --- HAT SUBROUTINE ---
1910 PRINT CHR$(12)
1920 FOR G=1 TO 14
1930 PRINT

```

```

1940 NEXT G
1950 IF S=0 THEN 1980
1960 PRINT TAB(9+(14-U)/2);MID$(F$,6,U-5);
1970 PRINT
1980 PRINT TAB(10);"MMMMMMMM"
1990 FOR H=1 TO 6
2000 PRINT TAB(12);
2010 FOR N=1 TO 4
2020 PRINT "M";
2030 NEXT N
2040 PRINT
2050 NEXT H
2060 PRINT
2070 RETURN

```

☐ If You Have . . .

APPLE II

Change: 100 HOME
 480 HOME
 510 HOME
 600 HOME
 740 HOME
 770 HOME
 1090 HOME
 1630 PRINT "HIT RETURN KEY TO CONTINUE"
 1910 HOME
 1960 PRINT TAB(9.5+(14-U)/2);MID(F\$,6,U-5);

ATARI

Add: 95 DIM A\$(25),B\$(25),C\$(25),D\$(25),E\$(25),F\$(27),
 W\$(25),L\$(1)
 2085 FOR AA=1 TO IDX
 2095 PRINT " ";:NEXT AA:RETURN

Change: 100 PRINT CHR\$(125)
 240 W\$="WATER "
 250 IDX=4:GOSUB 2085:PRINT "1";:GOSUB 2085:PRINT A\$
 260 GOSUB 2085:PRINT "2";:GOSUB 2085:PRINT B\$
 270 GOSUB 2085:PRINT "3";:GOSUB 2085:PRINT C\$
 280 GOSUB 2085:PRINT "4";:GOSUB 2085:PRINT D\$
 290 GOSUB 2085:PRINT "5";:GOSUB 2085:PRINT E\$
 300 GOSUB 2085:PRINT "6";:GOSUB 2085:PRINT F\$
 310 GOSUB 2085:PRINT "7";:GOSUB 2085:PRINT "QUIT"
 480 PRINT CHR\$(125)
 510 PRINT CHR\$(125)

(continued)

```

520 PRINT "PRESTO!":IDX=16:GOSUB 2085:PRINT
    "G O L D!"
600 PRINT CHR$(125)
610 IDX=M:GOSUB 2085:PRINT "WALK"
740 PRINT CHR$(125)
770 PRINT CHR$(125)
780 IDX=13:GOSUB 2085:PRINT CHR$(34);"A WO";CHR$(32);
    "MAN";CHR$(34)
800 IDX=5:GOSUB 2085:PRINT "SEE? I CUT A WOMAN
    IN HALF!"
880 IDX=8:GOSUB 2085
950 IDX=2:GOSUB 2085
1050 IDX=4:GOSUB 2085:PRINT "(( ))";:IDX=8:GOSUB
    2085:PRINT "(( ))"
1060 IDX=5:GOSUB 2085:PRINT "---";:IDX=10:GOSUB
    2085:PRINT "---"
1090 PRINT CHR$(125)
1130 IDX=9:GOSUB 2085:PRINT "POOF! GONE!"
1180 IDX=8:GOSUB 2085
1200 PRINT E$(X,X);
1340 ON INT(RND(0)*6)+1 GOTO 1350,1370,1390,1410,
    1430,1450
1630 PRINT "HIT RETURN KEY TO CONTINUE"
1700 IDX=4:GOSUB 2085:PRINT "THAT WAS ";
1710 ON (RND(0)*5)+1 GOTO 1720,1740,1760,1780,1800
1850 FOR Z=1 TO 37
1910 PRINT CHR$(125)
1960 IDX=(9+(14-U)/2):GOSUB 2085:PRINT F$(6,
    (6+CU-5));
1980 IDX=9:GOSUB 2085:PRINT "MMMMMMMM"
2000 IDX=11:GOSUB 2085

```

COMMODORE 64

Change:

```

100 PRINT CHR$(147)
480 PRINT CHR$(147)
510 PRINT CHR$(147)
600 PRINT CHR$(147)
740 PRINT CHR$(147)
770 PRINT CHR$(147)
1090 PRINT CHR$(147)
1340 ON INT(RND(0)*6)+1 GOTO 1350,1370,1390,1410,
    1430,1450
1630 PRINT "HIT RETURN KEY TO CONTINUE";
1710 ON (RND(0)*5)+1 GOTO 1720,1740,1760,1780,1800
1910 PRINT CHR$(147)

```

COMMODORE VIC-20

Add:

```

125 GOSUB 1620
705 GOSUB 1620
815 GOSUB 1620

```

1145 GOSUB 1620

1285 GOSUB 1620

Change: 100 PRINT CHR\$(147)
350 PRINT "CHOOSE ONE"
480 PRINT CHR\$(147)
510 PRINT CHR\$(147)
590 FOR M=1 TO 18
600 PRINT CHR\$(147)
620 FOR NN=1 TO 4
740 PRINT CHR\$(147)
770 PRINT CHR\$(147)
780 PRINT TAB(6);CHR\$(34);"A WO";CHR\$(32);"MAN";
CHR\$(34)
880 PRINT TAB(4);
890 FOR U=1 TO 10-(5-T/2)
960 FOR W=1 TO 20-(6-V/2)
1050 PRINT TAB(2);"(()";TAB(12);"(()"
1060 PRINT TAB(3);"---";TAB(13);"---"
1090 PRINT CHR\$(147)
1340 ON INT(RND(0)*6)+1 GOTO 1350,1370,1390,1410,
1430,1450
1630 PRINT "HIT RETURN KEY TO CONTINUE"
1710 ON (RND(0)*5)+1 GOTO 1720,1740,1760,1780,1800
1850 FOR Z=1 TO 21
1910 PRINT CHR\$(147)

TEXAS INSTRUMENTS 99/4A

Change: 100 CALL CLEAR
110 PRINT "I CAN PERFORM INCREDIBLE FEATS"
240 W\$="WATER"&CHR\$(32)&CHR\$(32)&CHR\$(32)
370 IF (K<1)+(K>7) THEN 350
480 CALL CLEAR
490 PRINT "HERE IT IS:PLAIN OLD D I R T"
510 CALL CLEAR
590 FOR M=1 TO 24
600 CALL CLEAR
620 FOR NN=1 TO 4
740 CALL CLEAR
770 CALL CLEAR
1090 CALL CLEAR
1200 PRINT SEG\$(E\$,X,1);
1210 FOR Q=1 TO 100
1260 PRINT "ANIMALS ARE HARD TO MOVE QUICKLY."
1340 ON INT(RND*6)+1 GOTO 1350,1370,1390,1410,
1430,1450
1390 PRINT "ONE SEC! THE RABBIT'S WEAR- ING THE
HAT!"
1410 PRINT "HEY! STOP! THE RABBIT'S EAT-ING THE
HAT!"
1450 PRINT "THE HAT JUST FELL ON THE RABBIT!"
1710 ON (RND*5)+1 GOTO 1720,1740,1760,1780,1800

(continued)

```

1850 FOR Z=1 TO 28
1910 CALL CLEAR
1960 PRINT TAB(9+(14-U)/2);SEG$(F$,6,U-5);

```

TRS-80 COLOR COMPUTER

Delete: 150,170,320,540,790,860,1070,1590

Add: 1825 INPUT X\$

Change:

```

100 CLS
130 REM
250 PRINT TAB(2);"1";TAB(5);A$
260 PRINT TAB(2);"2";TAB(5);B$
270 PRINT TAB(2);"3";TAB(5);C$
280 PRINT TAB(2);"4";TAB(5);D$
290 PRINT TAB(2);"5";TAB(5);E$
300 PRINT TAB(2);"6";TAB(5);F$
310 PRINT TAB(2);"7";TAB(5);"QUIT"
480 CLS
510 CLS
590 FOR M=1 TO 25
600 CLS
620 FOR NN=1 TO 5
740 CLS
770 CLS
1090 CLS
1340 ON RND(6) GOTO 1350,1370,1390,1410,1430,1450
1680 REM
1700 PRINT "THAT WAS ";
1710 ON RND(5) GOTO 1720,1740,1760,1780,1800
1820 PRINT "HIT ENTER KEY TO CONTINUE";
1850 FOR Z=1 TO 31
1910 CLS

```

Going Down!

Imagine! You're in the biggest computer store in the world, but you're stuck on the top floor — where all you can find is boring old business software.

Help is on the way!

☐ Sample Run

ENTER YOUR WHOLE NAME? LINDSAY WOOLSEY

HERE WE ARE IN THE BIGGEST COMPUTER
STORE IN THE WORLD. THE PROBLEM IS
THAT WE'RE ON THE TOP FLOOR AND THE
NEW SHIPMENT OF GAMES AND PROGRAMS
IS ON THE GROUND FLOOR. AND TO TOP
IT ALL OFF, THE ELEVATORS ARE NOT
WORKING. MAYBE IF WE TRIED HITTING
THE ENTER KEY WE COULD FIND AN
ESCALATOR DOWN?

L
I
N
D
S
A
Y

W
O
O
L
S
E
Y

Y

E
S
L
O
O
W

Y
A
S
D
N
I
L

Break

☐ Program Listing

```
10 REM GOING DOWN!
100 K=0
110 PRINT "ENTER YOUR WHOLE NAME";
120 INPUT WNAM$
130 LNAM=LEN(WNAM$)
140 IF LNAM=0 THEN 110
150 PRINT CHR$(12)
160 PRINT "HERE WE ARE IN THE BIGGEST COMPUTER"
170 PRINT "STORE IN THE WORLD. THE PROBLEM IS"
180 PRINT "THAT WE'RE ON THE TOP FLOOR AND THE"
190 PRINT "NEW SHIPMENT OF GAMES AND PROGRAMS"
200 PRINT "IS ON THE GROUND FLOOR. AND TO TOP"
210 PRINT "IT ALL OFF, THE ELEVATORS ARE NOT"
220 PRINT "WORKING. MAYBE IF WE TRIED HITTING"
230 PRINT "THE ENTER KEY WE COULD FIND AN"
240 PRINT "ESCALATOR DOWN";
250 INPUT D$
260 FOR A=1 TO 25
270 PRINT
280 NEXT A
290 FOR B=1 TO LNAM
300 PRINT TAB(2*B);MID$(WNAM$,B,1)
310 NEXT B
320 PRINT
330 FOR C=LNAM TO 1 STEP -1
340 PRINT TAB(2*C);MID$(WNAM$,C,1)
350 NEXT C
360 PRINT
370 K=K+1
380 IF K=500 THEN 400
390 GOTO 290
400 PRINT "HOORAY! WE FOUND ALL THE GAMES!"
```

☐ If You Have . . .

APPLE II

Change: 150 HOME
230 PRINT "THE RETURN KEY WE COULD FIND AN"

ATARI

Add: 95 DIM WNAM\$(25),D\$(1)
405 END
415 FOR AA=1 TO IDX
425 PRINT " ";:NEXT AA:RETURN

Change: 150 PRINT CHR\$(125)
230 PRINT "THE RETURN KEY WE COULD FIND AN"
300 IDX=(2*B):GOSUB 415:PRINT WNAM\$(B,B)
340 IDX=(2*C):GOSUB 415:PRINT WNAM\$(C,C)

COMMODORE 64

Change: 150 PRINT CHR\$(147)
230 PRINT "THE RETURN KEY WE COULD FIND AN"

COMMODORE VIC-20

Change: 110 PRINT "ENTER YOUR WHOLE NAME"
150 PRINT CHR\$(147)
230 PRINT "THE RETURN KEY WE COULD FIND AN"
240 PRINT "ESCALATOR DOWN"

TEXAS INSTRUMENTS 99/4A

Change: 150 CALL CLEAR
160 PRINT "HERE WE ARE IN THE BIGGEST COMPUTER"
170 PRINT "STORE IN THE WORLD. THE PROBLEM IS"
190 PRINT "NEW SHIPMENT OF GAMES AND PROGRAMS"
210 PRINT "IT ALL OFF, THE ELEVATORS ARE NOT"
300 PRINT TAB(2*B);SEG\$(WNAM\$,B,1)
340 PRINT TAB(2*C);SEG\$(WNAM\$,C,1)
380 IF K=20 THEN 400
400 PRINT "HOORAY! WE FOUND ALL THE GAMES!"

TRS-80 COLOR COMPUTER

Delete: 270

Change: 150 CLS

Computer Speedway

Now you can drive your very favorite sports car through the toughest race course anywhere. But watch out! Go too fast, and you may find yourself in big trouble!

(Important note! Each section of the course is exactly one-seventh of a mile long. And the average speeds are correct — even though they may seem wrong!

Say you do one mile at 3 mph, the next at 4 mph, and the last one at 5 mph. You might expect your average speed to be 4 mph — but it's not! You can't simply take an average of average speeds.

To figure an overall average speed, you need to know the total amount of time it took you to travel the course—in this case, one-third or .33 hours for the first lap, one-fourth or .25 hours for the second, and one-fifth or .20 for the third. The total time? $.33 + .25 + .20 = .78$ hours.

To find the average speed, you divide the total distance — 3 miles — by the total time — .78 hours. $3/.78 = 3.846$ miles per hour.

It sounds crazy — but it's true!)

☐ Sample Run

WE'RE HERE AT THE COMPUTER SPEEDWAY
WHERE RACERS CAN TRY THEIR LUCK AT THE
FASTEST, TWISTIEST TRACK IN THE WORLD.

WHAT'S YOUR NAME? ELIZABETH

WHAT CAR ARE YOU DRIVING? PORSCHE

HOW MANY LAPS (1 - 30) WILL YOU DO? 3

DO YOU WANT TO PLAY THE HARD VERSION? NO

OK, ELIZABETH! START YOUR PORSCHE AND
HIT THE ENTER KEY WHEN YOU'RE READY?

** YOU'RE NOW STARTING LAP NUMBER 1 **

YOU'RE IN THE STRAIGHTAWAY
THE TOP SAFE SPEED HERE IS 150
HOW FAST DO YOU WANT TO GO? 160

YOU'RE IN THE SHARP LEFT CURVE
THE TOP SAFE SPEED HERE IS 40
HOW FAST DO YOU WANT TO GO? 40

A NEARBY CAR SPINS OUT OF CONTROL...
BUT YOU DRIVE SKILLFULLY AROUND IT.

YOU'RE IN THE SNAKY S-CURVES
THE TOP SAFE SPEED HERE IS 80
HOW FAST DO YOU WANT TO GO? 85
A NEARBY CAR SPINS OUT OF CONTROL...
BUT YOU DRIVE SKILLFULLY AROUND IT.

YOU'RE IN THE HAIRPIN RIGHT TURN
THE TOP SAFE SPEED HERE IS 20
HOW FAST DO YOU WANT TO GO? 20
ANOTHER DRIVER CUTS RIGHT IN FRONT...
BUT YOU DRIVE SKILLFULLY AROUND HIM.

YOU'RE IN THE GENTLE RIGHT TURN
THE TOP SAFE SPEED HERE IS 140
HOW FAST DO YOU WANT TO GO? 150

YOU'RE IN THE MEDIUM LEFT TURN
THE TOP SAFE SPEED HERE IS 100
HOW FAST DO YOU WANT TO GO? 110

YOU'RE IN THE EASY RIGHT TURN
THE TOP SAFE SPEED HERE IS 160
HOW FAST DO YOU WANT TO GO? 155

** YOU'RE NOW STARTING LAP NUMBER 2 **
AVERAGE SPEED LAST LAP WAS 60.75129 MPH

YOU'RE IN THE STRAIGHTAWAY
THE TOP SAFE SPEED HERE IS 190
HOW FAST DO YOU WANT TO GO? 200

YOU'RE IN THE SHARP LEFT CURVE
THE TOP SAFE SPEED HERE IS 90
HOW FAST DO YOU WANT TO GO? 90
YOU SEE A DEEP GOUGE IN THE TRACK...
BUT YOU DRIVE SKILLFULLY AROUND IT.

YOU'RE IN THE SNAKY S-CURVES
THE TOP SAFE SPEED HERE IS 50
HOW FAST DO YOU WANT TO GO? 60
ANOTHER DRIVER CUTS RIGHT IN FRONT...
BUT YOU DRIVE SKILLFULLY AROUND HIM.

YOU'RE IN THE HAIRPIN RIGHT TURN
THE TOP SAFE SPEED HERE IS 50
HOW FAST DO YOU WANT TO GO? 55

YOU'RE IN THE GENTLE RIGHT TURN
THE TOP SAFE SPEED HERE IS 140
HOW FAST DO YOU WANT TO GO? 145

(continued)

.....SCREEECHHH.....CRASH!!!!
NOTHING 1 COAT OF PAINT WON'T CURE.
IT COSTS YOU 4 SECONDS.

YOU'RE IN THE MEDIUM LEFT TURN
THE TOP SAFE SPEED HERE IS 100
HOW FAST DO YOU WANT TO GO? 110
YOU SEE A DEEP GOUGE IN THE TRACK...
BUT YOU DRIVE SKILLFULLY AROUND IT.

YOU'RE IN THE EASY RIGHT TURN
THE TOP SAFE SPEED HERE IS 150
HOW FAST DO YOU WANT TO GO? 150

** YOU'RE NOW STARTING LAP NUMBER 3 **
THIS IS YOUR FINAL LAP!!
AVERAGE SPEED LAST LAP WAS 54.66266 MPH

YOU'RE IN THE STRAIGHTAWAY
THE TOP SAFE SPEED HERE IS 180
HOW FAST DO YOU WANT TO GO? 195
ANOTHER DRIVER CUTS RIGHT IN FRONT...
BUT YOU DRIVE SKILLFULLY AROUND HIM.

YOU'RE IN THE SHARP LEFT CURVE
THE TOP SAFE SPEED HERE IS 70
HOW FAST DO YOU WANT TO GO? 70
YOU SEE A DEEP GOUGE IN THE TRACK...
BUT YOU DRIVE SKILLFULLY AROUND IT.

YOU'RE IN THE SNAKY S-CURVES
THE TOP SAFE SPEED HERE IS 90
HOW FAST DO YOU WANT TO GO? 95
YOU SEE A DEEP GOUGE IN THE TRACK...
BUT YOU DRIVE SKILLFULLY AROUND IT.

YOU'RE IN THE HAIRPIN RIGHT TURN
THE TOP SAFE SPEED HERE IS 40
HOW FAST DO YOU WANT TO GO? 40
ANOTHER DRIVER CUTS RIGHT IN FRONT...
BUT YOU DRIVE SKILLFULLY AROUND HIM.

YOU'RE IN THE GENTLE RIGHT TURN
THE TOP SAFE SPEED HERE IS 100
HOW FAST DO YOU WANT TO GO? 105
.....SCREEECHHH.....CRASH!!!!
LEAVING LOTS OF METAL ON THE ASPHALT!
IT COSTS YOU 16 SECONDS.

YOU'RE IN THE MEDIUM LEFT TURN
THE TOP SAFE SPEED HERE IS 110
HOW FAST DO YOU WANT TO GO? 100

YOU'RE IN THE EASY RIGHT TURN
THE TOP SAFE SPEED HERE IS 120
HOW FAST DO YOU WANT TO GO? 125
YOU SEE A DEEP GOUGE IN THE TRACK...
BUT YOU DRIVE SKILLFULLY AROUND IT.

AVERAGE SPEED LAST LAP WAS 23.31454 MPH

STATS FOR PORSCHE DRIVEN BY ELIZABETH

LAP	MPH
1	60.75129
2	54.66266
3	23.31454

TOTAL LAPS: 3
AVERAGE OVERALL SPEED 38.63676 MPH
BUT YOU CRASHED 2 TIMES.

☐ Program Listing

```
10 REM COMPUTER SPEEDWAY
100 PRINT "WE'RE HERE AT THE COMPUTER SPEEDWAY"
110 PRINT "WHERE RACERS CAN TRY THEIR LUCK AT THE"
120 PRINT "FASTEST, TWISTIEST TRACK IN THE WORLD."
130 PRINT
140 PRINT "WHAT'S YOUR NAME";
150 INPUT NAM$
160 IF NAM$="" THEN 140
170 DIM GT(30)
180 PRINT "WHAT CAR ARE YOU DRIVING";
190 INPUT CAR$
200 IF CAR$="" THEN 180
210 PRINT
220 PRINT "HOW MANY LAPS (1 - 30) WILL YOU DO";
230 INPUT LT$
240 LT=VAL(LT$)
250 IF LT=0 OR LT>30 THEN 220
260 LT=LT+1
270 RANDOMIZE(LT*ASC(MID$(CAR$,1,1)))
280 PRINT
290 PRINT "DO YOU WANT TO PLAY THE HARD VERSION";
300 INPUT H$
310 H$=MID$(H$,1,1)
320 IF H$<>"Y" AND H$<>"N" THEN 290
330 PRINT
340 PRINT "OK, ";NAM$;"! START YOUR ";CAR$;" AND"
350 PRINT "HIT THE ENTER KEY WHEN YOU'RE READY";
360 INPUT X$
370 K=1
```

(continued)

```

380 LAP=LAP+1
390 PRINT
400 GOSUB 1300
410 IF LAP=LT THEN 450
420 PRINT "*** YOU'RE NOW STARTING LAP NUMBER";LAP;"**"
430 IF LAP=LT-1 THEN PRINT TAB(8);"THIS IS YOUR FINAL LAP!!"
440 IF LAP<2 THEN 560
450 MN=0
460 FOR J=1 TO 7
470 MN=MN+(1/TS(J)/7)+(WT/3600)
480 NEXT J
490 AV=1/MN
500 PRINT "AVERAGE SPEED LAST LAP WAS";AV;"MPH"
510 WT=0
520 M=M+1
530 GT(M)=MN
540 IF LAP=LT THEN 1360
550 LET K=1
560 GOSUB 1300
570 F=13
580 A$="STRAIGHTAWAY"
590 GOSUB 800
600 F=4
610 A$="SHARP LEFT CURVE"
620 GOSUB 800
630 F=3
640 A$="SNAKY S-CURVES"
650 GOSUB 800
660 F=2
670 A$="HAIRPIN RIGHT TURN"
680 GOSUB 800
690 F=9
700 A$="GENTLE RIGHT TURN"
710 GOSUB 800
720 F=8
730 A$="MEDIUM LEFT TURN"
740 GOSUB 800
750 F=10
760 A$="EASY RIGHT TURN"
770 GOSUB 800
780 GOTO 380
790 REM --- SPEED SUBROUTINE ---
800 PRINT
810 PRINT "YOU'RE IN THE ";A$
820 L=(INT(RND(1)*7)+F)*10
830 PRINT "THE TOP SAFE SPEED HERE IS ";L
840 PRINT "HOW FAST DO YOU WANT TO GO";
850 INPUT S$
860 S=VAL(S$)
870 IF S=0 THEN 840
880 TS(K)=S
890 K=K+1
900 HAZARD=0
910 X=INT(RND(1)*10)
920 IF S>L-X THEN GOSUB 1570

```

```

930 IF HAZARD<4 THEN 950
940 IF S>L-X THEN GOSUB 970
950 RETURN
960 REM --- CRASH SUBROUTINE ---
970 IF S/L>1.95 THEN 1000
980 IF H$="N" THEN 1100
990 PRINT
1000 PRINT "NOOO.....SCREEECHHH.....CRASH!!!!!"
1010 ON INT(RND(1)*4)+1 GOTO 1020,1040,1060,1080
1020 PRINT "YOU DISINTEGRATE INTO A WALL OF FLAME!"
1030 GOTO 1360
1040 PRINT "YOU ARE NOW A BURNED SMEAR ON THE ROAD!"
1050 GOTO 1360
1060 PRINT "YOU BREAK EVERY BONE BUT ARE ALIVE!"
1070 GOTO 1360
1080 PRINT "YOUR CAR IS TOTALLED BUT YOU WALK AWAY!"
1090 GOTO 1360
1100 IF S/L>1.6 THEN 1150
1110 IF S/L<1.3 THEN 1140
1120 IF INT(RND(1)*3)<>2 THEN 1280
1130 GOTO 1280
1140 IF INT(RND(1)*6)<>2 THEN 1280
1150 PRINT ".....SCREEECHHH.....CRASH!!!!!"
1160 ON INT(RND(1)*3)+1 GOTO 1170,1190,1230
1170 PRINT "LEAVING LOTS OF METAL ON THE ASPHALT!"
1180 GOTO 1240
1190 IF CR<1 THEN C=2
1200 C=CR+1
1210 PRINT "NOTHING";C;"COATS OF PAINT WON'T CURE."
1220 GOTO 1240
1230 PRINT "ANOTHER OF THOSE AND YOU'RE IN TROUBLE!"
1240 CR=CR+1
1250 Y=INT(RND(1)*15)+2
1260 PRINT "IT COSTS YOU ";Y; " SECONDS."
1270 WT=WT+Y
1280 RETURN
1290 REM --- LINE SUBROUTINE ---
1300 FOR A=1 TO 38
1310 PRINT "-";
1320 NEXT A
1330 PRINT "- "
1340 RETURN
1350 REM --- FINAL RESULTS ---
1360 IF LAP=1 THEN END
1370 PRINT
1380 PRINT "STATS FOR ";CAR$;" DRIVEN BY ";NAM$
1390 PRINT
1400 PRINT "LAP";TAB(17);"MPH"
1410 PRINT
1420 FOR J=1 TO LAP-1
1430 PRINT J,1/GT(J)
1440 ET=ET+GT(J)
1450 NEXT J
1460 PRINT
1470 PRINT "TOTAL LAPS:";LAP-1

```

(continued)

```

1480 PRINT "AVERAGE OVERALL SPEED";(LAP-1)/ET;"MPH"
1490 IF CR=0 THEN 1540
1500 IF CR>1 THEN 1530
1510 PRINT "BUT YOU CRASHED";CR;"TIME."
1520 GOTO 1540
1530 PRINT "BUT YOU CRASHED";CR;"TIMES."
1540 GOSUB 1300
1550 END
1560 REM --- RANDOM HAZARD SUBROUTINE ---
1570 HAZARD=INT(RND(1)*11)+1
1580 ON HAZARD GOTO 1600,1640,1720,1790,1850
1590 GOTO 1850
1600 PRINT "YOU SEE A DEEP GOUGE IN THE TRACK..."
1610 IF S/L>1.5 THEN 1830
1620 PRINT "BUT YOU DRIVE SKILLFULLY AROUND IT."
1630 GOTO 1850
1640 PRINT "ANOTHER DRIVER CUTS RIGHT IN FRONT..."
1650 IF S/L>1.4 THEN 1830
1660 PRINT "BUT YOU DRIVE SKILLFULLY AROUND HIM."
1670 GOTO 1850
1680 PRINT "SUDDENLY OIL IS ALL OVER THE TRACK..."
1690 IF S/L>1.3 THEN 1830
1700 PRINT "BUT YOU DRIVE SKILLFULLY AROUND IT."
1710 GOTO 1850
1720 PRINT "YOU NOTICE YOUR BRAKES HAVE FAILED..."
1730 IF S/L>1.2 THEN 1830
1740 PRINT "BUT YOU COAST ARTFULLY INTO THE PITS."
1750 Z=INT(RND(1)*15)+2
1760 PRINT "IT COSTS YOU ";Z;" SECONDS."
1770 WT=WT+Z
1780 GOTO 1850
1790 PRINT "A NEARBY CAR SPINS OUT OF CONTROL..."
1800 IF S/L>1.1 THEN 1830
1810 PRINT "BUT YOU DRIVE SKILLFULLY AROUND IT."
1820 GOTO 1850
1830 PRINT "AND YOU ARE GOING TOO FAST TO STOP!"
1840 GOTO 1000
1850 RETURN

```

☐ If You Have . . .

APPLE II

Delete: 270

Change: 350 PRINT "HIT THE RETURN KEY WHEN YOU'RE READY";
420 PRINT "** YOU'RE NOW STARTING LAP NUMBER ";
LAP;"**"
500 PRINT "AVERAGE SPEED LAST LAP WAS ";AV;"
MPH"
1210 PRINT "NOTHING ";C;" COATS OF PAINT WON'T
CURE."

```

1480 PRINT "AVERAGE OVERALL SPEED ";(LAP-1)/ET;
      " MPH"
1510 PRINT "BUT YOU CRASHED ";CR;" TIME."
1530 PRINT "BUT YOU CRASHED ";CR;" TIMES."

```

ATARI

Delete: 170,270

```

Add: 95 DIM NAM$(40),CAR$(17),LT$(2),H$(3),X$(1),A$
      (22),S$(3)
96 DIM TS(30),GT(30)
435 IDX=8:GOSUB 1865:PRINT "THIS IS YOUR FINAL
      LAP!!"
1865 FOR AA=I TO IDX
1875 PRINT " ";:NEXT AA:RETURN

```

Change:

```

310 H$=H$(1,1)
350 PRINT "HIT THE RETURN KEY WHEN YOU'RE READY";
430 IF LAP<>LT-1 THEN 440
820 L=(INT(RND(0)*7)+F)*10
910 X=INT(RND(0)*10)
1010 ON INT(RND(0)*4)+1 GOTO 1020,1040,1060,1080
1120 IF INT(RND(0)*3)<>2 THEN 1280
1140 IF INT(RND(0)*6)<>2 THEN 1280
1160 ON INT(RND(0)*3)+1 GOTO 1170,1190,1230
1250 Y=INT(RND(0)*15)+2
1300 FOR A=1 TO 36
1400 PRINT "LAP ";:IDX=14:GOSUB 1865:PRINT "MPH"
1570 HAZARD=INT(RND(0)*11)+1
1750 Z=INT(RND(0)*15)+2

```

COMMODORE 64

Change:

```

270 REM
350 PRINT "HIT THE RETURN KEY WHEN YOU'RE READY";
820 L=(INT(RND(0)*7)+F)*10
910 X=INT(RND(0)*10)
1010 ON INT(RND(0)*4)+1 GOTO 1020,1040,1060,1080
1120 IF INT(RND(0)*3)<>2 THEN 1280
1140 IF INT(RND(0)*6)<>2 THEN 1280
1160 ON INT(RND(0)*3)+1 GOTO 1170,1190,1230
1250 Y=INT(RND(0)*15)+2
1570 HAZARD=INT(RND(0)*11)+1
1750 Z=INT(RND(0)*15)+2

```

COMMODORE VIC-20 Program Will Not Run On Unexpanded VIC-20

(continued)

TEXAS INSTRUMENTS 99/4A

Add: 432 PRINT "THIS IS YOUR FINAL LAP!!"
922 GOSUB 1570
942 GOSUB 970
1192 C=2
1862 END

Change: 100 PRINT "WE'RE HERE AT THE COMPUTER SPEEDWAY"
110 PRINT "WHERE RACERS CAN TRY THEIR LUCK AT THE"
250 IF (LT=0)+(LT>30) THEN 220
270 RANDOMIZE
310 H\$=SEG\$(H\$,1,1)
320 IF (H\$<>"Y")*(H\$<>"N") THEN 290
350 PRINT "HIT THE ENTER KEY WHEN YOU ARE READY";
420 PRINT "** YOU'RE NOW STARTING LAP NUMBER";LAP;

430 IF LAP<>LT-1 THEN 440
820 L=(INT(RND*7)+F)*10
910 X=INT(RND*10)
920 IF S<=L-X THEN 930
940 IF S<=L-X THEN 950
1010 ON INT(RND*4)+1 GOTO 1020,1040,1060,1080
1040 PRINT "YOU ARE NOW A BURNED SMEAR ON THE ROAD!"
1120 IF INT(RND*3)<>2 THEN 1280
1140 IF INT(RND*6)<>2 THEN 1280
1160 ON INT(RND*3)+1 GOTO 1170,1190,1230
1190 IF CR>=1 THEN 1200
1250 Y=INT(RND*15)+2
1300 FOR A=1 TO 27
1360 IF LAP=1 THEN 1862
1570 HAZARD=INT(RND*5)+1
1620 PRINT "BUT YOU DRIVE SKILLFULLY AROUND IT."
1640 PRINT "ANOTHER DRIVER CUTS RIGHT IN FRONT..."
1660 PRINT "BUT YOU DRIVE SKILLFULLY AROUND HIM."
1700 PRINT "BUT YOU DRIVE SKILLFULLY AROUND IT."
1750 Z=INT(RND*15)+2
1790 PRINT "A NEARBY CAR SPINS OUT OF CONTROL..."
1810 PRINT "BUT YOU DRIVE SKILLFULLY AROUND IT."
1830 PRINT "AND YOU ARE GOING TOO FAST TO STOP!"

TRS-80 COLOR COMPUTER

Delete: 270,1370,1390

Add: 415 IF LAP=1 THEN 420
416 PRINT "HIT ENTER KEY TO START NEXT LAP"
417 INPUT X\$
1365 PRINT "HIT ENTER TO SEE FINAL RESULTS"
1366 INPUT X\$

Change: 430 IF LAP=LT-1 THEN PRINT TAB(5);"THIS IS YOUR
 FINAL LAP!!"
820 L=(INT(RND(7)+(F-1)))*10
910 X=RND(10)-1
1010 ON RND(4) GOTO 1020,1040,1060,1080
1120 IF RND(3)<>2 THEN 1280
1140 IF RND(6)<>2 THEN 1280
1160 ON RND(3) GOTO 1170,1190,1230
1250 Y=RND(15)+1
1300 FOR A=1 TO 30
1570 HAZARD=RND(11)
1750 Z=RND(15)+1

MORE FUN! LESS TYPING!

If you just finished typing in all the programs in this book and your pet gerbil ate the disk or cassette you saved them on...

If you're already enjoying the shorter programs in this book, but your fingers tremble at the thought of typing in the longer ones...

If your secretary does all your typing, but she's on a year-long vacation in Moose Jaw, Saskatchewan...

Or if you'd just like to have all these programs handy on a disk or cassette that'll run on your machine...

You're in luck!

You can get every single program in this book (along with extra bonus programs not included in this book) on ready-to-run cassettes or disks, for only \$19.95 per book. You can also get tapes or disks for any of the *other* exciting books in this series.

(As a special bonus, order any three, and we'll throw in the fourth one *absolutely free*—so you can enjoy the entire set—and we'll pay all postage and handling!)

Programs are available in one format only (disk *or* cassette) for each specific computer, so please make sure you have the proper equipment before ordering. Remember to check off BOTH the set of programs you want AND the type of machine you have.

Hard/Soft Inc., PO Box 1277, Riverdale, NY 10471

Yes! Please send me the ready-to-run programs I've checked off below.

1. Check off which format you want:

- ☐ ATARI cassette tapes
- ☐ APPLE II/II PLUS/IIe disks
- ☐ COMMODORE 64 disks
- ☐ COMMODORE VIC-20 disks
- ☐ IBM PC/PCjr disks
- ☐ TEXAS INSTRUMENTS 99/4A cassette tapes
- ☐ TRS-80 COLOR COMPUTER cassette tapes

2. Check off which programs you are ordering:

- ☐ Computer Crazyness (\$19.95)
- ☐ Computer Monsters (\$19.95)
- ☐ Computer Olympics (\$19.95)
- ☐ Computer Space Adventures (\$19.95)
- ☐ ALL FOUR BOOKS (SPECIAL PRICE \$59.85)

3. PRINT your name and address, and fill in the amount you are enclosing: (Add applicable sales taxes!)

I am enclosing \$19.95 PLUS \$2.00 postage and handling and applicable sales tax for each. Total: \$_____

☐ *SPECIAL BONUS: Check this box, enclose \$59.85, and we'll send you all FOUR sets of programs for the price of three—and we'll pay all postage! This is a savings of over \$20.00.*

PRINT your name _____

PRINT your address _____

City _____ State _____ Zip _____

Make checks payable to Hard/Soft, Inc. Allow 2-8 weeks for delivery.



YOUR COMPUTER IS CRAZIER THAN YOU THINK!

COMPUTER CRAZINESS is the wildest, funniest, most entertaining book of programs ever assembled. It's guaranteed to drive you and your friends bananas! These easy-to-type programs can make your computer grant your wishes, convince other people to handle your household chores, insult your friends, and turn the tables on your parents by giving *them* report cards. The dozens of puzzles and games here will keep you fascinated for hours.

This book will help you and your computer play terrific tricks on your friends. But watch out! Once you type in these programs, your computer will know a few tricks of its own. It may even pretend it's broken!

COMPUTER CRAZINESS includes programs with special secrets that may fool even you! And when you're not holding your sides from laughing, you can find your way out of a fun house, race a car at 200 MPH down a twisty track, or admire your own personal picture creations. This is the most fun you and your computer will ever have!

Other books by these authors:

**COMPUTER OLYMPICS
COMPUTER MONSTERS
COMPUTER SPACE ADVENTURES**



ISBN 0-590-33175-2


SCHOLASTIC INC.